> **#Prob 4.2**

> *restart* :

> $sys := D(y1)(t) = 2\cdot y1(t) - 3\cdot y2(t), D(y2)(t) = \left(\frac{3}{4}\right)\cdot y1(t) - 3\cdot y2(t);$

$$sys := D(y1)(t) = 2\,y1(t) - 3\,y2(t), D(y2)(t) = \frac{3\,y1(t)}{4} - 3\,y2(t)$$

> $dsolve(\{sys, y1(0) = 10, y2(0) = 0\});$

$$\left\{ y1(t) = -\frac{5\,e^{-\frac{5\,t}{2}}}{4} + \frac{45\,e^{\frac{3\,t}{2}}}{4}, y2(t) = -\frac{15\,e^{-\frac{5\,t}{2}}}{8} + \frac{15\,e^{\frac{3\,t}{2}}}{8} \right\}$$

> *with*(*DEtools*) :

>

>

> $DEplot([sys], [y1(t), y2(t)], t = 0..1, [[y1(0) = 3, y2(0) = 0]])$



> $DEplot([sys], [y1(t), y2(t)], t = 0..1, linecolor = black, [[y1(0) = 10, y2(0) = 0]], scaling = constrained, y1 = 0..40)$



> $DEplot(\{sys[1], sys[2]\}, \{y1(t), y2(t)\}, t = 0..1, linecolor = black, \{[y1(0) = 10, y2(0) = 0]\}, y1 = 0..40)$



> **#Prob 4.4**

> *with(DEtools)* :

> $sys := D(y1)(t) = -y1(t) + 9\cdot y2(t), D(y2)(t) = -4\cdot y1(t) - y2(t);$

$$sys := D(y1)(t) = -y1(t) + 9\, y2(t), D(y2)(t) = -4\, y1(t) - y2(t)$$

> $y := dsolve(\{sys[1], sys[2], y1(0) = 1, y2(0) = 1\});$

$$y := \left\{ y1(t) = e^{-t}\left(\cos(6\,t) + \frac{3\sin(6\,t)}{2}\right), y2(t) = \frac{2\,e^{-t}\left(\frac{3\cos(6\,t)}{2} - \sin(6\,t)\right)}{3} \right\}$$

> $init := [0, 1, 1];$

$$init := [0, 1, 1]$$

>

> $DEplot(\{sys[1], sys[2]\}, [y1(t), y2(t)], t = -4..4, y1 = -4..4, y2 = -4..4, scaling = constrained, \{init\}, linecolor = black, stepsize = 0.05, obsrange = false);$



Pr.4.10 Differentiate the second equation The first equation has the form needed for writing the system in the usual standard form,

```
> with(LinearAlgebra):
```

```
> restart:
```

The two equations are

```
> eq1 := D(i1)(t) = -3*i1(t) + 3*i2(t) + 24;
```

$$eq1 := \mathrm{D}(i1)(t) = -3\, i1(t) + 3\, i2(t) + 24$$

and (after differentiation)

```
> eq2 := 8*D(i2)(t) + 3*(D(i2)(t) - D(i1)(t)) + 4*i2(t) = 0;
```

$$eq2 := 11\,\mathrm{D}(i2)(t) - 3\,\mathrm{D}(i1)(t) + 4\, i2(t) = 0$$

or, using eq1,

```
> eq2 := 8*D(i2)(t) + 3*(D(i2)(t) - rhs(eq1)) + 4*i2(t) = 0;
```

$$eq2 := 11D(i2)(t) + 9i1(t) - 5i2(t) - 72 = 0$$

```
> solve(eq2, D(i2)(t));
```

\# Resp. $\frac{72}{11} - \frac{9}{11}i1(t) + \frac{5}{11}i2(t)$

```
> J := <<I1(t), I2(t)>>;
```

```
> A := <<-3  3> , <-9/11  5/11>>;
```

\# Resp. $A := \begin{bmatrix} -3 & 3 \\ -\frac{9}{11} & \frac{5}{11} \end{bmatrix}$

```
> g := <<24, 72/11>>:
```

```
> <<D(i1)(t), D(i2)(t)>> = A.J + g;
```

$$\begin{bmatrix} (D(i1))(t) \\ (D(i2))(t) \end{bmatrix} = \begin{bmatrix} -3I1(t) + 3I2(t) + 24 \\ -\frac{9}{11}I1(t) + \frac{5}{11}I2(t) + \frac{72}{11} \end{bmatrix}$$

For the homogeneous system, $\mathbf{J}' = \mathbf{AJ}$, and, letting $\mathbf{J} = \mathbf{x}e^{\lambda t}$ gives $\mathbf{J}' = \lambda \mathbf{x}e^{\lambda t}$ For a nontrivial solution, we use the eigenvalues and eigenvectors as follows.

```
> Eig := Eigenvectors(A);
```

\# Resp. $\begin{bmatrix} -2 \\ -\frac{6}{11} \end{bmatrix}, \begin{bmatrix} 3 & \frac{11}{9} \\ 1 & 1 \end{bmatrix}$

```
> Jh := c1*Eig[2][1..2, 1]*exp(Eig[1][1]*t)
   + c2*Eig[2][1..2, 2]*exp(Eig[1][2]*t);
```

$$Jh := \begin{bmatrix} 3c1e^{-2t} + \frac{11}{9}c2e^{-\frac{6}{11}t} \\ c1e^{-2t} + c2e^{-\frac{6}{11}t} \end{bmatrix}$$

For the particular solution of the non-homogeneous system $\mathbf{J}' = \mathbf{AJ} + \mathbf{g}$, we note that $\mathbf{g}$ is constant so we assume a solution $\mathbf{J_p} = \mathbf{a}$ where $\mathbf{a}$ is independent of t. Hence

```
> a := LinearSolve(A, -g);
```

\# Resp. $\begin{bmatrix} 8 \\ 0 \end{bmatrix}$

```
> convert(Jh, Matrix) + a;
```

$$\begin{bmatrix} 3c1e^{-2t} + \frac{11}{9}c2e^{-\frac{6}{11}t} + 8 \\ c1e^{-2t} + c2e^{-\frac{6}{11}t} \end{bmatrix}$$

Pr.4.14 Type the ODE as a system

```
> sys := D(y1)(t) = y2(t), D(y2)(t) = (1 - y1(t)^2)*y2(t)/4 - y1(t);
```

$$sys := D(y1)(t) = y2(t),\ D(y2)(t) = \frac{1}{4}\left(1 - y1(t)^2\right) y2(t) - y1(t)$$

Choose some points (triples t, y_1, y_2) through which your trajectories should pass,

```
> inits := [0, 0, 1], [0, 0, 2], [0, 0, 3]:
```

Plot the direction field and the trajectories,

```
> with(DEtools):
```

```
> DEplot([sys[1], sys[2]], [y1(t), y2(t)], t = -5..5, y1 = -4..4,
  y2 = -4..4, [inits], linecolor=[black, blue, red], stepsize = 0.1);
```



Problem 4.14. Van der Pol equation with parameter $\mu = 1/2$

The second curve is a ***limit cycle***, which the trajectories approach from outside and inside; one of each of them is shown.

Pr.5.2 Type the given function and its series (responses not shown).

```
> f := sin(Pi*x):
```

```
> s := series(f, x, 20);
```

```
> p := convert(s, polynom);
```

```
> plot(p, f, x = -3..3, y = -2..1.2);
```



Problem 5.2. $\sin \pi x$ and approximation by a partial sum of the Maclaurin series

Pr.5.4 Type $P_6(x)$ and its derivative and evaluate both at $x = 0$. Of course, for these simple operations one would not need a computer or calculator, so the problem serves to illustrate the commands needed in more involved cases.

```
> with(orthopoly):
```

```
> p := P(6, x);
```

Resp. $p := -\frac{5}{16} + \frac{231}{16}x^6 - \frac{315}{16}x^4 + \frac{105}{16}x^2$

```
> p0 := subs(x = 0, p);
```

Resp. $p0 := -\frac{5}{16}$

```
> pp := diff(p, x);
```

Resp. $pp := \frac{693}{8}x^5 - \frac{315}{4}x^3 + \frac{105}{8}x$

```
> p1 := subs(x = 0, pp);
```

Resp. $p1 := 0$

More quickly,

```
> P(6, 0);
```

Resp. $-\frac{5}{16}$

```
> subs(x = 0, diff(P(6, x), x));
```

Resp. 0

Hence the initial value problem is as follows, and `dsolve` gives the expected response.

```
> ode := (1 - x^2)*diff(y(x), x, x) - 2*x*diff(y(x), x) + 42*y(x) = 0;
```

$$ode := (-x^2 + 1)\left(\frac{d^2}{dx^2}y(x)\right) - 2x\left(\frac{d}{dx}y(x)\right) + 42\,y(x) = 0$$

```
> dsolve(ode, y(0) = p0, D(y)(0) = p1);
```

$$y(x) = -\frac{5}{16} + \frac{231}{16}x^6 - \frac{315}{16}x^4 + \frac{105}{16}x^2$$

It is interesting that both independent solutions come out in closed form, as `dsolve` shows.

```
> dsolve(ode);
```

$$y(x) = \_C1\left(-\frac{5}{16} + \frac{231}{16}x^6 - \frac{315}{16}x^4 + \frac{105}{16}x^2\right) + \_C2\left(\frac{1}{160}(1155\,x^6 - 1575\,x^4 + 525\,x^2 - 25)\ln\left(\frac{-x-1}{x-1}\right) - \frac{231}{16}x^5 + \frac{119}{8}x^3 - \frac{231}{80}x\right)$$

Pr.5.8

```
> restart:
> ode := (x - 1)^2*diff(y(x), x, x) + (x - 1)*diff(y(x), x)
   - 9*y(x) = 0;
> dsolve(ode);                          # Resp.
```

$$y(x) = \_C1\ (x-1)^3 + \frac{\_C2}{(x-1)^3}$$

This suggests that the solution should be a power series in $x-1$ — the singular point. To find the indicial equation, consider the first few terms in the series

```
> ser3 := add(a[m]*(x -1)^(m+r), m = 0..3);
```

and substitute them into the ODE (long output omitted)

```
> ser3ode := (x - 1)^2*diff(ser3, x, x) +(x - 1)*diff(ser3, x) -9*ser3;
```

If any methods are used to simplify this expression, the $x-1$ terms get expanded. To avoid this, we do the following

```
> ser3odeX := subs({(x-1) = X}, ser3ode);
```

$$ser3odeX := X^2\left(\frac{a_0X^rr^2}{X^2} - \frac{a_0X^rr}{X^2} + \frac{a_1X^{1+r}(1+r)^2}{X^2} - \frac{a_1X^{1+r}(1+r)}{X^2} + \frac{a_2X^{2+r}(2+r)^2}{X^2} - \frac{a_2X^{2+r}(2+r)}{X^2} + \frac{a_3X^{3+r}(3+r)^2}{X^2} - \frac{a_3X^{3+r}(3+r)}{X^2}\right) + X\left(\frac{a_0X^rr}{X} + \frac{a_1X^{1+r}(1+r)}{X} + \frac{a_2X^{2+r}(2+r)}{X} + \frac{a_3X^{3+r}(3+r)}{X}\right) - 9\,a_0X^r - 9\,a_1X^{1+r} - 9\,a_2X^{2+r} - 9\,a_3X^{3+r}$$

```
> tempA := subs({X^(r) = A, X^(r+1) = A^2, X^(r+2) = A^3},
   simplify(ser3odeX));
```

$$tempA := X^{3+r}r^2a_3 + a_0Ar^2 + A^3r^2a_2 + A^2r^2a_1 + 6\,X^{3+r}ra_3 + 4\,A^3ra_2 + 2\,A^2ra_1 - 5\,a_2A^3 - 8\,a_1A^2 - 9\,a_0A$$

```
> Low := coeff(tempA, A);               # Resp.
```

$$Low := r^2a_0 - 9\,a_0$$

```
> rt := solve(Low = 0, r);              # Resp.
```

$$rt := 3, -3$$

If both solutions can be found they will come from the minimum value of r

```
> rm := min(rt);                        # Resp.
```

$$rm := -3$$

so find the corresponding $m+3$ term of the series

```
> ser := sum(a[s]*(x-1)^(s+rm), s = m + 1..m + 3);
```

$$ser := a_{m+1}(x-1)^{-2+m} + a_{2+m}(x-1)^{-1+m} + a_{m+3}(x-1)^m$$

```
> Rec := (x - 1)^2*diff(op(3, ser), x, x) + (x - 1)*diff(op(3, ser), x)
   - 9*op(3, ser);
```

$$Rec := (x-1)^2\left(\frac{a_{m+3}(x-1)^m m^2}{(x-1)^2} - \frac{a_{m+3}(x-1)^m m}{(x-1)^2}\right) + a_{m+3}(x-1)^m m - 9\,a_{m+3}(x-1)^m$$

```
> Rec1 := simplify(expand(Rec, x-1));
```

$$Rec1 := a_{m+3}(x-1)^m(m^2-9)$$

This must equal 0 for all m. For $m=-3$ and $m=3$, a_{-3} and a_3 can be non zero (giving the two arbitrary constants) while a_m must be 0 for all other m. The solution is thus

$$y = \frac{a_{-3}}{(x-1)^3} + a_3\,(x-1)^3.$$

Pr.5.16

```
> ode := x^2*diff(y(x), x, x) + x*diff(y(x), x) + (9*x^6 - 1/9)*y(x)
    = 0;
```

$$ode := x^2 \left(\frac{d^2}{dx^2} y(x)\right) + \left(\frac{d}{dx} y(x)\right) x + \left(9\,x^6 - \frac{1}{9}\right) y(x) = 0$$

```
> dsolve(ode);      # Resp.
```

$$y(x) = \_C1 \operatorname{BesselJ}\left(\frac{1}{9},\, x^3\right) + \_C2 \operatorname{BesselY}\left(\frac{1}{9},\, x^3\right)$$

This result shows that the transformation $x^3 = z$ would lead to Bessel's equation with parameter $1/9$.