Pr.6.2

```
> assume(s, nonnegative):
> int(exp(-s·t)·cos(omega·t)^2, t = 0..infinity);
```

$$\frac{2\,\omega^2 + s\sim^2}{s\sim\left(4\,\omega^2 + s\sim^2\right)}$$

The assumption that s is nonnegative is indicated by the tilde and is necessary and sufficient for the existence of the integral.

Pr.6.4 The exponential functions, which you first obtain, can be converted to hyperbolic functions as shown. Such conversions often need patience and use of trial and error.

```
> with(inttrans):
> invlaplace((s - 1)/(s^2 - 9), s, t);
```

Resp. $\frac{1}{3}e^{3t} + \frac{2}{3}e^{-3t}$

```
> convert(%, trig);
```

Resp. $\cosh(3t) - \frac{1}{3}\sinh(3t)$

Pr.6.6 Type the given ODE and obtain from it the subsidiary equation.

```
> restart:
> ode := diff(y(t), t) + 5*y(t) = (7/2)*exp(-5*t);
```

$$ode := \frac{d}{dx}y(t) + 5\,y(t) = \frac{7}{2}e^{-5t}$$

```
> with(inttrans):
> subsid := laplace(ode, t, s);
```

$$subsid := s\,laplace(y(t), t, s) - y(0) + 5\,laplace(y(t), t, s) = \frac{7}{2(s+5)}$$

Solve the subsidiary equation and insert the initial condition into it.

```
> Y := solve(subsid, laplace(y(t), t, s));
```

$$Y := \frac{1}{2}\,\frac{2\,s\,y(0) + 10\,y(0) + 7}{(s+5)^2}$$

```
> subs(y(0) = 1, Y);
```

Resp. $\frac{1}{2}\,\frac{2s+17}{(s+5)^2}$

Apply the inverse transform to the solution `Y` of the subsidiary equation.

```
> invlaplace(%, s, t);
```

Resp. $\frac{1}{2}(7t+2)e^{-5t}$

This calculation automatically proceeds in terms of decimal fractions if, in `ode`, you type `3.5` instead of `(7/2)`.

Pr.6.8

```
> with(inttrans):
> f := sin(t)*Heaviside(t - Pi);
```

$$f := \sin(t)\,\mathrm{Heaviside}(-\pi + t)$$

```
> plot([f, Heaviside(t - Pi)], t = 0..5*Pi, xtickmarks = [Pi, 3*Pi, 5*Pi],
  ytickmarks = [-1, 0, 1], linestyle = [solid, dash]);
```



Problem 6.8. Given function $f(t)$ and Heaviside $(t-\pi)$ (dashed)

```
> laplace(f, t, s);                                    # Resp.
```

$$-\frac{e^{-s\pi}}{s^2+1}$$

Pr.6.12 The right-hand side changes. Type the differential equation as

```
> eq := R*i(t) + 1/C*int(i(tau), tau = 0..t) = K*Dirac(t - 1);
```

$$eq := Ri(t) + \frac{\int_0^t i(\tau)\,d\tau}{C} = K\,\mathrm{Dirac}(t-1)$$

Obtain the subsidiary equation by typing

```
> with(inttrans):
> subsid := laplace(eq, t, s);
```

$$subsid := R\,laplace(i(t), t, s) + \frac{laplace(i(t), t, s)}{Cs} = Ke^{-s}$$

Now solve the subsidiary equation algebraically. Call the solution J because $I = \sqrt{-1}$ is protected.

```
> J := solve(subsid, laplace(i(t), t, s));             # Resp.
```

$$J := \frac{Ke^{-s}Cs}{RCs+1}$$

Now obtain the inverse of J by typing

```
> j := invlaplace(J, s, t);
```

$$j := -\frac{\mathrm{Heaviside}(t-1)\,Ke^{-\frac{t-1}{RC}}}{C\,R^2}$$