Chapter 10

Vector Integral Calculus. Integral Theorems

Content. Line integrals (Exs. 10.1, 10.2, Prs. 10.1–10.4)
Double integrals, Green's theorem (Exs. 10.3, 10.4, Prs. 10.5–10.9)
Surface integrals, Gauss and Stokes theorems
(Exs. 10.5–10.7, Prs. 10.10–10.13, 10.16–10.20)
Triple integrals (Prs. 10.14, 10.15)

`int(fC, t = a..b)` evaluates line integrals (the `fC` is used here to refer to f taken along the path C) converted to integrals over the parameter interval `t = a..b` of that path of integration C (see Ex. 10.1). `int(int(...))` evaluates double integrals (Ex. 10.3). `diff(F, x)`, `diff(F, y)` give partial derivatives (Ex. 10.4).

Examples for Chapter 10

EXAMPLE 10.1 LINE INTEGRALS

A **line integral** of a function f over a curve $C : \mathbf{r}(t)$, $a \leq t \leq b$, in space or in the plane (called the **path of integration**) can be defined by

$$\int_C f(\mathbf{r})\, d\mathbf{r} = \int_a^b f(\mathbf{r}(t))\, dt.$$

For example, let C be the helix $\mathbf{r}(t) = [\sin\, t,\ \cos, t,\ 5t]$, $a = 0$, $b = 2\pi$, and

$$f(\mathbf{r}) = f(x, y, z) = (x^2 + y^2 + z^2)^2.$$

Then type

```
> r := <sin(t) | cos(t) | 5*t>;
```
`# Resp.` $r := \begin{bmatrix} \sin(t) & \cos(t) & 5\,t \end{bmatrix}$

```
> f := (x^2 + y^2 + z^2)^2;
```
`# Resp.` $f := \left(x^2 + y^2 + z^2\right)^2$

Substituting $x = \cos\, t$, $y = \sin\, t$, $z = 3t$ gives $f(\mathbf{r}(t))$, that is, f on C. Denote this by `fC`. Accordingly, type (`r[1]` being the first component of `r`, etc.)

```
> subs(x = r[1], y = r[2], z = r[3], f);
```
$$\left(\sin(t)^2 + \cos(t)^2 + 25\,t^2\right)^2$$

```
> fC := simplify(%);
```
`# Resp.` $fC := 1 + 50\,t^2 + 625\,t^4$

```
> answer := int(fC, t = 0..2*Pi);
```
`# Resp.` $answer := 2\,\pi + \frac{400}{3}\,\pi^3 + 4000\,\pi^5$

```
> evalf[7](answer);
```
`# Resp.` $1.228220\,10^6$

Work integrals. This is a practical (nonstandard) name for another very useful kind of line integral in which a vector function $\mathbf{F}(\mathbf{r})$ is given and one finds and integrates the tangential component of $\mathbf{F}(\mathbf{r})$ in the tangent direction of the path of integration

C. Hence this is the work done by a force $\mathbf{F}$ in a displacement of a body along C. In terms of a formula,

$$\int_C \mathbf{F}(\mathbf{r}) \bullet d\mathbf{r} = \int_a^b \mathbf{F}(\mathbf{r}(t)) \bullet \frac{d\mathbf{r}}{dt} dt.$$

For instance, let $\mathbf{F(r)} = [z,\ x,\ y]$, and let C and $\mathbf{r}(t)$ be as before. Denote $\mathbf{F}(\mathbf{r}(t))$ by `FC` and $\mathbf{r}' = d\mathbf{r}/dt$ by `r1`. Then

```
> F := <z | x | y>;
```
\# Resp. $F := \begin{bmatrix} z & x & y \end{bmatrix}$

```
> r1 := VectorCalculus[diff](r, t);
```
$$r1 := (\cos(t))\, e_x - \sin(t)\, e_y + 5\, e_z$$

```
> FC := eval(subs(x = r[1], y = r[2], z = r[3], <F[1] | F[2] | F[3]>));
```
$$FC := \begin{bmatrix} 5\,t & \sin(t) & \cos(t) \end{bmatrix}$$

In this command, `x = r[1]` is the first component $\sin t$ of the position vector $\mathbf{r}$, etc. Similarly, `F[1]` is the first component of $\mathbf{F}$, namely, z, etc. To compute the inner product (the dot product), you need the `LinearAlgebra` package.

```
> with(LinearAlgebra):
```

```
> integrand := DotProduct(FC, r1, conjugate=false);
```
$$integrand := 5\,t\,\cos(t) - \sin(t)^2 + 5\,\cos(t)$$

```
> answer := int(integrand, t = 0..2*Pi);
```
\# Resp. $answer := -\pi$

```
> with(plots):
```

```
> spacecurve(r, t = 0..2*Pi, axes = NORMAL);
```



Example 10.1. Path of integration C (circular helix)

Similar Material in AEM: Sec. 10.1

EXAMPLE 10.2 INDEPENDENCE OF PATH

A line integral from a given point A to a given point B will generally depend on the path along which you integrate from A to B. A line integral is called **independent of path in a domain** D in space if, for every pair of endpoints A, B in D, the integral has the same value for all paths in D that begin at A and end at B.

For instance, show that the work integral of $\mathbf{F} = [4x,\ 6y,\ 6z]$ is independent of path in any domain D in space and find its value if you integrate from (0,0,0) to (2,2,2).

Solution. Necessary and sufficient for path independence in D is that $\mathbf{F}$ (with continuous components) have a potential f in D, that is, that $\mathbf{F}$ be the gradient of a function f in D. Thus type

```
> with(LinearAlgebra):
```

```
> VectorCalculus[SetCoordinates]('cartesian'[x,y,z]):
```

```
> F := <4*x | 6*y |6*z>;                    # Resp.
```
$F := \begin{bmatrix} 4\,x & 6\,y & 6\,z \end{bmatrix}$

```
> f := int(F[1], x) + int(F[2], y) + int(F[3], z);
```
$$f := 2\,x^2 + 3\,y^2 + 3\,z^2$$

```
> <diff(f, x) | diff(f, y) | diff(f, z)>;        # Resp.
```
$\begin{bmatrix} 4\,x & 6\,y & 6\,z \end{bmatrix}$

```
> gradf := VectorCalculus[Gradient](f);
```
$$gradf := 4x\bar{e}_x + 6y\bar{e}_y + 6z\bar{e}_z$$

Because grad$f = F$ we have path independence as claimed and f gives the potential.

Furthermore, if the components of $\mathbf{F}$ are continuous and have continuous first partial derivatives and if $\mathbf{F}$ is path independent in a domain D, then its curl is the zero vector. In our case,

```
> VectorCalculus[Curl](VectorCalculus[VectorField](F));
```
$$0\bar{e}_x$$

This condition is also sufficient for path independence in D, provided D is simply connected.

Path independence being guaranteed, you may now choose the most convenient path, namely, the straight-line segment C from the origin $A : (0, 0, 0)$ to $B : (3, 3, 3)$, and perform the integration as in the previous example. Let `r` be the position vector of C and `r1` its derivative. Let `FC` denote $\mathbf{F}$ on C, that is, $\mathbf{F}(\mathbf{r}(\mathrm{t}))$.

```
> r := <t | t | t>; r1 := VectorCalculus[diff](r, t);
```
$$r := \begin{bmatrix} t & t & t \end{bmatrix}$$
$$r1 := e_x + e_y + e_z$$

```
> FC := eval(subs(x = r[1], y = r[2], z = r[3], <F[1] | F[2] | F[3]>));
```
$$FC := \begin{bmatrix} 4\,t & 6\,t & 6\,t \end{bmatrix}$$

```
> answer := int(DotProduct(FC, r1), t = 0..3);        # Resp.
```
$answer := 72$

This result is much more quickly obtained by using the potential f and noting that the integral equals $f(B)-f(A)$ (the analog of a well-known formula from calculus) valid in the case of path independence.

```
> subs(x = 2, y = 2, z = 2, f) - subs(x = 0, y = 0, z = 0, f);
```
$$32$$

Of course, this simple example merely serves to explain the basic facts and techniques, and a computer or even a calculator would hardly be needed here.

Similar Material in AEM: Sec. 10.2

EXAMPLE 10.3 DOUBLE INTEGRALS. MOMENTS OF INERTIA

Find the moments of inertia I_x and I_y of a mass of density $\sigma = 1$ in the triangle with vertices $(0, 0)$, $(b, 0)$, (b, h) about the coordinate axes, as well as the polar moment $I_0 = I_x + I_y$. (Sketch the triangle.)

```
> with(plots):
```

```
> a := plot(<<0, 5> | <0, 4>>, style = line):
  b := plot(<<5, 5> | <4, 0>>, style = line):
  display([a, b], scaling = constrained, labels = [x, y]);
  a := 'a':  b := 'b':
```



Example 10.3. Region of integration when $b = 5$ and $h = 4$

Solution. Because $\sigma = 1$, the integrand of I_x is y^2. If you integrate stepwise first over y from 0 to xh/b and then over x from 0 to b, you obtain

```
> int(y^2, y = 0..h*x/b);
```
Resp. $\frac{1}{3}\frac{h^3x^3}{b^3}$

```
> Ix := int(%, x = 0..b);
```
Resp. $Ix := \frac{1}{12}h^3b$

If you combine the two steps into one, you obtain the same result,

```
> int(int(y^2, y = 0..h*x/b), x = 0..b);
```
Resp. $\frac{1}{12}h^3b$

If you integrate y^2 first over x, you must integrate from by/h to b and then over y from 0 to h, obtaining

```
> int(int(y^2, x = b*y/h..b), y = 0..h);
```
Resp. $\frac{1}{12}h^3b$

Similarly, the integrand of I_y is x^2 and integration gives

```
> Iy := int(int(x^2, y = 0..h*x/b), x = 0..b);
```
Resp. $Iy := \frac{1}{4}hb^3$

The **polar moment of inertia** I_0 is the sum $I_x + I_y$,

```
> I0 := Ix + Iy;
```
Resp. $I0 = \frac{1}{12}h^3b + \frac{1}{4}hb^3$

```
> factor(I0);
```
Resp. $I0 = \frac{1}{12}bh\left(h^2 + 3\,b^2\right)$

Similar Material in AEM: Sec. 10.3

EXAMPLE 10.4 GREEN'S THEOREM IN THE PLANE

Green's theorem in the plane transforms a double integral over a region R in the xy-plane into a line integral along the boundary C of R and conversely. The formula is

$$\iint\limits_R \left[\left(\frac{\partial}{\partial x} F_2\right) - \left(\frac{\partial}{\partial y} F_1\right)\right] dx\, dy = \int_C (F_1\, dx + F_2\, dy).$$

It is valid under suitable regularity assumptions on R, C, F_1, and F_2, usually satisfied in applications (see AEM, Sec. 10.4).

For instance, proceeding first in terms of components, (and later repeating the calculation in terms of vectors), let $F_1 = y^3 - 11y$, $F_2 = 3xy + 11x$. Type

```
> F1 := y^3 - 11*y; F2 := 3*x*y^2 + 11*x;
```

$$F1 := y^3 - 11\,y$$
$$F2 := 3\,xy^2 + 11\,x$$

The integrand of the double integral is

```
> diff(F2, x) - diff(F1, y);                              # Resp. 22
```

Hence the integrand is constant, so that the integral equals 22 times the area of the region of integration R. For instance, if R is a circular disk of radius 1, the integral equals 22π.

We verify the formula of Green's theorem for this case, assuming that the disk has the center at the origin of the xy-plane. Then its boundary curve C is the circle $x^2 + y^2 = 1$. In polar coordinates, the position vector $\mathbf{r}$ of C (oriented counterclockwise!) and its derivative $\mathbf{r}'$ have the components

```
> x := cos(t); y := sin(t); x1 := diff(x, t); y1 := diff(y, t);
```

$$x := \cos(t)$$
$$y := \sin(t)$$
$$x1 := -\sin(t)$$
$$y1 := \cos(t)$$

Hence the line integral on the right-hand side of the formula is

```
> int(F1*x1 + F2*y1, t = 0..2*Pi);                        # Resp. 22 π
```

This verifies Green's theorem for the present example.

The same example in vectorial notation. Let

```
> x := 'x':  y := 'y':
```

```
> r := <cos(t) | sin(t) | 0>; r1 := VectorCalculus[diff](r, t);
```

$$r := \left[\begin{array}{ccc} \cos(t) & \sin(t) & 0 \end{array}\right]$$
$$r1 := -\sin(t)\, e_x + \cos(t)\, e_y$$

```
> F := <F1 | F2 | 0>;
```

Resp. $F := \left[\begin{array}{ccc} y^3 - 11\,y & 3\,xy^2 + 11\,x & 0 \end{array}\right]$

```
> with(LinearAlgebra):
```

```
> VectorCalculus[SetCoordinates]('cartesian'[x,y,z]):
```

```
> CU := VectorCalculus[Curl](VectorCalculus[VectorField](F));
```

$$CU := 22\bar{e}_z$$

Hence the integrand $(\operatorname{curl}\mathbf{F}) \bullet \mathbf{k}$ of the double integral (with $\mathbf{k}$ the unit vector in the z-direction) equals 22, as before,

```
> DotProduct(CU, <0 | 0 | 1>);                                    # Resp. 22
```

For the line integral you get (with `r[1]` and `r[2]` the x and y components of $\mathbf{r}$), as before,

```
> FC := subs(x = r[1], y = r[2], F);
```

$$FC := \left[\ \sin(t)^3 - 11\sin(t) \quad 3\cos(t)\sin(t)^2 + 11\cos(t) \quad 0\ \right]$$

```
> int(DotProduct(FC, r1), t = 0..2*Pi);                           # Resp. 22π
```

Note that, in using the curl, Maple requires that you carry along a third component 0 in $\mathbf{r}$. (Can you find the reason?)

Similar Material in AEM: Sec. 10.4

EXAMPLE 10.5 SURFACE INTEGRALS. FLUX

These are integrals taken over a surface S. We assume S to be given parametrically in the form $\mathbf{r}(u, v)$, where (u, v) varies over a region R in the uv-plane. The integrand is a scalar function given in the form $\mathbf{F} \bullet \mathbf{n}$, where the vector function $\mathbf{F}$ is given and $\mathbf{n}$ is a unit normal vector of S ($-\mathbf{n}$ being the other one). This is very practical in flow problems, where $\mathbf{F} = \rho\mathbf{v}$ (ρ the density, $\mathbf{v}$ the velocity) and $\mathbf{F} \bullet \mathbf{n}$ is the **flux** across S, that is, the mass of fluid crossing S per unit time. The integral is evaluated by reducing it to a double integral over R, namely,

$$\iint_S \mathbf{F} \bullet \mathbf{n}\, dA = \iint_R \mathbf{F}(\mathbf{r}(u, v)) \bullet \mathbf{N}(u, v)\, du\, dv,$$

where dA is the element of area, $\mathbf{N} = \mathbf{r}_u \times \mathbf{r}_v$ is a normal vector of S, and $\mathbf{r}_u$ and $\mathbf{r}_v$ are the partial derivatives of $\mathbf{r}(u, v)$.

For instance, let $\mathbf{r} = [u,\ u^2,\ v]$ with u varying from 0 to 2 and v from 0 to 3. You see that $y = x^2$, so that this surface intersects the xy-plane along the parabola $y = x^2$. See the figure. To understand the plot command, type `?plot3d` and `?plot3d[options]`. `orientation = [50, 70]` gives the spherical coordinates θ and ϕ of the point from which the plot is viewed. Try other angles.

Let $\mathbf{v} = \mathbf{F} = [5z^2,\ 7,\ 7xz]$ and $\rho = 1\,\text{gram/cm}^3 = 1000\,\text{kg/meter}^3 = 1\,\text{tonne/meter}^3$ (density of water), speed being measured in meters/sec. Type

```
> r := <u | u^2 | v>;                                  # Resp. r := [ u  u^2  v ]
```

```
> r_u := VectorCalculus[diff](r, u); r_v := VectorCalculus[diff](r, v);
```

$$r_u := e_x + 2\,u e_y$$
$$r_v := e_z$$

```
> with(LinearAlgebra):
```

```
> N := CrossProduct(r_u, r_v);                         # Resp. N := [ 2u  -1  0 ]
```

```
> F := <5*z^2 | 7 | 7*x*z>;                            # Resp. F := [ 5z^2  7  7xz ]
```

On S, the vector function $\mathbf{F}$ takes the form $\mathbf{F}(\mathbf{r}(u, v))$. Call it `FS` and type it and the integrand on the right-hand side of the general formula

```
> FS := subs(x = r[1], y = r[2], z = r[3], F);
```

$$FS := \left[\begin{array}{ccc} 5\,v^2 & 7 & 7\,uv \end{array}\right]$$

```
> integrand := DotProduct(FS, N, conjugate = false);
```

$$integrand := -7 + 10\,uv^2$$

Now integrate over u and v,

```
> flux := int(int(integrand, u = 0..2), v = 0..3);   # Resp. flux := 138
```

Because speed is measured in meters/sec, the flux is measured in tonnes/sec; this gives the answer 138 tonnes/sec or 138000 liters/sec.

```
> plot3d(<r[1] | r[2] | r[3]>, u = 0..2, v = 0..3, axes = NORMAL,
   labels = [x, y, z], orientation = [70, 40]);
```



Example 10.5. Surface over which the integral is extended

Similar Material in AEM: Sec. 10.6

EXAMPLE 10.6 DIVERGENCE THEOREM OF GAUSS

The **divergence theorem** transforms a triple integral over a region T in space into a surface integral over the boundary surface S of T and conversely. The formula is

$$\iiint_T \operatorname{div} \mathbf{F}\, dV = \iint_S \mathbf{F} \bullet \mathbf{n}\, dA$$

where $\mathbf{F}$ is a given vector function, dV is the volume element, $\mathbf{n}$ is the outer unit normal vector of S, and dA is the element of area of S. The formula is valid under suitable regularity assumptions on T, S, and $\mathbf{F}$ usually satisfied in applications (see AEM, Sec. 10.7). Note that if in a flow problem, $\mathbf{F} = \rho\mathbf{v}$ (ρ the density, $\mathbf{v}$ the velocity), then the integral over S gives the flux through S. This is similar to the previous example.

For instance, let T be the solid circular cylinder of radius a with the z-axis as the axis, extending from $z = 0$ to $z = b$ (> 0), and let $\mathbf{F} = [x^3, x^2\,y, x^2\,z]$. Using the divergence theorem, evaluate the surface integral of $\mathbf{F} \bullet \mathbf{n}$ over the surface S of T (consisting of the upper and lower disks and the lateral cylindrical part).

Solution. Type

```
> F := <x^3 | x^2*y | x^2*z>;                # Resp. F := [ x^3  x^2y  x^2z ]
```

```
> with(LinearAlgebra):
```

```
> VectorCalculus[SetCoordinates]('cartesian'[x, y, z]):
```

```
> divF := VectorCalculus[Divergence](VectorCalculus[VectorField](F));
```

$$divF := 5\,x^2$$

The form of T suggests the use of cylindrical coordinates $x = r\cos\theta$, $y = r\sin\theta$, $z = z$. Accordingly, make the substitution (note that div **F** depends only on x)

```
> f := subs(x = r*cos(theta), divF);              # Resp.
```
$f := 5\,r^2\cos(\theta)^2$

This is the integrand of the triple integral in the divergence theorem, which must be multiplied by the volume element $r\,dr\,d\theta\,dz$ and integrated over r from 0 to a, over θ from 0 to 2π, and over z from 0 to b. Type

```
> answer := int(int(int(f*r, r = 0..a), theta = 0..2*Pi), z = 0..b);
```

$$answer := \frac{5}{4}\pi\,a^4 b$$

The direct evaluation of the surface integral should give the same result. It is more elaborate and thus illustrates the usefulness of the divergence theorem. Begin with the disk on the top, whose outer unit normal vector is pointing vertically upward,

```
> n := <0 | 0 | 1>;                               # Resp.
```
$n := \begin{bmatrix} 0 & 0 & 1 \end{bmatrix}$

You need the dot product $\mathbf{F} \bullet \mathbf{n}$,

```
> ip := DotProduct(F, n);                         # Resp.
```
$ip := x^2 z$

Introduce polar coordinates by setting $x = r\cos\theta$ and then integrate over r from 0 to a and over θ from 0 to 2π. Use $dA = r\,dr\,d\theta$. Call the integral J1.

```
> g := subs(x = r*cos(theta), z = b, ip);         # Resp.
```
$g := r^2\cos(\theta)^2\,b$

```
> J1 := int(int(g*r, r = 0..a), theta = 0..2*Pi);
```

$$J1 := \frac{1}{4}\pi\,a^4 b$$

For the bottom, $z = 0$, the inner product is $-x^2 z$, hence it is zero. (The outer normal vector points downward; this gives the minus sign.)

Now turn to the vertical cylinder surface. Represent it by

```
> r := <a*cos(theta) | a*sin(theta) | z>;
```

$$r := \begin{bmatrix} a\cos(\theta) & a\sin(\theta) & z \end{bmatrix}$$

Obviously, its normal is horizontal and at each point has the direction of the position vector. Hence the outer unit normal vector n2 of the cylinder is

```
> n2 := <cos(theta) | sin(theta) | 0>;
```

$$n2 := \begin{bmatrix} \cos(\theta) & \sin(\theta) & 0 \end{bmatrix}$$

Let FC denote **F** on the cylinder. Recall that r[1], r[2], r[3] are the components of **r**. Hence type

```
> FC := subs(x = r[1], y = r[2], z = r[3], F);
```

$$FC := \begin{bmatrix} a^3\cos(\theta)^3 & a^3\cos(\theta)^2\sin(\theta) & a^2\cos(\theta)^2 z \end{bmatrix}$$

```
> ip2 := DotProduct(FC, n2, conjugate = false);
```

$$ip2 := a^3\cos(\theta)^4 + a^3\cos(\theta)^2\sin(\theta)^2$$

```
> simplify(ip2);                                  # Resp.
```
$a^3\cos(\theta)^2$

This is the integrand. Multiply it by the element of area $a d\theta dz$ and integrate over θ from 0 to 2π and over z from 0 to b. Call the integral `J2`. Add it to `J1` to get the answer, which agrees with the previous one.

```
> J2 := int(int(ip2*a, theta = 0..2*Pi), z = 0..b);
```

$$J2 := \pi a^4 b$$

```
> answer2 := J1 + J2;
```

Resp. $answer2 := \frac{5}{4}\pi a^4 b$

Similar Material in AEM: Sec. 10.7

EXAMPLE 10.7 STOKES'S THEOREM

Stokes's theorem transforms a surface integral over a surface S into a line integral over the boundary curve C of S and conversely. The formula is

$$\iint_S (\operatorname{curl} \mathbf{F}) \bullet \mathbf{n}\, dA = \oint_C \mathbf{F} \bullet \mathbf{r}'(s)\, ds$$

where $\mathbf{n}$ is a unit surface normal vector of S and $\mathbf{r}'(s)$ is a unit tangent vector of C and the direction of integration around C appears counterclockwise if one looks from the terminal point of $\mathbf{n}$ onto the surface. s is the arc length of C. The formula holds under suitable regularity conditions on S, C, and $\mathbf{F}$ usually satisfied in applications (see AEM, Sec. 10.9).

For instance, using Stokes's theorem, evaluate the line integral in the formula when $\mathbf{F} = [5\,z^2,\ x,\ -7y^3]$ and C is the circle $x^2 + y^2 = a^2$, $z = b\ (> 0)$.

Solution. Type $\mathbf{F}$ as given. Then obtain the curl.

```
> F := <5*z^2 | x | -7*y^3>;
```

Resp. $F := \begin{bmatrix} 5\,z^2 & x & -7\,y^3 \end{bmatrix}$

```
> with(LinearAlgebra):
> VectorCalculus[SetCoordinates]('cartesian'[x,y,z]):
> curlF := VectorCalculus[Curl] (VectorCalculus[VectorField](F));
```

$$curlF := -21\,y^2\bar{e}_x + 10\,z\bar{e}_y + \bar{e}_z$$

As surface S choose the disk bounded by C. Obtain the curl on S; denote it by `curlFS`.

```
> r := <u*cos(v) | u*sin(v) | b>;
```

Resp. $r := \begin{bmatrix} u\cos(v) & u\sin(v) & b \end{bmatrix}$

```
> curlFS := <0 | 0 | subs(y = r[2], curlF[3])>;
```

$$curlFS := \begin{bmatrix} 0 & 0 & 1 \end{bmatrix}$$

A unit normal vector of the disk is

```
> n := <0 | 0 | 1>;
```

Resp. $n := \begin{bmatrix} 0 & 0 & 1 \end{bmatrix}$

You can now type the integrand `ip` and then integrate over u from 0 to a and over the angle v from 0 to 2π.

```
> ip := DotProduct(curlFS, n);
```

Resp. $ip := 1$

```
> answer := int(int(ip*u, u = 0..a), v = 0..2*Pi);
```

$$answer := a^2\pi$$

Confirmation by direct evaluation of the line integral around the boundary circle $C : x^2 + y^2 = a^2$, $z = b$. A representation of C is the following, in which t goes from 0 to 2π.

```
> r := <a*cos(t) | a*sin(t) | b>;     # Resp.
```
$r := \begin{bmatrix} a\cos(t) & a\sin(t) & b \end{bmatrix}$

You can use t as a variable of integration because in the integral, by calculus, $\mathbf{r}'(s)\,ds = (d\mathbf{r}/ds)\,ds = (d\mathbf{r}/dt)\,dt$. Denote $\mathbf{F}$ on C by `FC` and obtain it by typing

```
> FC := subs(x = r[1], y = r[2], z = r[3], F);
```
$$FC := \begin{bmatrix} 5\,b^2 & a\cos(t) & -7\,a^3\sin(t)^3 \end{bmatrix}$$

Denote $d\mathbf{r}/dt$ by `r1` and type

```
> r1 := VectorCalculus[diff](r, t); # Resp.
```
$r1 := -a\sin(t)\,e_x + a\cos(t)\,e_y$

Now obtain the integrand `ip` and finally the integral, which will agree with the previous result.

```
> ip := DotProduct(FC, r1, conjugate = false);
```
$$ip := -5\,b^2 a\sin(t) + a^2\cos(t)^2$$

```
> int(ip, t = 0..2*Pi);                              # Resp.
```
$a^2\pi$

Similar Material in AEM: Sec. 10.9

Problem Set for Chapter 10

Pr.10.1 (Line integral. Work) Evaluate the work integral (see Example 10.1 in this Guide) of the force $\mathbf{F} = [2z,\ 7x,\ -3y]$ from $(1, 0, 0)$ to $(1, 0, 4\pi)$ along the helix C : $\mathbf{r} = [\cos t,\ \sin t,\ 2t]$. (*AEM* Sec. 10.1)

Pr.10.2 (Path dependence, same endpoints) Show that the integral in Pr.10.1 changes its value if you integrate from $(1, 0, 0)$ to $(1, 0, 4\pi)$ along the straight-line segment with these endpoints. (*AEM* Sec. 10.1)

Pr.10.3 (Independence of path. Potential) Using a suitable curl, show that the integral $\int_C (3x^2\,dx + 2yz\,dy + y^2\,dz)$ is independent of path in any domain in space. Find a potential and use it to obtain the value of the integral from A: $(0, 1, 2)$ to B: $(1, -1, 7)$. (*AEM* Sec. 10.2)

Pr.10.4 (Independence of path) Is the integral of $ze^x\,dx + 2y\,dy + e^z\,dz$ independent of path in space? If this is the case, find a potential by integration . Using the potential, integrate the given form from the origin to the point (a, b, c). (*AEM* Sec. 10.2)

Pr.10.5 (Double integral. Center of gravity) Find the center of gravity of a mass of density $\sigma = 1$ in the portion of the disk $x^2 + y^2 \le 1$ in the second quadrant of the xy-plane. (*AEM* Sec. 10.3)

Pr.10.6 (Double integral. Moment of inertia) Consider a mass of density $\sigma = 1$ in the portion of the disk $x^2 + y^2 \le a^2$ in the lower half-plane. Find the polar moment of inertia I_0 of this mass about the origin. (*AEM* Sec. 10.3)

Pr.10.7 (Green's theorem in the plane) Using the formula of Green's theorem (see Example 10.4 in this Guide), integrate $\mathbf{F}(\mathbf{r}(t)) \cdot \mathbf{r}'(t)$ counterclockwise around the boundary of the region $R : 1 + x^6 \le y \le 32$; here, $\mathbf{F} = [e^y/x,\ e^y \ln x + 2x]$. (*AEM* Sec. 10.4)

Pr.10.8 (Green's theorem in the plane) Use Green's theorem in the plane (see Example 10.4 in this Guide), to integrate $\mathbf{F}(\mathbf{r}(t)) \bullet \mathbf{r}'(t)$ with $\mathbf{F} = [2x \sinh 6y,\ 7x^2 \cosh 6y]$ counterclockwise around the boundary of the region $R : x^2 \le y \le x$. (*AEM* Sec. 10.4)

Pr.10.9 (Area) Choosing $F_1 = 0$, $F_2 = x$ in the formula of Green's theorem in Example 10.4 of this Guide gives $\iint_R dxdy = \oint_C x\,dy$. Similarly, $\iint_R dxdy = -\oint_C y\,dx$ by choosing $F_1 = -y$, $F_2 = 0$. The double integral is the area A of R. Together,

$$A = \frac{1}{2}\oint_C (x\,dy - y\,dx).$$

Obtain from this the area of an ellipse $x^2/a^2 + y^2/b^2 = 1$. (*AEM* Sec. 10.4)

Pr.10.10 (Experiment on surface normal) Find a representation of the **ellipsoid**

$$S : \mathbf{r}(u, v) = [a \cos v \cos u,\ b \cos v \sin u,\ c \sin v]$$

in terms of Cartesian coordinates. Find a normal vector of S. Plot S for some triples a, b, c, for instance, $a = 10$, $b = 4$, $c = 3$. Choose other triples and observe how the surface and its normal change. (*AEM* Sec. 10.5)

Pr.10.11 (Surface integral) Find the flux integral (see Example 10.5 in this Guide for the definition) of $\mathbf{F} = [x^3,\ 0,\ 6y^4]$ over the portion of the plane $3x + 7y + z = 4$ in the first octant in space. (*AEM* Sec. 10.6)

Pr.10.12 (Surface integral) Find the flux integral (as defined in Example 10.5 of this Guide) of $\mathbf{F} = [x^3,\ y^3,\ z^3]$ over the **helicoid** $\mathbf{r} = [u \cos v,\ u \sin v,\ 3v]$, where $0 \le u \le 1$, $0 \le v \le 2\pi$. Sketch the surface. Explain its name. (*AEM* Sec. 10.6)

Pr.10.13 (Integral over a sphere) Find the flux integral (as defined in Example 10.5 in this Guide) of $\mathbf{F} = [0,\ 0,\ y]$ over the portion of the sphere $x^2 + y^2 + z^2 = 1$ in the first octant. (*AEM* Sec. 10.6)

Pr.10.14 (Triple integral) Find the total mass in the box $|x| \le 4$, $|y| \le 2$, $|z| \le 5$ if the mass density is $\sigma = x^2 + y^2 + z^2$. (*AEM* Sec. 10.7)

Pr.10.15 (Triple integral. Moment of inertia) Find the moment of inertia I_x about the x-axis of a mass of density σ in the solid circular cylinder of radius a about the x-axis, extending in x-direction from 0 to h. (*AEM* Sec. 10.7)

Pr.10.16 (Surface integral. Divergence theorem) Using the divergence theorem (see Example 10.6 in this Guide), find the integral of the normal component of the vector function $\mathbf{F} = [\,3\,x,\ x^3\,y^5,\ y^3\,z^4\,]$ over the surface of the tetrahedron with vertices $(0, 0, 0)$, $(1, 0, 0)$, $(0, 1, 0)$, $(0, 0, 1)$. (*AEM* Sec. 10.7)

Pr.10.17 (Surface integral. Divergence theorem) Using the divergence theorem (see Example 10.6 in this Guide), integrate the normal component of the vector function $\mathbf{F} = [9x,\ y \cosh^2 x,\ -z \sinh^2 x]$ over the ellipsoid $4\,x^2 + y^2 + 9\,z^2 = 36$. (*AEM* Sec. 10.8)

Pr.10.18 (Laplacian, normal derivative) If $\mathbf{F} = \operatorname{grad} f$ in the formula of the divergence theorem in Example 10.6 in this Guide, show that $\operatorname{div} \mathbf{F} = \nabla^2 f$ and $\mathbf{F} \bullet \mathbf{n} = \partial f / \partial n$ (the normal derivative). Verify the resulting formula

$$\iiint_T \nabla^2 f \, dV = \iint_S \frac{\partial f}{\partial n} \, dA$$

for $f = (x^2 + y^2 + z^2)^2$ and S a sphere of radius a and center at the origin. *Hint*: Represent the interior of S by $\mathbf{R} = [r \cos u \sin v,\ r \sin u \sin v,\ r \cos v]$ (where r is variable!) and use the corresponding volume element $r^2 \sin v \, dr \, dv \, du$. (*AEM* Sec. 10.8)

Pr.10.19 (Surface integral. Divergence theorem) Using the divergence theorem, find the integral of the normal derivative of $\mathbf{F} = [x^5,\ y^5,\ z^5]$ over the sphere $S: x^2+y^2+z^2 = 9$. Represent S as in the previous problem. (*AEM* Sec. 10.7)

Pr.10.20 (Stokes's theorem) Using Stokes's theorem (see Example 10.7 in this Guide), integrate the tangential component of $\mathbf{F} = [e^z,\ e^z \sin y,\ e^z \cos y]$ around the boundary of the surface $S: z = y^2$, where $0 \le x \le 4$, $0 \le y \le 2$. (*AEM* Sec. 10.9)