Pr.25.2 (Confidence interval for the mean) Find a 99% confidence interval for the mean of a normal population with standard deviation 2.7, using the sample 25.5, 24.7, 24.6, 24.8, 26.4, 28.7. (*AEM* Sec. 25.3)

Pr.25.2 $-c$ is the 0.5%-point and c is the 99.5%-point of the standardized normal distribution.

```
> with(Statistics): Digits := 5: sd := 2.7:
> Sa := [25.5, 24.7, 24.6, 24.8, 26.4, 28.7]:
> xbar := Mean(Sa);                          # Resp. xbar := 25.783
> n := Count(Sa);                            # Resp. n := 6
> c := Quantile('Normal'(0, 1), 0.995);      # Resp. c := 2.5758
> k := evalf(c*sd/sqrt(n));                  # Resp. k := 2.8392
> conf1 := xbar - k;                         # Resp. conf1 := 22.944
> conf2 := xbar + k;                         # Resp. conf2 := 28.622
```

This is what you do when you use a table of the standardized normal distribution. On the computer you can proceed more directly, noting that $\bar{X}$ has standard deviation $\sigma/\sqrt{n} = 2.7/\sqrt{6}$ and that the sample mean is the midpoint of the confidence interval. Type

```
> Quantile('Normal'(xbar, sd/sqrt(n)), 0.005);      # Resp. 22.944
> Quantile('Normal'(xbar, sd/sqrt(n)), 0.995);      # Resp. 28.622
```

Pr.25.4 (Confidence interval for the mean) What confidence interval would you obtain in Example 25.3 in this Guide if σ were known and equal to $s = 3.2514$ (the value in that example), the other data being as before?

Pr.25.4 The new confidence interval is slightly shorter due to the additional information used.

```
> with(Statistics): Digits := 5:
> sample := [242, 251, 248, 245, 250, 247, 244]:
> n := Count(sample);                         # Resp. n := 7
> xbar := Mean(sample);                       # Resp. xbar := 246.71
> s := StandardDeviation(sample);             # Resp. s := 3.2514
> c := Quantile('Normal'(0, 1), 0.995);       # Resp. c := 2.5758
> k := evalf(c*s/sqrt(n));                    # Resp. k := 3.1654
```

Pr.25.6 (Test for the mean) Test the hypothesis $\mu_0 = 24$ against the alternative $\mu_1 = 27$, choosing $\alpha = 5\%$ and using a sample of size 10 with mean 25.8 from a normal population with variance 9. Is the power of the test sufficiently large?
(*AEM* Sec. 25.4)

Pr.25.6 The test is right-sided. Hence you need the 95%-point.

```
> with(Statistics):  Digits := 5:  sd := 3/sqrt(10):
```

```
> c := Quantile('Normal'(24, sd), 0.95);                # Resp. c := 25.560
```

$25.8 > c$. Reject the hypothesis. The power is large, 93.5%.

```
> power := 1 - CDF('Normal'(27, sd), c);
```

$$power := 0.935479347450266$$

Pr.25.10 (Comparison of means) Will an increase of temperature increase the yield (measured in grams/min) of some chemical process? Test this, using the following independent samples, assuming normality, and choosing $\alpha = 5\%$. (*AEM* Sec. 25.4)

Yield x at 55^o C	97	108	115	103	113	117	130	127	111	107
Yield y at 70^o C	115	123	138	118	105	130	132	127		

Pr.25.10 Hypothesis $\mu_y = \mu_x$. Alternative $\mu_y > \mu_x$. Right-sided test, rejection region extends from c to the right. Independent samples. Sample sizes $n_1 = 10$ (x-values), $n_2 = 8$ (y-values). Use the t-distribution with 16 degrees of freedom.

```
> with(Statistics):  Digits := 5:
> sa1 := [97, 108, 116, 103, 113, 117, 129, 127, 111, 107];
> sa2 := [115, 121, 138, 118, 104, 130, 132, 127];
> n1 := Count(sa1);                          # Resp. n1 := 10
> n2 := Count(sa2);                          # Resp. n2 := 8
```

Now obtain the means and variances of the x-values and of the y-values.

```
> xbar := Mean(sa1);                         # Resp. xbar := 112.80
> ybar := Mean(sa2);                         # Resp. ybar := 123.12
> xvar := Variance(sa1);                     # Resp. xvar := 99.734
> yvar := Variance(sa2);                     # Resp. yvar := 117.83
```

Now obtain an observed value of the t-distributed random variable T used in this test as well as the 95%-point of the t-distribution of T with $n_1 + n_2 - 2 = 16$ degrees of freedom.

```
> t0 := evalf(sqrt((n1*n2*(n1+ n2 - 2))/(n1+ n2))*(ybar - xbar)/
   sqrt(n1*xvar + n2*yvar));
                              t0 := 1.9759
> c := Quantile('StudentT'(n1 + n2 - 2), 0.95);      # Resp. c := 1.7459
```

Because $t_0 > c$ and the test is right-sided, reject the hypothesis and assert that there will be an increase in yield if the temperature is raised.