



**Objective:** Structural Design using Verilog modules and using the RTL viewer to verify the nested structure of the design.

**Design:**

- (a) Write a gate-level model describing the operation of a TwoBit comparator. The module will have 2 Two-bit inputs: **a**, and **b**. Three outputs indicate whether a equals b, a is less than b, or a is greater than b. Create the Karnaugh map and find the expression for the switching function for **the three outputs**.
- (b) Use the TwoBit comparator to design a FourBit comparator. Add any additional logic needed.
- (c) Verify the connection using the RTL Viewer, and verify the interconnections of the instantiated module.
- (d) The Verilog Code should be well documented.

```

module twoBit_comparator (a, b, a_less_b, a_greater_b, a_equal_b);
input [1:0] a,b;
output      a_less_b, a_greater_b, a_equal_b;

wire a0,    a1,    b0,    b1;
wire a0bar, a1bar, b0bar, b1bar;

assign a0 = a[0];    assign a0bar = ~a0;
assign a1 = a[1];    assign a1bar = ~a1;
assign b0 = b[0];    assign b0bar = ~b0;
assign b1 = b[1];    assign b1bar = ~b1;

assign a_less_b    = (b1 & a1bar) | (a0bar & a1bar & b0) | (b0 & b1 & a0bar);
assign a_greater_b = (a1 & b1bar) | (a0 & b1bar & b0bar) | (a0 & a1 & b0bar);
assign a_equal_b    = (a0bar & a1bar & b0bar & b1bar) |
(a0 & a1bar & b0 & b1bar) |
(a0bar & a1 & b0bar & b1) |
(a0 & a1 & b0 & b1);

endmodule

```

```

module twoBit_comparator_tb ();


reg [3:0] ab;
wire A1B, AgB, AeB;

twoBit_comparator uut (.a (ab[1:0]),
                        .b (ab[3:2]),
                        .a_less_b (A1B),
                        .a_greater_b (AgB),
                        .a_equal_b (AeB) );

initial
begin
ab = 4'b0000;
forever
#10 ab = ab + 4'b0001;
end

endmodule

```



```

module fourBit_comparator (A,B,A_less_B, A_greater_B, A_equal_B);
input [3:0] A,B;
output      A_less_B, A_greater_B, A_equal_B;

wire MSB_A_less_B, MSB_A_greater_B, MSB_A_equal_B;
wire LSB_A_less_B, LSB_A_greater_B, LSB_A_equal_B;

twoBit_comparator MSB (A[3:2], B[3:2], MSB_A_less_B, MSB_A_greater_B, MSB_A_equal_B);
twoBit_comparator LSB (A[1:0], B[1:0], LSB_A_less_B, LSB_A_greater_B, LSB_A_equal_B);

assign A_equal_B = MSB_A_equal_B & LSB_A_equal_B;
assign A_less_B = MSB_A_less_B | (MSB_A_equal_B & LSB_A_less_B);
assign A_greater_B = MSB_A_greater_B | (MSB_A_equal_B & LSB_A_greater_B);

endmodule

```

```

module fourBit_comparator_tb ();

reg [7:0] AB;
wire      A_less_B, A_greater_B, A_equal_B;


//fourBit_comparator (A,B,A_less_B, A_greater_B, A_equal_B);

fourBit_comparator UUT (.A (AB[3:0]),
                        .B (AB[7:4]),
                        .A_less_B (A_less_B),
                        .A_greater_B (A_greater_B),
                        .A_equal_B (A_equal_B) );



initial
begin
    AB = 8'b0000_0000;
    forever
    #10 AB = AB + 1;
end


endmodule

```



1. Expand the created signal AB into single bits in the simulation waveform window.
2. Select the bits [3:0] and **combine signals** them into one bus and call it **A**.
3. Select the bits [7:4] and **combine signals** them into one bus and call it **B**.
4. Select A and B buses and choose **group**.
5. Call the group inputs adjust its height as required.





You can change the order of the signals by dragging them up or down.

You can also right click on any of the buses and change the **radix** to **unsigned** to display unsigned decimal or choose from the different radix options such as: binary, hexadecimal or octal.