Chapter 9

Vector Differential Calculus
Grad, Div, Curl

Content. Addition of vectors, scalar multiplication (Ex. 9.1, Prs. 9.1–9.4)
Inner product, cross product, triple product (Ex. 9.2, Prs. 9.5–9.13)
Vector fields (Pr. 9.14)
Derivatives, curves (Ex. 9.3, Prs. 9.15–9.18)
Gradient, divergence, curl (Exs. 9.4, 9.5, Prs. 9.19–9.25)

LinearAlgebra package. Load by typing `with(LinearAlgebra):` You will need it.

For vectors in R^2 and R^3 various commands are those for vectors in R^n in Chap. 7, for instance, `DotProduct(u,v)`. New are `VectorAngle(u, v)`, `CrossProduct(u, v)`, and because of the changes in packages involving vectors, a new `VectorCalculus` package has been introduced with several commands migrating from the old `linalg` package.

VectorCalculus package. You can load it by typing `with(VectorCalculus)`. In the `linalg` package, the information about the coordinate system that you were using was included in the command. For example `grad(f(x,y,z), [x, y, z])` for 3-dimensional Cartesian coordinates. In the `VectorCalculus` package you specify the coordinate system by `SetCoordinates('cartesian'[x,y,z])` before using the following commands

`Gradient(f(x,y,z)`	(Ex. 9.4)
`Divergence(VectorField(v))`	(Ex. 9.5)
`Curl(VectorField(v))`	(Ex. 9.5)
`ScalarPotential(VectorField(v))`	(Ex. 9.4)
`Laplacian(f)`	(Ex. 9.5)

Note that, for those commands with a vector argument, the argument must be specified as a `VectorField` and not as a simple vector.

Examples for Chapter 9

EXAMPLE 9.1 VECTORS, LENGTH, ADDITION, SCALAR MULTIPLICATION

In the applications in this chapter, vectors **v** are often given as "arrows" from an initial point P to a terminal point Q, where, for instance,

```
> P := Vector([2, 3, 7]); Pr := Vector[row]([2, 3, 7]);
```

$$P := \begin{bmatrix} 2 \\ 3 \\ 7 \end{bmatrix} \qquad Pr := \begin{bmatrix} 2 & 3 & 7 \end{bmatrix}$$

(note that the responses are arranged horizontally, rather than vertically, to save space) or

```
> Q := <5, -6, 3>; Qr := <5 | -6 | 3>;
```

$$Q := \begin{bmatrix} 5 \\ -6 \\ 3 \end{bmatrix} \qquad Qr := \begin{bmatrix} 5 & -6 & 3 \end{bmatrix}$$

The vector **v** has the components

```
> v := Q - P;                                  # Resp.
```

$$v := \begin{bmatrix} 3 \\ -9 \\ -4 \end{bmatrix}$$

and the length

```
> with(LinearAlgebra):
```

```
> lengthv := sqrt(DotProduct(v, v));           # Resp.
```

$$lengthv := \sqrt{106}$$

Vector addition and scalar multiplication proceed as in Chap. 7. For instance, let (with **v** as before)

```
> w := <4, -3, 7>;                             # Resp.
```

$$w := \begin{bmatrix} 4 \\ -3 \\ 7 \end{bmatrix}$$

```
> v + w; -3*v; 0.5*w;
```

$$\begin{bmatrix} 7 \\ -12 \\ 3 \end{bmatrix} \qquad \begin{bmatrix} -9 \\ 27 \\ 12 \end{bmatrix} \qquad \begin{bmatrix} 2. \\ -1.50000000000000 \\ 3.50000000000000 \end{bmatrix}$$

Similar Material in AEM: Sec. 9.1

EXAMPLE 9.2 INNER PRODUCT. CROSS PRODUCT

Inner products (dot products) have already occurred in Chap. 7 (see Example 7.1 in this Guide). In 2- and 3-space they can be motivated as the **work done by a force** in a displacement. For instance, find the work W done by the force $\mathbf{p} = [3,\ 2,\ 5]$ in the displacement from $A : (2, 3, -4)$ to $B : (3, 2, 6)$. Also find the angle between **p** and a vector in the direction of the displacement.

Solution. W is defined by $W = \mathbf{p} \bullet \mathbf{d}$, where **d** is the "displacement vector" from A to B; thus,

```
> with(LinearAlgebra):
```

```
> A := <2 | 3 | -4>; B := <3 | 2 | 6>; p := <3 | 2 | 5>;
```

$$A := \begin{bmatrix} 2 & 3 & -4 \end{bmatrix}$$
$$B := \begin{bmatrix} 3 & 2 & 6 \end{bmatrix}$$
$$p := \begin{bmatrix} 3 & 2 & 5 \end{bmatrix}$$

```
> d := B - A;                                   # Resp. d := [ 1  -1  10 ]
> W := DotProduct(p, d);                        # Resp. W := 51
```

For the **angle** you have $\mathbf{p} \bullet \mathbf{d} = \|\mathbf{p}\|\|\mathbf{d}\| \cos\theta$, hence $\theta = \arccos(\mathbf{p} \bullet \mathbf{d}/(\|\mathbf{p}\|\|\mathbf{d}\|))$. Hence type

```
> theta := arccos(DotProduct(p, d)/(Norm(p, 2)*Norm(d, 2)));
```

$$\theta := \arccos\left(\frac{1}{76}\sqrt{38}\sqrt{102}\right)$$

```
> VectorAngle(p, d);                            # Resp.
```

$$\left(\frac{1}{76}\sqrt{38}\sqrt{102}\right)$$

```
> evalf[4](convert(%, degrees));                # 4 gives 4 digits
```

$$34.97\ degrees$$

Cross products $\mathbf{v} = \mathbf{a} \times \mathbf{b}$ are vectors $\mathbf{v}$ perpendicular to $\mathbf{a}$ and $\mathbf{b}$ and of length $\|\mathbf{v}\|$ (equal to the area of the parallelogram with $\mathbf{a}$ and $\mathbf{b}$ as adjacent sides). (If $\mathbf{a}$ and $\mathbf{b}$ are parallel or at least one of them is $\mathbf{0}$, then $\mathbf{v} = \mathbf{0}$ by definition.) Hence, among other applications, cross products can be used to calculate the area of a triangle when its vertices are given, say $A: (2, 1, 4)$, $B: (3, -7, 2)$, $C: (6, 3, 8)$.

Solution. You need two vectors **a** and **b** representing two sides of the triangle, say, AB and AC. Accordingly, type

```
> A := <2 | 1 | 4>; B := <3 | -7 | 2>; C := <6 | 3 | 8>;
> a := B - A; b := C - A;
```

$$a := \begin{bmatrix} 1 & -8 & -2 \end{bmatrix}$$
$$b := \begin{bmatrix} 4 & 2 & 4 \end{bmatrix}$$

The command for the cross product is `CrossProduct(a, b)`. To obtain its length, type `Norm(..., 2)`, and the area of the triangle is 1/2 of this length. Thus,

```
> v := CrossProduct(a, b);                      # Resp. v := [ -28  -12  34 ]
> answer := evalf[4](Norm(v, 2)/2);             # Area to 4 digits
```

$$answer := 22.83$$

Recalling that a cross product can be written as a symbolical determinant whose first row is **i, j, k** (unit vectors in the positive directions of the coordinate axes) and whose second and third rows are the two vectors, you can check your cross product by typing

```
> r := <i | j | k>:
> C := <r, a, b>;                               # Resp. C :=
```

$$\begin{bmatrix} i & j & k \\ 1 & -8 & -2 \\ 4 & 2 & 4 \end{bmatrix}$$

```
> Determinant(C);                               # Resp. -28i + 34k - 12j
```

Similar Material in AEM: Secs. 9.2, 9.3

EXAMPLE 9.3 DIFFERENTIATION OF VECTORS. CURVES AND THEIR PROPERTIES

Investigate the main geometrical properties of the **helix**

$$\mathbf{r}(t) = [a\cos t,\ a\sin t,\ ct],$$

which lies on a cylinder of radius a, is right-handed if $c > 0$, and has pitch $2\pi c$.

Solution. You will need the first two derivatives and their inner products (dot products). Type

```
> restart:
```

```
> with(LinearAlgebra):
```

```
> r := <a*cos(t) | a*sin(t) | c*t>; # Resp.
```
$r := \begin{bmatrix} a\cos(t) & a\sin(t) & ct \end{bmatrix}$

You need to differentiate the vector, but `diff` doesn't work with a vector

```
> r1 := diff(r, t);
Error, non-algebraic expressions cannot be differentiated
```

so you need to use the "VectorCalculus" commands. Because the commands are often similar to standard Maple commands, you may prefer to avoid problems by not loading the `VectorCalculus` package but, instead, using the long form

```
> r1 := VectorCalculus[diff](r, t); r2 := VectorCalculus[diff](r1, t);
```

$$r1 := -a\sin(t)\,e_x + (a\cos(t))e_y + (c)e_z$$
$$r2 := -a\cos(t)\,e_x - a\sin(t)\,e_y$$

```
> i11 := DotProduct(r1, r1, conjugate = false); i12 := DotProduct(r1,
  r2, conjugate = false); i22 := DotProduct(r2, r2, conjugate = false);
```

$$i11 := a^2\sin(t)^2 + a^2\cos(t)^2 + c^2$$
$$i12 := 0$$
$$i22 := a^2\cos(t)^2 + a^2\sin(t)^2$$

```
> i22 := simplify(%);                          # Resp.
```
$i22 := a^2$

Arc length. The arc length s is the integral of the square root of `i11`,

```
> s := int(sqrt(simplify(i11)), t);            # Resp.
```
$s := \sqrt{a^2+c^2}\,t$

Curvature. The curvature κ of a curve represented by $\mathbf{r} = \mathbf{r}(t)$ is given by

```
> kappa := (i11*i22 - i12^2)^(1/2)/i11^(3/2);
```

$$\kappa := \frac{\sqrt{\left(a^2\sin(t)^2 + a^2\cos(t)^2 + c^2\right)a^2}}{\left(a^2\sin(t)^2 + a^2\cos(t)^2 + c^2\right)^{3/2}}$$

```
> assume(a, positive); assume(c, positive):
```

```
> simplify(kappa);                             # Resp.
```
$\dfrac{a\sim}{(a\sim^2 + c\sim^2)}$

Type `?sqrt` to understand what Maple is doing here; taking the square root was conditional on the radicand's being positive. Tildes indicate that a and c are conditioned to be positive. (Without this condition, Maple would not reduce $\sqrt{a^2}$ to a!)

Torsion. You will need the third derivative of $\mathbf{r}(t)$. The numerator of the torsion τ is the determinant of the matrix with $\mathbf{r}', \mathbf{r}'', \mathbf{r}'''$ as rows (or columns), and the denominator is $i11 * i22 - i12^2$. Hence type

```
> r3 := VectorCalculus[diff](r2, t);
```

$$r3 := (a\sim \sin(t))e_x - a\sim \cos(t)\, e_y$$

```
> tau := Determinant(<r1, r2, r3>)/(i11*i22 - i12^2);
```

$$\tau := \frac{a\sim^2 \cos(t)^2 c\sim + a\sim^2 \sin(t)^2 c\sim}{(a\sim^2 \sin(t)^2 + a\sim^2 \cos(t)^2 + c\sim^2)\, a\sim^2}$$

```
> simplify(tau);                                    # Resp.
```

$$\frac{c\sim}{a\sim^2 + c\sim^2}$$

Note that the helix has the remarkable property that both its curvature and its torsion are constant.

If you want to plot the helix, you have to choose specific values for a and c, for instance, $a = 3$ and $c = 2$. The optional `axes = NORMAL` introduces coordinate axes in space. The optional `numpoints = 400` makes the curve smooth (try plotting without these optons). Move the curve in space by clicking on the figure and 'dragging' it.

```
> r0 := <3*cos(t), 3*sin(t), 2*t>:
```

```
> with(plots):
```

```
> spacecurve([r0[1], r0[2], r0[3], t = 0..6*Pi], axes = NORMAL,
   numpoints = 500, orientation = [45, 70]);
```



Example 9.3. Helix on a cylinder of radius 2

Similar Material in AEM: Secs. 9.4, 9.5

EXAMPLE 9.4 GRADIENT. DIRECTIONAL DERIVATIVE. POTENTIAL

Gradient. Type `?grad` for information. The 'long' command to compute the gradient is `VectorCalculus[Gradient](f, [x, y, z])`. For instance, type

```
> restart:
```

```
> with(LinearAlgebra):
```

```
> f := 4*x^2 + 7*y^2 + z^2;                     # Resp.
```

$$f := 4x^2 + 7y^2 + z^2$$

```
> v := VectorCalculus[Gradient](f, [x, y, z]);
```
$$v := 8x\bar{e}_x + 14y\bar{e}_y + 2z\bar{e}_z$$

Main applications of the gradient occur in connection with directional derivatives, surface normals, and potentials, as we shall now see.

Directional derivative. The directional derivative of a function f at a point P in the direction of a vector **a** can be expressed in terms of the gradient by

$$\mathrm{D}_a\, f = (1/\|\mathbf{a}\|)\mathbf{a} \bullet \operatorname{grad} f.$$

For instance, find the directional derivative of the above f at the point (2, 1, 3) in the direction of the vector [1, 0, -2].

Solution. Type **a** and a corresponding unit vector $\mathbf{b} = \mathbf{a}/\|\mathbf{a}\|$, then the directional derivative, call it `deriv`. Finally take the value of the derivative at the given point.

```
> a := <-3 | 0 | 2>; b := a/Norm(a, 2);
```
$$a := \begin{bmatrix} -3 & 0 & 2 \end{bmatrix}$$
$$b := \begin{bmatrix} -\frac{3}{13}\sqrt{13} & 0 & \frac{2}{13}\sqrt{13} \end{bmatrix}$$

The command `DotProduct` expects its arguments to be vectors.

```
> deriv := DotProduct(b, Vector(v), conjugate = false);
```
$$deriv := -\frac{24}{13}\sqrt{13}\,x + \frac{4}{13}\sqrt{13}\,z$$

```
> subs(x = 2, y = 1, z = 3, deriv);                    # Resp. -36/13 sqrt(13)
```
$$-\frac{36}{13}\sqrt{13}$$

```
> evalf[4](%);                                          # Resp. -9.985
```

Hence f is decreasing in the direction of **a**.

Surface normal. The gradient grad f is a normal vector to a level surface $f = const$ of a function $f(x, y, z)$, that is, if grad f is not the zero vector, it is perpendicular to the surface $f =$ const passing through the point considered. For instance, find a normal vector to the cone $z^2 = 3\,(x^2 + y^2)$ at the point (2, 1, -3).

Solution. The cone S is the level surface $f = 0$ of the function

```
> f := 3*x^2 + 3*y^2 - z^2;                             # Resp. f := 3x^2 + 3y^2 - z^2
```

Thus, a normal vector of S is $\mathbf{N} = \operatorname{grad} f$,

```
> N := VectorCalculus[Gradient](f, [x, y, z]);
```
$$N := 6x\bar{e}_x + 6y\bar{e}_y - 2\bar{e}_z$$

```
> NP := subs(x = 2, y = 1, z = -3, [N[1], N[2], N[3]]);
```
$$NP := [12, 6, 6]$$

Hence a unit normal vector of S at the point $(2, 1, -3)$ is (multiply by the reciprocal length of `NP`)

```
> n := 1/Norm(Vector(NP), 2)*NP;                        # Resp. n := 1/36 sqrt(6) [12, 6, 6]
```
$$n := \frac{1}{36}\sqrt{6}\,[12, 6, 6]$$

Potential. A potential is a scalar function f associated with a vector function **v** such that $\mathbf{v} = \operatorname{grad} f$. Not every **v** has a potential. If a **v** does, it is generally more convenient to work with a single scalar function (the potential) than with the triple of component functions of $\mathbf{v}$. Consider the following basic example. Begin by typing the distance r of a point (x, y, z) from the origin. Then type $\mathbf{v}$. Note that $\|\mathbf{v}\| = c/r^2$, as it appears in Newton's law of gravitation or in Coulomb's law of attraction or repulsion of electrically charged particles, where c is a constant. Find the integral of all three components of $\mathbf{v}$. If they are the same, that common value is the potential for this vector function

```
> with(VectorCalculus):
> SetCoordinates('cartesian'[x,y,z]):
> r := sqrt(x^2 + y^2 + z^2);                    # Resp. r := sqrt(x^2+y^2+z^2)
> v := VectorField(<-c*x/r^3, -c*y/r^3, -c*z/r^3>);
```

$$v := -\frac{cx}{(x^2+y^2+z^2)^{3/2}}\bar{e}_x - \frac{cy}{(x^2+y^2+z^2)^{3/2}}\bar{e}_y - \frac{cz}{(x^2+y^2+z^2)^{3/2}}\bar{e}_z$$

```
> ScalarPotential(v);                             # Resp. c/sqrt(x^2+y^2+z^2)
```

This is the potential at (x, y, z) of a point mass or electrical point charge located at the origin.

Similar Material in AEM: Sec. 9.7

EXAMPLE 9.5 DIVERGENCE, LAPLACIAN, CURL

The command `Diverge(v)` is used to obtain the **divergence** ($\operatorname{div} \mathbf{v}$) of a vector function $\mathbf{v}(x, y, z) = [v_1(x, y, z),\ v_2(x, y, z),\ v_3(x, y, z)]$.

```
> with(LinearAlgebra):
> v := VectorCalculus[VectorField](<v1(x,y,z), v2(x,y,z), v3(x,y,z)>,
  'cartesian'[x,y,z]);
```

$$v := (v1\,(x, y, z))\bar{e}_x + (v2\,(x, y, z))\bar{e}_y + (v3\,(x, y, z))\bar{e}_z$$

```
> div := VectorCalculus[Divergence](v);
```

$$div := \frac{\partial}{\partial x} v1\,(x, y, z) + \frac{\partial}{\partial y} v2\,(x, y, z) + \frac{\partial}{\partial z} v3\,(x, y, z)$$

The process can be simplified by loading the package

```
> with(VectorCalculus):
> SetCoordinates('cartesian'[x,y,z]):
> v := VectorField(<v1(x,y,z), v2(x,y,z), v3(x,y,z)>);
```

$$v := (v1\,(x, y, z))\bar{e}_x + (v2\,(x, y, z))\bar{e}_y + (v3\,(x, y, z))\bar{e}_z$$

```
> div := Divergence(v);
```

$$div := \frac{\partial}{\partial x} v1\,(x, y, z) + \frac{\partial}{\partial y} v2\,(x, y, z) + \frac{\partial}{\partial z} v3\,(x, y, z)$$

For example, find the divergence of the following vector function $\mathbf{w}$:

```
> w := VectorField(x*y*z*<x | y | z>);
```

$$w := (x^2yz)\bar{e}_x + (xy^2z)\bar{e}_y + (xyz^2)\bar{e}_z$$

```
> divw := Divergence(w);                          # Resp. divw := 6xyz
```

Laplacian. The Laplacian of a function f referred to Cartesian coordinates is

$$\nabla^2 f = \operatorname{div}(\operatorname{grad} f) = f_{xx} + f_{yy} + f_{zz}.$$

For instance,

```
> f := x*y^2/z:
> lapf := diff(f, x, x) + diff(f, y, y) + diff(f, z, z);
```

$$lapf := \frac{2x}{z} + \frac{2xy^2}{z^3}$$

or, more quickly,

```
> Laplacian(f);
```

`# Resp.` $\frac{2x}{z} + \frac{2xy^2}{z^3}$

The Laplacian of a function f equals the divergence of grad f. Indeed,

```
> f := 'f':                    # f was defined - remove the definition
> Divergence(Gradient(f(x, y, z)));
```

$$\frac{\partial^2}{\partial x^2} f(x,y,z) + \frac{\partial^2}{\partial y^2} f(x,y,z) + \frac{\partial^2}{\partial z^2} f(x,y,z)$$

Curl. The command `Curl(v)` is used to obtain the curl of a vector function $\mathbf{v}$. The curl plays a role in connection with rotations. A **rotation** can be described by a rotation vector $\mathbf{w}$, whose direction is that of the axis of rotation and whose length equals the angular velocity. Then the velocity $\mathbf{v}$ at a point P with position vector $\mathbf{r}$ (whose initial point lies on the axis of rotation) is $\mathbf{v} = \mathbf{w} \times \mathbf{r}$.

```
> w := <w1, w2, w3>:  r := <x, y, z>:
> v := CrossProduct(w, r);
```

$$v := (w2\, z - w3\, y)\, e_x + (-w1\, z + w3\, x)\, e_y + (w1\, y - w2\, x)\, e_z$$

If you take the curl of $\mathbf{v}$, you obtain

```
> Curl(VectorField(v));
```

`# Resp.` $2w1\bar{e}_x + 2w2\bar{e}_y + 2w3\bar{e}_z$

This proves that the curl of the velocity vector of the rotation equals twice the rotation vector. This is a basic relation on rotations which characterizes the nature of the curl in this connection.

If a vector field $\mathbf{v}$ has a potential f so that $\mathbf{v} = \operatorname{grad} f$, then $\mathbf{v}$ is **irrotational,** that is, its curl is the zero vector. To prove this, type

```
> v := Gradient(f);
```

`# Resp.` $v := 0\bar{e}_x$

```
> v := Gradient(f(x,y,z));
```

$$v := \left(\frac{\partial}{\partial x} f(x,y,z)\right)\bar{e}_x + \left(\frac{\partial}{\partial y} f(x,y,z)\right)\bar{e}_y + \left(\frac{\partial}{\partial z} f(x,y,z)\right)\bar{e}_z$$

```
> Curl(v);
```

`# Resp.` $0\bar{e}_x$

Similarly, the divergence of the curl of $\mathbf{v}$ is zero. Indeed,

```
> Divergence(Curl(VectorField(<v1, v2, v3>)));
```

`# Resp.` 0

To avoid problems that might arise from having a package installed, it can be removed by

```
> unwith(VectorCalculus):
```

Similar Material in AEM: Secs. 9.8, 9.9

Problem Set for Chapter 9

Pr.9.1 (Components, length) Find the components and length of the vector $\mathbf{v}$ with initial point (6, 3, 7) and terminal point (5, 2, 3. (*AEM* Sec. 9.1)

Pr.9.2 (Addition, scalar multiplication) Find $4\mathbf{a} + 8\mathbf{c}$ and $4(\mathbf{a} + 2\mathbf{c})$, where $\mathbf{a} = [-2, -3, 5]$ and $\mathbf{c} = [7, -2, 8]$. (*AEM* Sec. 9.1)

Pr.9.3 (Resultant force) Find the resultant of the forces $\mathbf{p} = [7, 5, -6], \mathbf{q} = [14, -3, 1]$, $\mathbf{u} = [-21, -2, 5]$. (*AEM* Sec. 9.1)

Pr.9.4 (Equilibrium) Find **p** such that **p, q** = [-5, 7, -1], and **u** = [4, -4, -1] are in equilibrium. Sketch these forces. (*AEM* Sec. 9.1)

Pr.9.5 (Inner product) Find $7\mathbf{b} \bullet 3\mathbf{c}$ and $8\mathbf{b} \bullet \mathbf{c}$, where $\mathbf{b} = [6, -2, -4]$ and $\mathbf{c} = [2, -1, -1]$. (*AEM* Sec. 9.2)

Pr.9.6 (Angle) Find the angle between **b** and **c** in Pr.9.5 in two ways, (a) by using inner products, (b) by using the command `VectorAngle.` (*AEM* Sec. 9.2)

Pr.9.7 (Work) Find the work done by the force $\mathbf{p} = [5,\ 3,\ 3]$ in the displacement from the point $A : (6, 7, 0)$ to the point $B : (9, 5, 0$. Sketch the vectors involved. (*AEM* Sec. 9.2)

Pr.9.8 (Vector product) Find $\mathbf{a}\times\mathbf{c}$, $\mathbf{c}\times\mathbf{a}$, $|\mathbf{a}\times\mathbf{c}|$, $|\mathbf{c}\times\mathbf{a}|$, $\mathbf{a}\bullet\mathbf{c}$, where $\mathbf{a}$ = [5, 3, -4] and $\mathbf{c} = [7, -4, 3]$. (*AEM* Sec. 9.3)

Pr.9.9 (Scalar triple product) Find $(\mathbf{a}\times\mathbf{b})\bullet\mathbf{c}$ and $\mathbf{a}\bullet(\mathbf{b}\times\mathbf{c})$, where $\mathbf{b} = [6, -4, 0]$ and **a** and **c** are as in Pr.9.8. (*AEM* Sec. 9.3)

Pr.9.10 (Tetrahedron) Find the volume of the tetrahedron with vertices (5, 7, 8), (1, 5, 11), (6, 7, 8), (4, 6, 7). (*AEM* Sec. 9.3)

Pr.9.11 (Linear independence) Are the vectors $[7, 5, 9]$, $[73, 59, 71]$, $[5, 7, -5]$ linearly independent? (*AEM* Sec. 9.3)

Pr.9.12 (Plane) Find the plane through the points (6, 3, 7), (7, 5, -3,) (0, 5, 8). (*AEM* Sec. 9.3)

Pr.9.13 (Length of cross product) Show that for any vectors **a** and **b** the length of $\mathbf{a}\times\mathbf{b}$ equals the square root of $(\mathbf{a}\bullet\mathbf{a})(\mathbf{b}\bullet\mathbf{b}) - (\mathbf{a}\bullet\mathbf{b})^2$ (*AEM* Sec. 9.3)

Pr.9.14 (Vector field) Plot the vector field $\mathbf{v} = [y^2, x{-}1]$. (Type `?fieldplot`.) (*AEM* Sec. 9.4)

Pr.9.15 (Length of a curve) Find the length of the **catenary** $\mathbf{r} = [t, \cosh t]$ from $t = 0$ to $t = 1$. Plot this portion of the curve. (*AEM* Sec. 9.5)

Pr.9.16 (Lissajous curves) Plot the special Lissajous curve $[\sin t, \cos 3t]$ for $t = 0 \ldots 2\pi$ (named after the French physicist J. A. Lissajous, 1822-1880). First guess what the curve might look like. (*AEM* Sec. 9.5)

Pr.9.17 (Torsion) Find the torsion $\tau = (\mathbf{r}'\ \mathbf{r}''\ \mathbf{r}''')/[(\mathbf{r}'\bullet\mathbf{r}')(\mathbf{r}''\bullet\mathbf{r}'') - (\mathbf{r}'\bullet\mathbf{r}'')^2]$ of the curve $\mathbf{r}(t) = [t,\ t^2,\ t^3]$. (*AEM* Sec. 9.5)

Pr.9.18 (Tangential acceleration) Find the tangential acceleration of the curve $\mathbf{r}(t) = [\sin\, t,\ \cos\, 2t,\ -\sin\, 2t]$. Plot the curve. (*AEM* Sec. 9.5)

Pr.9.19 (Gradient) Find the gradient of the function $f(x, y) = 10 \ln (x^4 + y^4)$ and its value at the point (2, 0). Plot the gradient field in the first quadrant near the origin, say, for x and y from 0.1 to 0.5. (*AEM* Sec. 9.7)

Pr.9.20 (Directional derivative) Find the directional derivative of the function in Pr.9.19 at the point (0, 4) in the direction of the vector $\mathbf{a} = [1, -1]$. (*AEM* Sec. 9.7)

Pr.9.21 (Surface normal) Find a unit normal vector of the cone $z^2 = 5x^2 + 5y^2$ at the point (2, 2, 5). (*AEM* Sec. 9.7)

Pr.9.22 (Potential) Does $\mathbf{v} = [2xy \cos (yz),\ x^2 \cos (y\, z) - x^2\, y\, z\, \sin (y\, z), -x^2\, y^2\, \sin (y\, z)]$ have a potential? If so, find it. (*AEM* Sec. 9.7)

Pr.9.23 (Divergence) Find the divergence of $(x^2 + y^2)^{-1}[2y, -2x]$. (*AEM* Sec. 9.8)

Pr.9.24 (Divergence) Prove $\nabla \bullet (f\mathbf{v}) = f\nabla \bullet \mathbf{v} + \mathbf{v} \bullet \operatorname{grad} f$. (*AEM* Sec. 9.8)

Pr.9.25 (Curl) Illustrate the important relations $\operatorname{curl} (\operatorname{grad} f) = \mathbf{0}$ and $\nabla \bullet (\operatorname{curl} \mathbf{v}) = 0$ by examples of your own and list some typical physical applications of these relations. (*AEM* Sec. 9.9)