PART A. ORDINARY DIFFERENTIAL EQUATIONS (ODEs)

Content. First-order ODEs (Chap. 1)
Second and higher order ODEs (Chap. 2, 3)
Systems of ODEs (Chap. 4)
Series solution of ODEs (Chap. 5)
Solution of ODEs by Laplace transforms (Chap. 6)

DEtools package. For some techniques you will need this package. Load it by typing `with(DEtools):`. Typing `?DEtools` shows that the package contains commands for plotting, for solving special ODEs, etc. Click on any of its keywords listed, for instance, `linearsol`, to see what it means and how you can use it. You will ***not*** need the package all the time, but it can be helpful, for instance, for more complicated ODEs. For further help type `?odeadvisor` and then click on `separable` or `exact`, etc., to see in detail what it can do for you.

Derivatives y', y'', y''',... may be typed as `diff(y(x), x)`, `diff(y(x), x, x)`, `diff(y(x), x, x, x)`,... . For instance,

```
> y := exp(3*x)*sin(7*x);                                    # 3*x not 3x
```

$$y := e^{3\,x}\sin(7\,x)$$

```
> diff(y, x);
```
Resp. $3\,e^{3\,x}\sin(7\,x) + 7\,e^{3\,x}\cos(7\,x)$

```
> diff(y, x, x);
```
Resp. $-40\,e^{3\,x}\sin(7\,x) + 42\,e^{3\,x}\cos(7\,x)$

D-notation for derivatives. Those derivatives may also be typed as `D(y)(x)`, `(D@@2)(y)(x)`, `(D@@3)(y)(x)`, etc. For instance (type `?D` for information) The D-notation

```
> D(y)(x);
```
Resp. $\mathtt{D}(y)(x)$

```
> D(y)(2*x);
```
Resp. $\mathtt{D}(y)(2\,x)$

```
> D(sin)(x);
```
Resp. $\cos(x)$

```
> (D@@2)(sin)(x);
```
Resp. $-\sin(x)$

In order to use this method on functions of functions, you need to write such functions as ***composite functions*** by using the composition operator `@`.

```
> D(tan@(2*x));
```
Resp. $2\left(\tan^2+1\right)@(2\,x)\mathtt{D}(x)$

```
> D(exp@(3*x)*(sin@(7*x)));
```

$$3\ \exp@(3\,x)\mathtt{D}(x)\sin@(7\,x) + 7\ \exp@(3\,x)\cos@(7\,x)\mathtt{D}(x)$$

```
> factor(%);
```
Resp. $\exp@(3\,x)\mathtt{D}(x)\,(3\ \sin@(7\,x) + 7\ \cos@(7\,x))$

Keeping in mind that you are differentiating with respect to x (so that $D(x) = 1$), the derivative is $e^{3x}(3\sin 7x + 7\cos 7x)$, as before. We shall, in general, use the first of these two notations.

Integration. `int(f, x)` gives the indefinite integral (the antiderivative) and `int(f, x = a..b)` the definite integral. For instance,

```
> f := 3*x*exp(7*x);
```
Resp. $f := 3\,xe^{7\,x}$

```
> int(f, x);            # Remember to give the variable for integration.
```

$$\frac{3}{49}\,(7\,x-1)\,\mathrm{e}^{7\,x}$$

```
> int(f, x = -infinity..3);
```
Resp. $\frac{60}{49}\,\mathrm{e}^{21}$

```
> evalf[7](%);
```
Resp. $1.614877\ 10^{9}$

Command for solving ODEs and systems. `dsolve` gives general solutions as well as particular solutions of initial value problems. See the various examples.

Chapter 1

First-Order ODEs

Content. General solutions (Ex. 1.1)
Direction fields (Ex. 1.2, Prs. 1.1, 1.2)
Separable ODEs (Prs. 1.3-1.8)
Exact ODEs, integrating factors (Ex. 1.4, Prs. 1.9, 1.10)
Linear ODEs, mixing problems, electric circuits (Exs. 1.3, 1.6, Prs. 1.11, 1.13, 1.17, 1.18)
Bernoulli ODE, Verhulst population model (Ex. 1.5, Prs. 1.14, 1.15)
Picard iteration, do-loop (Prs. 1.19, 1.20)

Examples for Chapter 1

EXAMPLE 1.1 GENERAL SOLUTIONS

General solutions are obtained by `dsolve`. For instance,

```
> restart:
```

Type the ODE.

```
>  ode := diff(y(x), x) = -13*y(x);
```

$$ode := \frac{\mathrm{d}}{\mathrm{d}x}y\,(x) = -13\,y\,(x)$$

```
> dsolve(ode);
```
Resp. $y\,(x) = \_C1\,\mathrm{e}^{-13\,x}$

Give a name to the solution by which you can use it further, say, `sol`.

```
> sol := dsolve(ode);
```
Resp. $sol := y\,(x) = \_C1\,\mathrm{e}^{-13\,x}$

$\_C1$ is the Maple notation for the ***arbitrary constant***.

Initial value problems can also solved by `dsolve` by the simple step of placing the initial condition into the command. For instance,

```
> ypartic := dsolve({ode, y(0) = 2));
Error, ‘)‘ unexpected
```

***You must take care to distinguish between braces* {} *and parentheses* ().** Here,

```
> ypartic := dsolve({ode, y(0) = 2});
```

$$ypartic := y(x) = 2\,\mathrm{e}^{-13\,x}$$

Checking solutions obtained on the computer is at least as important as it is in working with paper and pencil – the computer will sometimes fool you. Type

```
> y1 := k*exp(-13*x):          # The general solution you just obtained.
> subs(y(x) = y1, ode);                              # The original ODE.
```

$$\frac{\partial}{\partial x}\left(k\mathrm{e}^{-13\,x}\right) = -13\,k\mathrm{e}^{-13\,x}$$

```
> eval(%);                                            # Does the LHS = RHS?
```

$$-13\,k\mathrm{e}^{-13\,x} = -13\,k\mathrm{e}^{-13\,x}$$

```
> subs(x = 0, ypartic);                        # Check the initial condition.
```

$$y(0) = 2\,\mathrm{e}^{0}$$

```
> simplify(%);                                         # Resp. y(0) = 2
> plot(ypartic, x = 0..1, ytickmarks = [2, 10, 20, 30, 40]);
Error, invalid input: plot expects its 1st argument, p, to be of type set,
array, list, rtable, algebraic, procedure, And(`module`, appliable), but
received y(x) = 2*exp(-13*x)
```

```
> plot(rhs(ypartic), x = 0..1, ytickmarks = [2, 10, 20, 30, 40]);
```



Example 1.1. Particular solution $y(x) = 2\,e^{3\,x}$

`rhs` means **right-hand side** and gives the function $2\,e^{-13\,x}$, which you want to plot, whereas `ypartic` alone gives the whole equation $y = 2\,e^{-13\,x}$.

Similar Material in AEM: Sec. 1.1

EXAMPLE 1.2 DIRECTION FIELDS

Direction fields and approximate solution curves of ODEs can be plotted on the computer by first loading the **DEtools package**, typing

```
> with(DEtools):
```

and then typing the ODE and points (x, y) through which you want to have approximate solution curves. Show this for the ODE $y' = x^3y$ and the two points (0, 1) and

(0, 2).

Solution. Type the ODE

```
> ode := diff(y(x), x) = x^3*y(x);
```

$$ode := \frac{d}{dx}y(x) = x^3 y(x)$$

and the given points (initial conditions for particular solutions represented by those curves)

```
> inits := {[0, 1], [0, 2]};                    # Resp. inits := {[0,1],[0,2]}
```

The plot command for the direction field and solution curves is `DEplot`. It must contain `y(x)` as shown, x and y ranges, and the initial values. `scaling = constrained` (equal scales on both axes) is optional. Try plotting it without.*

```
> DEplot(ode, y(x), x = -3..3, y = 0..3, inits, scaling = constrained,
  linecolor = black);
```



Example 1.2. Direction field for $y' = x^3 y$

The equation can be solved by separating variables, $y'/y = x^3$, and integration, $y = c \exp(x^4/4)$.

Similar Material in AEM: Sec. 1.2

EXAMPLE 1.3 MIXING PROBLEMS

Mixing problems involve a tank into which some substance such as brine flows, the content of the tank is stirred (this is the 'mixing'), and the mixture flows out. The model is the ODE

$$y' = \text{Salt inflow rate minus Salt outflow rate},$$

where $y(t)$ is the amount of salt in the tank at any time t and $y' = dy/dt$ is the time rate of change of $y(t)$. Assume the following. At $t = 0$ the tank contains 300 gal of water in which 70 lb of salt are dissolved. The inflow is 15 lb/min (5 gal of brine, each containing 3 lb of salt). 5 gal/min of mixture flows out. Hence the model is

$$y' = 15 - (5/300)y, \qquad y(0) = 40.$$

Solve this initial value problem by typing

```
> ode := diff(y(t), t) = 15 - (5/300)*y(t);
```

$$ode := \frac{d}{dt}y(t) = 15 - \frac{1}{60}\,y(t)$$

```
> ypartic := dsolve(ode, y(0) = 70);
```

$$ypartic := y(t) = 900 - 830\,e^{-\frac{1}{60}t}$$

We wish to plot this particular solution but make some common errors – in the first attempt, `rhs` (right-hand side) is missing, in the second attempt a parenthesis after `ypartic` is missing (although Maple reports a premature `;`).

```
> plot(ypartic, t = 0..250, labels = [t, y], title = `Salt content`);
Error, invalid input: plot expects its 1st argument, p, to be of type
set, array, list, rtable, algebraic, procedure, And (`module`, appliable),
but received y(t) = 900-830*exp(-(1/60)*t)
```

```
> plot(rhs(ypartic, t = 0..250, labels = [t, y], title = `Salt content`);
Error, `;` unexpected
```

```
> plot(rhs(ypartic), t = 0..250, y = 0..1000, labels = [t, y],
    xtickmarks = [0, 50, 100, 150, 200, 250], ytickmarks =
    [0, 70, 200, 400, 600, 800, 1000], title = `Salt content`);
```



Example 1.3. Salt content $y(t)$ in the tank

The plot illustrates that $y(t)$ approaches the limit of 900 lb.

Similar Material in AEM: Sec. 1.3

EXAMPLE 1.4 INTEGRATING FACTORS

Integrating factors convert nonexact ODEs into exact ODEs. Let the given ODE be

(1) $$P\,dx + Q\,dy = 5\,\sin(y^2)\,dx + 3\,x\,y\,\cos(y^2)\,dy = 0.$$

Thus

```
> P := 5*sin(y^2);                         # Resp.  P := 5 sin(y^2)
> Q := 3*x*y*cos(y^2);                     # Resp.  Q := 3 x y cos(y^2)
```

The exactness test fails. Indeed,

```
> diff(P, y) - diff(Q, x);                 # Resp.  7 cos(y^2) y
```

In the case of exactness the response would be zero.

Is there an integrating factor $F(x)$ depending only on x? [Ordinarily, an integrating factor (if it exists) would depend on both x and y.] The exactness condition for (1) multiplied by $F(x)$ is

```
> eq1 := diff(F(x)*P, y) - diff(F(x)*Q, x) = 0;
```

$$eq1 := 7\,F(x)\cos\left(y^2\right)y - 3\left(\frac{d}{dx}F(x)\right)xy\cos\left(y^2\right) = 0$$

Division by $y\cos(y^2)$ gives an equation no longer containing y. Hence this becomes a first-order ODE for $F(x)$, which you can solve by `dsolve`, so that you will get an integrating factor.

```
> simplify(eq1/(y*cos(y^2)));          # Resp.
```
$$-3\left(\frac{d}{dx}F(x)\right)x + 7\,F(x) = 0$$

```
> sol := dsolve(%);                    # Resp.
```
$$sol := F(x) = \_C1\,x^{7/3}$$

(If you write F on the left, you would get a warning. Try it.) Hence $x^{7/3}$ is an integrating factor. (You can choose $\_C1 = 1$.) You can now obtain an implicit solution $u(x, y) = const$ by integrating $x^{7/3}P$ with respect to x and $x^{7/3}Q$ with respect to y, typing

```
> x^(7/3)*P;                           # Resp.
```
$$5\,x^{7/3}\sin\left(y^2\right)$$

```
> x^(7/3)*Q;                           # Resp.
```
$$3\,x^{10/3}y\cos\left(y^2\right)$$

```
> int(x^(7/3)*P, x);                   # Resp.
```
$$\frac{3}{2}\,x^{10/3}\sin\left(y^2\right)$$

```
> int(x^(7/3)*Q, y);                   # Resp.
```
$$\frac{3}{2}\,x^{10/3}\sin\left(y^2\right)$$

Hence a solution is $x^{10/3}\sin(y^2) = const$.

Similar Material in AEM: Sec. 1.4

EXAMPLE 1.5 BERNOULLI'S EQUATION

Bernoulli's equation includes as a special case an important population model, the **Verhulst equation** $y' - Ay = -By^2$, where A and B are positive constants. Type this equation as

```
> ode := diff(y(x), x) - A*y(x) = -B*y(x)^2;
```

$$ode := \frac{d}{dx}y(x) - Ay(x) = -B\,(y(x))^2$$

where x is time. Solve it by `dsolve(...)`,

```
> sol := dsolve(ode);                  # Resp.
```
$$sol := y(x) = \frac{A}{e^{-Ax}\_C1\,A + B}$$

The `DEtools` package has a special command for Bernoulli equations, which confirms your result. Type

```
> with(DEtools):
```

```
> bernoullisol(ode);                   # Resp.
```
$$\left\{y(x) = \frac{A}{e^{-Ax}\_C1\,A + B}\right\}$$

For plotting you must choose specific values of A and B. Try $A = B = 1$.

```
> sol2 := subs(A = 1, B = 1, sol);
```
\# Resp. $sol2 := y(x) = \dfrac{1}{e^{-x}\_C1 + 1}$

From this obtain and plot three typical particular solutions

```
> y1 := subs(_C1 = 10, sol2);
```
\# Resp. $y1 := y(x) = \dfrac{1}{10\,e^{-x} + 1}$

```
> y2 := subs(_C1 = 0, sol2):
> y3 := subs(_C1 = -0.5, sol2):
```

Use ; instead of : to see responses. Plot the three solutions on common axes:

```
> plot(rhs(y1), rhs(y2), rhs(y3), x = 0..5, y = 0..2.3,labels = [x, y],
    ytickmarks = [0, 0.5, 1., 1.5, 2], title = `Verhulst population model`);
```



Example 1.5. Typical solution curves of the Verhulst ODE

Similar Material in AEM: Sec. 1.5

EXAMPLE 1.6 *RL*-CIRCUIT

The current $i(t)$ in an RL-circuit with $R = 70$ ohms, $L = 3$ henry, and electromotive force $110\cos 5t$ volts is obtained by solving the ODE

```
> ode := 3*diff(i(t), t) + 70*i(t) = 110*cos(5*t);
```

$$ode := 3\left(\frac{d}{dx} i(t)\right) + 70\, i(t) = 110\cos(5\,t)$$

Assume that $i(0) = 0$. The solution obtained by `dsolve` is

```
> sol := dsolve(ode, i(0) = 0);
```

$$sol := i(t) = \frac{308}{205}\cos(5\,t) + \frac{66}{205}\sin(5\,t) - \frac{308}{205}\,e^{-\frac{70}{3}t}$$

```
> sol2 := evalf[5](sol);
```

$$sol2 := i(t) = 1.5024\cos(5.\,t) + 0.32195\sin(5.\,t) - 1.5024\,e^{-23.333\,t}$$

It might be better to solve the general equation and then substitute the values for L and R.

```
> ode := L*diff(i(t), t) + R*i(t) = 110*cos(5*t);
```

$$ode := L\left(\frac{d}{dx}i(t)\right) + Ri(t) = 110\cos(5t)$$

```
> sol := dsolve(ode, i(0) = 0);
```

$$sol := i(t) = -\frac{110\,e^{-\frac{Rt}{L}}R}{R^2 + 25L^2} + \frac{110(\cos(5t)R + 5\sin(5t)L)}{R^2 + 25L^2}$$

```
> subs(R = 70, L = 3, sol);
```

$$i(t) = \frac{308}{205}\cos(5t) + \frac{66}{205}\sin(5t) - \frac{308}{205}e^{-\frac{70}{3}t}$$

```
> evalf[5](%);
```

$$i(t) = 1.5024\cos(5.t) + 0.32195\sin(5.t) - 1.5024\,e^{-23.333\,t}$$

The last result shows a reasonable number of digits. The steady-state solution is a harmonic motion with the frequency of the electromotive force. The exponential term dies out very quickly because $R/L = 70/3$ is large.

Problem Set for Chapter 1

Pr.1.1 (Direction field) Plot the direction field of the ODE $y' = -y^2$ and approximate solution curves through the points (0, 0.5), (0, 1), and (0, 3). (*AEM* Sec. 1.2)

Pr.1.2 (Direction field) Plot the direction field of $y' = -13x^3/17y$ and approximate solution curves through (0, 1) and (0, 1.4). (*AEM* Sec. 1.2)

Pr.1.3 (Exponential growth) Find and plot the solution of $y' = -3\,y$ satisfying $y(0) = 2$. (*AEM* Sec. 1.3)

Pr.1.4 (Exponential approach) Solve the initial value problem $y' + 0.5y = 1$, $y(0) = 0$. Plot the solution for $t = 0 \ldots 5$. (*AEM* Sec. 1.3)

Pr.1.5 (Exponential decay) Find the particular solution of $y' = ky$ satisfying $y(0) = 7$. Determine k such that, at $t = 2$, the solution $y(t)$ has decreased to half its initial value. (*AEM* Sec. 1.3)

Pr.1.6 (Initial value problem) Solve $y' = 0.5 + y^2$, $y(0) = 0$ and plot the solution curve. (*AEM* Sec. 1.3)

Pr.1.7 (Checking solutions) Check whether $y = 3\tan 3x$ satisfies $y' = 9 + y^2$. (*AEM* Sec. 1.3)

Pr.1.8 (Separable equation) Solve the initial value problem $y^2y' + x^2 = 0$, $y(0) = 1$. (*AEM* Sec. 1.3)

Pr.1.9 (Test for exactness) Is the following equation exact?

$$(5x^4 + 4\,x\,y^2)\,dx + (4\,x^2\,y + 4y^3)\,dy = 0.$$

Solve this equation. (*AEM* Sec. 1.4)

Pr.1.10 (Integrating factor) Show that $\cos{(2x)}$ is an integrating factor of $2\cos(y)\,dx - \tan{(2x)}\sin{(y)})\,dy = 0$ and solve the exact equation. (*AEM* Sec. 1.4)

Pr.1.11 (Linear differential equation) Find the general solution of $y' + y\cot\,x = \cos\,xx$ and from it the solution y satisfying the initial condition $y(\pi/2) = 4$. Plot y. (*AEM* Sec. 1.5)

Pr.1.12 (Beats) Find and plot the solution of the initial value problem

$$\csc\,x\,dy - (y\cot\,x\,\csc\,x + 100\cos\,30x)\,dx = 0, \qquad y(3\pi/2) = 0.$$

Pr.1.13 (Linear differential equation) The general solution of $y' + p(x)y = r(x)$ is

$$y(x) = e^{-h}\left[\int e^h\,r\,dx + c\right], \qquad h = \int p(x)\,dx.$$

Solve $y' + sin\,x\,y = xe^{\cos\,x}$ by this integral formula. (*AEM* Sec. 1.5)

Pr.1.14 (Bernoulli equation) Solve $y' + (1/3)y = (1/3)(1-2x)y^4$ (a) directly by `dsolve`, and (b) by setting $u = 1/y^3$, simplifying the ODE in u, and then applying `dsolve`. (*AEM* Sec. 1.5)

Pr.1.15 (Verhulst equation) Solve the Verhulst equation $y' - 7y = -11y^2$. Find three initial conditions such that the corresponding solutions are (1) increasing, (2) constant, (3) decreasing. Plot these solutions. (*AEM* Sec. 1.5)

Pr.1.16 (Orthogonal trajectories) Plot some of the hyperbolas $x^2 - y^2 = \mathrm{c}^2 = \mathrm{const}$. Find and plot some of their orthogonal trajectories, all curves and trajectories on common axes. (*AEM* Sec. 1.6)

Pr.1.17 (*RC*-circuit) The current $i(t)$ in an *RC*-circuit is governed by the ODE

$$R\,di/dt + i/C = dE/dt.$$

Solve this ODE for a general resistance R, capacitance C, and electromotive force $E(t) = E_0\,\sin{(\omega\,t)}$. Plot $i(t)$, assuming that $R = 3$ ohm, $C = 7$ farad, $\omega = 1\,\mathrm{sec}^{-1}$, $E_0 = 220$ volts, and $i(0) = 0$ ampere.

Pr.1.18 (*RL*-circuit) Model the current in an *RL*-circuit with $L = 0.5$ henry, $R = 7$ ohms, and a 5-volt battery. Determine and plot (on common axes) the current $i(t)$ when $i(0)$ equals 5, 2.5, 1, 0 amps.

Pr.1.19 (Picard iteration) Integrating $y' = f(x, y)$, $y(x_0) = y_0$ gives

$$y(x) = y_0 + \int_{x_0}^{x} f(t,\, y(t))\, dt.$$

This suggests the Picard iteration

$$y_n(x) = y_0 + \int_{x_0}^{x} f(t,\, y_{n-1}(t))\, dt, \qquad n = 1, 2, \dots .$$

Solve $y' = 1 + y^3$, $y(0) = 0$ by Picard iteration. (*AEM* Sec. 1.7)

Pr.1.20 (Experiment on Picard iteration by a do-loop) Obtain and plot the solution of Pr.1.19 (and a few initial value problems of your choice) by the do-loop with $N = 5$–10 steps

```
> N := 5:  y0 := 0:  pic(0) := 0:
  for n from 1 to N do
    pic(n) := sort(y0 + int(1 + (subs(x = t, pic(n - 1)))^3,
        t = 0..x));
  end do:
```

```
> S := seq(pic(n), n = 1..N):
```

```
> with(plots):
```

```
> plot(S, x = 0..1.3);
```

Explanation. `sort` has the effect that the powers come out in their natural order. The do-loop begins with the line that contains `do` and terminates with the line `end do`. Call this the opening line and the closing line of the do-loop. The closing line ends with a colon `:`, not with a `;`. The present do-loop consists of a single command (except for the opening line and the closing line) because Picard's iteration can be written as a single formula. (*AEM* Sec. 1.7)