Chapter 6

Laplace Transform Method for Solving ODEs

Content. Transforms, inverse transforms (Ex. 6.1, Prs. 6.1–6.5)
Solving ODEs (Exs. 6.2–6.4, Prs. 6.6, 6.7, 6.11–6.14)
Forced oscillations, resonance (Ex. 6.3)
Unit step, Dirac delta (Ex. 6.4, Prs. 6.9–6.15)
Solving systems (Ex. 6.5)
Differentiation, s-shifting, t-shifting, convolution, etc. (Ex. 6.6, Pr. 6.8)
Electric circuits, rectifiers (Prs. 6.11, 6.12, 6.14, 6.15)

```
> with(inttrans):
```

Laplace transforms. Type `laplace(f, t, s)`, f the given function, t its variable, s the variable of the transform F. For instance,

```
> laplace(1 + t + t^2 + exp(-a*t) + cosh(b*t) + sinh(c*t), t, s);
```

$$\frac{1}{s} + \frac{1}{s^2} + \frac{2}{s^3} + \frac{1}{s+a} + \frac{s}{-b^2+s^2} + \frac{c}{-c^2+s^2}$$

Inverse transforms. Type `invlaplace(F, s, t)`. For instance,

```
> invlaplace(1/s^5 + 1/(s^2 + 49), s, t);
```

$$\frac{1}{24}\,t^4 + \frac{1}{7}\,\sin(7\,t)$$

Transforms of the derivatives $\mathcal{L}(f\,') = s\mathcal{L}(f) - f(0)$ and $\mathcal{L}(f\,'') = s^2\mathcal{L}(f) - sf(0) - f\,'(0)$ (where $f' = df/dt$, etc.) are typed as

```
> laplace(diff(f(t), t), t, s);
```

$$s\,\mathit{laplace}\,(f\,(t)\,, t, s) - f\,(0)$$

```
> laplace(diff(f(t), t, t), t, s);
```

$$s^2\,\mathit{laplace}\,(f\,(t)\,, t, s) - \mathtt{D}\,(f)\,(0) - s f\,(0)$$

Solving ODEs by `dsolve(ode, y(t), method = laplace)` see Ex. 6.2, etc.

Examples for Chapter 6

EXAMPLE 6.1 FURTHER TRANSFORMS AND INVERSE TRANSFORMS

```
> with(inttrans):
```

For $e^{-a\,t}$, cosh bt, and sinh ct see before. For other frequently used functions such as cos at, sin at, $e^{a\,t}$ cos ωt, e^{at} sin ωt have the transforms

```
> laplace(cos(a*t), t, s);                         # Resp. s/(s^2 + a^2)
```

$$\frac{s}{s^2+a^2}$$

```
> laplace(sin(a*t), t, s);
```
Resp. $\frac{a}{s^2+a^2}$

```
> laplace(exp(a*t)*cos(omega*t), t, s);
```
Resp. $\frac{s-a}{(s-a)^2+\omega^2}$

```
> laplace(exp(a*t)*sin(omega*t), t, s);
```
Resp. $\frac{\omega}{(s-a)^2+\omega^2}$

Powers of t are transformed as follows.

```
> laplace(1, t, s);
```
Resp. $\frac{1}{s}$

```
> laplace(t, t, s);
```
Resp. $\frac{1}{s^2}$

```
> assume(n, positive);
> laplace(t^n, t, s);
```
Resp. $\Gamma\left(n\sim+1\right)s^{-n\sim-1}$

Hence t^n, with $n = 0, 1, 2, ...,$ has the transform $n!/s^{n+1}$ because $\Gamma(n+1) = n!$ when n is a positive integer. (The tilde is just a reminder that n is assumed to be positive.) **Inverse transforms** are more difficult to obtain than transforms. Corresponding tables are available (e.g. in AEM, Sec. 6.9), and you can use them just as you would use a table of integrals in integration. The Maple command `invlaplace(F, s, t)` gives help. (Note that now, `s` comes before `t` in the command!) For example,

```
> invlaplace((s + 1)/(s^2 + 5*s + 6), s, t);
```
Resp. $2\,\mathrm{e}^{-3\,t}-\mathrm{e}^{-2\,t}$

```
> invlaplace(1/s^6, s, t);
```
Resp. $\frac{1}{120}t^5$

```
> invlaplace((s - 2)/(s^2 - 9), s, t);
```
Resp. $\frac{1}{6}\mathrm{e}^{3\,t}+\frac{5}{6}\mathrm{e}^{-3\,t}$

```
> invlaplace(laplace(f, t, s), s, t);
```
Resp. f

```
> laplace(invlaplace(F, s, t), t, s);
```
Resp. F

Thus the two commands are inverses of each other, as had to be expected.

Similar Material in AEM: Sec. 6.1

EXAMPLE 6.2 DIFFERENTIAL EQUATIONS

Consider the differential equation and initial conditions

$$y'' + 2y' + 5y = 0, \qquad y(0) = 1, \quad y'(0) = -1.$$

I. Find a general solution $y(t)$ in three ways, first by `dsolve`, then by `dsolve(..., method = Laplace)`, and finally by using the subsidiary equation and transforming its solution $Y(s)$ back.

II. Apply the same three methods to the initial value problem.

Solution. I. Type the equation in the following form, and then apply `dsolve`.

```
> restart:
> ode := diff(y(t), t, t) + 2*diff(y(t), t) + 5*y(t) = 0;
```

$$ode := \frac{\mathrm{d}^2}{\mathrm{d}x^2}y\left(t\right)+2\left(\frac{\mathrm{d}}{\mathrm{d}x}y\left(t\right)\right)+5\,y\left(t\right)=0$$

```
> dsolve(ode);
```
\# Resp. $y(t) = \_C1\, e^{-t}\sin(2\,t) + \_C2\, e^{-t}\cos(2\,t)$

```
> factor(%);
```
\# Resp. $y(t) = e^{-t}(\_C1\ \sin(2\,t) + \_C2\ \cos(2\,t))$

This is a general solution in the form expected. (The arbitrary constants may appear interchanged.) Now turn to the Laplace transform method. Type

```
> with(inttrans):
```

and

```
> dsolve(ode, y(t), method = laplace);
```

$$y(t) = \frac{1}{2}\,e^{-t}\left(2\,y(0)\cos(2\,t) + \sin(2\,t)\left(\mathtt{D}(y)(0) + y(0)\right)\right)$$

```
> factor(%);
```

$$y(t) = \frac{1}{2}\,e^{-t}\left(\sin(2\,t)\,\mathtt{D}(y)(0) + \sin(2\,t)\,y(0) + 2\,y(0)\cos(2\,t)\right)$$

This form of the solution should not surprise you. It is needed to make the solution equal to $y(0)$ when $t = 0$ (so that the sine terms are zero) and its derivative equal to $y'(0)$.

In the third method you get the **subsidiary equation** by applying the command `laplace` to the differential equation,

```
> subsid := laplace(ode, t, s);
```

$$subsid := s^2\, laplace(y(t), t, s) - \mathtt{D}(y)(0) - sy(0) + 2\,s\, laplace(y(t), t, s) - 2\,y(0) + 5\ laplace(y(t), t, s) = 0$$

As the next step, solve the subsidiary equation *algebraically* for the transform (call it `Y`) of the unknown solution; thus,

```
> Y := solve(subsid, laplace(y(t), t, s));
```

$$Y := \frac{s\,y(0) + \mathtt{D}(y)(0) + 2y(0)}{s^2 + 2\,s + 5}$$

Finally, obtain $y(t)$ itself by taking the inverse of the expression for Y. This gives the general solution

```
> sol := invlaplace(Y, s, t);
```

$$sol := \frac{1}{2}\,e^{-t}\left(2\,y(0)\cos(2\,t) + \sin(2\,t)\left(\mathtt{D}(y)(0) + y(0)\right)\right)$$

in agreement with the previous result.

II. Now apply the same three methods to the initial value problem, remembering from Chap. 2 what will change when initial conditions are given. Basically, not much; put braces {} around as shown.

```
> dsolve({ode, y(0) = 1, D(y)(0) = -1});
```

$$y(t) = e^{-t}\cos(2\,t)$$

```
> dsolve({ode, y(0) = 1, D(y)(0) = -1}, y(t), method = laplace);
```

$$y(t) = e^{-t}\cos(2\,t)$$

```
> subsid2 := subs(y(0) = 1, D(y)(0) = -1, subsid);
```

$$subsid2 := s^2\, laplace(y(t), t, s) - 1 - s + 2\,s\, laplace(y(t), t, s) + 5\ laplace(y(t), t, s) = 0$$

```
> Y2 := solve(subsid2, laplace(y(t), t, s));
```

$$Y2 := \frac{1+s}{s^2+2\,s+5}$$

```
> y2 := invlaplace(Y2, s, t);                    # Resp.
```
$y2 := \mathrm{e}^{-t}\cos(2\,t)$

Similar Material in AEM: Sec. 6.2

EXAMPLE 6.3 FORCED VIBRATIONS. RESONANCE

Resonance occurs in undamped mechanical or electrical systems if the driving force (or the electromotive force, respectively) has the frequency equal to the natural frequency of the free vibrations of the system. This is the case for the differential equation

$$y'' + \omega_0^2\, y = K \sin \omega_0\, t.$$

Using the Laplace transform method, find the solution of this equation satisfying the initial conditions $y(0) = 0$, $y'(0) = 0$. Specifying K and ω_0 to be 1, plot that solution.

Solution. Type the equation; then obtain the subsidiary equation.

```
> ode := diff(y(t), t, t) + omega[0]^2*y(t) = K*sin(omega[0]*t);
```

$$ode := \frac{\mathrm{d}^2}{\mathrm{d}x^2}y(t) + \omega_0^2 y(t) = K\sin(\omega_0 t)$$

```
> with(inttrans):
```

```
> subsid := laplace(ode, t, s);
```

$$subsid := s^2\, laplace(y(t), t, s) - \mathrm{D}(y)(0) - s y(0) + \omega_0^2\, laplace(y(t), t, s) = \frac{K\omega_0}{s^2+\omega_0^2}$$

Now include the initial conditions in the subsidiary equation,

```
> subsid2 := subs(y(0) = 0, D(y)(0) = 0, subsid);
```

$$subsid2 := s^2\, laplace(y(t), t, s) + \omega_0^2\, laplace(y(t), t, s) = \frac{K\omega_0}{s^2+\omega_0^2}$$

Next solve the subsidiary equation *algebraically* for the Laplace transform, call it Y, of the unknown solution y:

```
> Y := solve(subsid2, laplace(y(t), t, s));
```

$$Y := \frac{K\omega_0}{(s^2+\omega_0^2)^2}$$

The inverse Laplace transform then gives the solution

```
> yp := invlaplace(Y, s, t);  # Resp.
```
$yp := \frac{1}{2}K\left(-\frac{t\cos(\omega_0 t)}{\omega_0} + \frac{\sin(\omega_0 t)}{\omega_0^2}\right)$

For plotting choose $K = 1$ and $\omega_0 = 1$ by the command

```
> yp0 := subs(K = 1, omega[0] = 1, yp);
```

$$yp0 := -\frac{1}{2}\,t\cos(t) + \frac{1}{2}\,\sin(t)$$

The maximum amplitude of the solution grows beyond bound as t increases indefinitely.

```
> plot(yp0, t = 0..100);
```



Example 6.3. Particular solution exhibiting resonance

Similar Material in AEM: Sec. 6.5

EXAMPLE 6.4 UNIT STEP FUNCTION (HEAVISIDE FUNCTION), DIRAC'S DELTA

The **unit step function** $u(t-a)$ (or **Heaviside function**) is 0 for $t < a$, has a jump of size 1 at $t = a$ (where we need not assign a value to it) and is 1 for $t > a$. Here, $a \geq 0$. The Maple notation is `Heaviside(t - a)`. Type `?Heaviside` for information. The transform of $u(t-a)$ is obtained by typing

```
> with(inttrans):
> assume(a >= 0):
> laplace(Heaviside(t - a), t, s);                    # Resp.
```

$$\frac{e^{-s a\sim}}{s}$$

$-s\,a\sim$ indicates that we have assumed something about a – in this case that a is non-negative. The unit step function is the basic building block for representing "piecewise functions". By this, Maple means piecewise continuous functions that are given by different expressions over different intervals. (Type `?piecewise`.) For instance, type

```
> f := piecewise(t < Pi, 2*t/Pi, t < 2*Pi, 2, t > 2*Pi, 2*cos(t));
```

$$f := \begin{cases} \frac{2t}{\pi} & t < \pi \\ 2 & t < 2\,\pi \\ 2\,\cos(t) & 2\,\pi < t \end{cases}$$

To explain: for $t < \pi$ the function equals $2t/\pi$. For π to $2\,\pi$ it equals 2. For $t > 2\,\pi$ it equals $2\cos t$. See the following figure. To represent `f` in terms of unit step functions, use the command `convert(..., Heaviside)`.

```
> f := convert(f, Heaviside);
```

$$f := \frac{2\,t}{\pi} - \frac{2\,t\,\mathrm{Heaviside}\,(-\pi+t)}{\pi} + 2\ \mathrm{Heaviside}\,(-2\pi+t) - 2\ \mathrm{Heaviside}\,(-\pi+t)$$
$$+2\ \cos(t)\ \mathrm{Heaviside}\,(-2\,\pi+t)$$

To plot f, type the following, where the (optional) command `scaling = constrained` gives equal scales on both axes.

```
> plot(f, t = 0..6*Pi, labels = [t, y], scaling = constrained);
```



Example 6.4.A. "Piecewise function"

The transform of f is

```
> F := laplace(f, t, s);          # Resp.
```

$$F := -\frac{2\,e^{-2\,s\pi}}{s\,(s^2+1)} + \frac{2\,(-e^{-s\pi}+1)}{\pi\,s^2}$$

Functions such as f may occur as driving forces in mechanics or as electromotive forces in circuits. For instance, the response of an undamped mass-spring system of mass 1 and spring constant 1 (initially at rest at equilibrium $y = 0$) to a single rectangular wave, say, $g(t) = 4$ if $2 < t < 10$ and 0 otherwise, is the solution of the ODE

```
> ode := diff(y(t), t, t) + y(t) = 4*Heaviside(t - 2)
   - 4*Heaviside(t - 10);
```

$$ode := \frac{d^2}{dx^2}y(t) + y(t) = 4\,\mathrm{Heaviside}(t-2) - 4\,\mathrm{Heaviside}(t-10)$$

```
> yp := dsolve(ode, y(0) = 0, D(y)(0) = 0, y(t), method = laplace);
```

$$yp := y(t) = 8\,\mathrm{Heaviside}(t-2)\sin\left(\frac{1}{2}t-1\right)^2 - 8\,\mathrm{Heaviside}(t-10)\sin\left(\frac{1}{2}t-5\right)^2$$

```
> plot([rhs(yp), 4*Heaviside(t - 2) - 4*Heaviside(t - 10)], t = 0..20,
    linestyle = [solid, dash]);
```



Example 6.4.B. Response of a mass-spring system to a single rectangular wave (dashed)

You see that the motion begins at $t = 2$. Until $t = 10$ there is a transition period (sinusoidal), followed by the steady-state harmonic oscillation that begins at $t = 10$ when the driving force shuts off.

The **Dirac delta function** $\delta(t-a)$ (also known as the **unit impulse function**) models an impulse at $t = a$ (practically, for instance, a hammer blow). The command is `Dirac(t - a)` (type `?Dirac`). Recall that we have assumed $a \geq 0$.

```
> Dirac(t - a);                                    # Resp. Dirac(t − a~)
> laplace(Dirac(t - a), t, s);                     # Resp. e^(−s a~)
```

For instance, if at $t = 2$ a hammer blow is imposed on an undamped mass-spring system of mass 1 and spring constant 3, the model is as follows and the solution [the displacement $y(t)$] is obtained by `dsolve(.., method = laplace)`, as before.

```
> ode := diff(y(t), t, t) + 9*y(t) = Dirac(t - 2);
```

$$ode := \frac{\mathrm{d}^2}{\mathrm{d}x^2} y(t) + 9\, y(t) = \mathrm{Dirac}(t-2)$$

```
> sol := dsolve(ode, y(0) = 0, D(y)(0) = 0, y(t), method = laplace);
```

$$sol := y(t) = \frac{1}{3}\, \mathrm{Heaviside}(t-2) \sin(3\,t-6)$$

```
> plot(rhs(sol), t = 0..10, xtickmarks = [0, 1, 5, 10]);
```



Example 6.4.C. Response of a mass-spring system to a hammer blow at $t = 2$

You see that the system remains at rest until the blow happens, and then immediately begins its steady-state harmonic oscillation.

Similar Material in AEM: Secs. 6.3, 6.4

EXAMPLE 6.5 SOLUTION OF SYSTEMS BY LAPLACE TRANSFORM

The process of solving systems by `dsolve(..., method = laplace)` is quite similar to that for single differential equations. For instance, type [where `D(y1)(t)` $= y_1{}'(t)$, etc.]

```
> restart:
> sys := D(y1)(t) + D(y2)(t) = cosh(t),
         D(y2)(t) + D(y3)(t) = 2 - exp(-t),
         D(y3)(t) + D(y1)(t) = exp(t);
```

$$sys := \mathtt{D}(y1)(t) + \mathtt{D}(y2)(t) = \cosh(t),\ \mathtt{D}(y2)(t) + \mathtt{D}(y3)(t) = 2 - \mathrm{e}^{-t},$$
$$\mathtt{D}(y3)(t) + \mathtt{D}(y1)(t) = \mathrm{e}^{t}$$

Then type initial conditions, so that you will obtain particular solutions; say,

```
> inits := y1(0) = 1, y2(0) = 0, y3(0) = 1;
```

$$inits := y1(0) = 1,\ y2(0) = 0,\ y3(0) = 1$$

Now apply the Laplace solution command. (y1, y2, y3 may appear in a different order.)

```
> with(inttrans):
```

```
> dsolve(sys, inits, y1(t), y2(t), y3(t), method = laplace);
```

$$\left\{y1\,(t) = -t + 1 + \frac{3}{2}\sinh(t),\, y2\,(t) = t - \frac{1}{2}\sinh(t),\, y3\,(t) = t + \frac{1}{4}e^{t} + \frac{3}{4}e^{-t}\right\}$$

Another system with unit step functions on the right, representing constant driving forces acting from $t = 0$ to 1 only, is the following.

```
> sys2 := D(y1)(t) = -y2(t) + 6 - 6*Heaviside(t - 1),
          D(y2)(t) = y1(t) + 6 - 6*Heaviside(t - 1);
```

$$sys2 := \mathtt{D}(y1)(t) = -y2\,(t) + 6 - 6\ \mathrm{Heaviside}\,(t-1),$$
$$\mathtt{D}(y2)(t) = y1\,(t) + 6 - 6\ \mathrm{Heaviside}\,(t-1)$$

Applying `dsolve(..., method = laplace)` and prescribing initial conditions, say, $y_1(0) = 0$, $y_2(0) = 0$, you obtain particular solutions, first in terms of unit step functions,

```
> y := dsolve(sys2, y1(0) = 0, y2(0) = 0, y1(t), y2(t), method = laplace);
```

$$y := \left\{y1(t) = -6 + 6\sin(t) + 6\cos(t) + 6\,\mathrm{Heaviside}(t-1)\left(2\sin\left(\frac{1}{2}t - \frac{1}{2}\right)^2 - \sin(t-1)\right),\, y2(t) = 6 + 6\sin(t) - 6\cos(t) - 6\,\mathrm{Heaviside}(t-1)\left(2\sin\left(\frac{1}{2}t - \frac{1}{2}\right)^2 + \sin(t-1)\right)\right\}$$

You can convert these solutions component-wise to the usual form

```
> p1 := convert(y[1], piecewise);
```

$$p1 := y1\,(t) = \begin{cases} -6 + 6\sin(t) + 6\cos(t) & t < 1 \\ 6\sin(1) + 6\cos(1) + \mathit{undefined} & t = 1 \\ -6 + 6\sin(t) + 6\cos(t) + 12\sin\left(\frac{1}{2}t - \frac{1}{2}\right)^2 - 6\sin(t-1) & 1 < t \end{cases}$$

```
> p2 := convert(y[2], piecewise);
```

$$p2 := y2\,(t) = \begin{cases} 6 + 6\sin(t) - 6\cos(t) & t < 1 \\ 6\cos(1) - 6\sin(1) + \mathit{undefined} & t = 1 \\ 6 + 6\sin(t) - 6\cos(t) - 12\sin\left(\frac{1}{2}t - \frac{1}{2}\right)^2 - 6\sin(t-1) & 1 < t \end{cases}$$

```
> plot(rhs(p1), rhs(p2), t = 0..5, linestyle = [dash, solid]);
```



Example 6.5. Solutions $y_1(t)$ (dashed curve) and $y_2(t)$ of the second system `sys2`

At $t = 1$, where the driving force jumps from 1 down to 0, the curves have a cusp, where the derivative (the slope) suddenly decreases, as you can see.

Similar Material in AEM: Sec. 6.7

EXAMPLE 6.6 FORMULAS ON GENERAL PROPERTIES OF THE LAPLACE TRANSFORM

The Laplace transform has various general properties that are essential for its practical usefulness. In this example we collect some of the most important formulas.

```
> restart:
> with(inttrans):
> laplace(a*f(t) + b*g(t), t, s);
```

$$a\,laplace\left(f\left(t\right),t,s\right)+b\,laplace\left(g\left(t\right),t,s\right)$$

```
> laplace(diff(f(t), t), t, s);              # Resp.
```
$s\,laplace\left(f\left(t\right),t,s\right)-f(0)$

```
> laplace(D(f)(t), t, s);                    # Resp.
```
$s\,laplace\left(f\left(t\right),t,s\right)-f(0)$

```
> laplace((D@@2)(f)(t), t, s);
```

$$s^2\,laplace\left(f\left(t\right),t,s\right)-\mathtt{D}\left(f\right)\left(0\right)-s\,f(0)$$

Thus, roughly speaking, differentiation of a function $f(t)$ corresponds to the multiplication of its transform by s. And integration of $f(t)$ from 0 to t (with the variable of integration denoted by τ) corresponds to division of the transform by s, namely,

```
> laplace(int(f(tau), tau = 0..t), t, s);        # Resp.
```
$\dfrac{laplace\left(f\left(t\right),t,s\right)}{s}$

Multiplying a function by e^{at} corresponds to replacing s by $s-a$ in the transform. This is called ***s*-shifting**. For instance,

```
> laplace(exp(a*t)*sin(t), t, s);                # Resp.
```
$\dfrac{1}{\left(s-a\right)^2+1}$

Multiplying a transform by e^{-as} corresponds to replacing t by $t-a$ in the function and to equating it to 0 for $t \le a$. This is called ***t*-shifting**. For instance,

```
> assume(a > 0):
> invlaplace(exp(-a*s)*laplace(sin(t), t, s), s, t);
```

$$-\,\mathrm{Heaviside}\,(t - a\sim)\sin(-t + a\sim)$$

Multiplying a function by t corresponds to differentiating its transform (and multiplying it by -1),

```
> laplace(t*f(t), t, s);
```
\# Resp. $-\frac{\partial}{\partial s}\, laplace\,(f\,(t)\,, t, s)$

Thus, $\mathcal{L}(t\,f(t)) = -F'(s)$. Taking the inverse transform on both sides (and interchanging the two sides), you thus have $\mathcal{L}^{-1}(F'(s)) = -t\,f(t)$; indeed,

```
> invlaplace(diff(laplace(f(t), t, s), s), s, t);
```
\# Resp. $-tf\,(t)$

Finally, taking the product of two transforms corresponds to taking the transform of the **convolution** (the integral shown) of the two corresponding functions,

```
> laplace(int(f(tau)*g(t - tau), tau = 0..t), t, s);
```

$$laplace\,(f\,(t)\,, t, s)\;\; laplace\,(g\,(t)\,, t, s)$$

For instance, if $f(t) = \cos\,t$ and $g(t) = e^t$, their transforms are $s/(s^2{+}1)$ and $1/(s{-}1)$, and you should get the product of these two transforms on the right; indeed,

```
> laplace(int(cos(tau)*exp(t - tau), tau = 0..t), t, s);
```

$$\frac{s}{(s-1)\,(s^2+1)}$$

Similar Material in AEM: Sec. 6.8

Problem Set for Chapter 6

Pr.6.1 (Transform) Find $\mathcal{L}(\sin\,\pi t)$ by evaluating the defining integral. (*AEM* Sec. 6.1)

Pr.6.2 (Transform by integration) Find $\mathcal{L}(\cos^2\omega t)$ by evaluating the defining integral of the transform. (*AEM* Sec. 6.1)

Pr.6.3 (Transform by integration) Find the transform of $f(t) = k$ if $0 < t < c$, $f(t) = 0$ otherwise, by integration. (*AEM* Sec. 6.1)

Pr.6.4 (Inverse transform) Using `invlaplace`, find the function whose transform is $(s - 1)/(s^2 - 9)$. (*AEM* Sec. 6.1)

Pr.6.5 (Inverse transform) Find the inverse transform of $\dfrac{s}{L^2\,s^2 + n^2\,\pi^2}$. (*AEM* Sec. 6.1)

Pr.6.6 (Initial value problem, subsidiary equation) Find the solution of the initial value problem $y' + 5y = 3.5\,e^{-5\,t}$, $y(0) = 1$ by obtaining the subsidiary equation, solving it, and transforming the solution back. (*AEM* Sec. 6.2)

Pr.6.7 (Initial value problem, subsidiary equation) Solve the initial value problem $y'' + y' - 6y = 6e^{-5\,t}$, $y(0) = 2$, $y'(0) = -14$ by using the subsidiary equation. (*AEM* Sec. 6.2)

Pr.6.8 (t-shifting) Plot the function $u(t-\pi)\sin t$ and find its transform. (*AEM* Sec. 6.3)

Pr.6.9 (Unit step function) Find and plot the inverse of $4(1-e^{-\pi\, s})s/(s^2+25)$. (*AEM* Sec. 6.3)

Pr.6.10 (Unit step function) Find and plot the inverse transform of $e^{-\pi\, s}/(s^2+1\,s+1)$. (*AEM* Sec. 6.3)

Pr.6.11 (RC-circuit, unit step) Find the subsidiary equation of

$$R\,i(t)+\frac{1}{C}\int_0^t i(\tau)\,d\tau = v(t), \qquad i(0)=0, \qquad i'(0)=0,$$

determining the current $i(t)$ in an RC-circuit, assuming that $v(t)=K=\text{const}$ for t from 1 to 3 and $v(t)=0$ otherwise. Find $i(t)$. Plot $i(t)$ when $R=2$ ohm, $C=1/2$ farad, and $K=110$ volts. (*AEM* Sec. 6.3)

Pr.6.12 (RC-circuit, Dirac's delta) Solve Pr.6.11 when $v(t)=K\,\delta(t-1)$, the other data being as before. (*AEM* Sec. 6.4)

Pr.6.13 (Experiment on repeated Dirac's delta) Solve $y''+4y=4\delta(t-\pi)-4\delta(t-4\,\pi)$, $y(0)=0$, $y'(0)=1$. First guess what the solution may look like. Then solve and plot. What will happen if you add further terms $\delta(t-3\pi)-\delta(t-4\pi)+-\ldots$? (*AEM* Sec. 6.4).

Pr.6.14 (RL-circuit) Using the command `dsolve(..., method = laplace)`, solve $Li'+Ri=v(t)$, $i(0)=0$, where $v(t)=\sin t$ if $0<t<3\pi$ and 0 otherwise. Plot the solution with $R=1$, $L=1$ and comment. (*AEM* Sec. 6.3)

Pr.6.15 (Full-wave rectifier) Plot the function obtained by the full-wave rectification of $\cos t$, that is, by multiplying the negative half-waves by -1. Let t go from 0 to $9\pi/2$. (*AEM* Sec. 6.4)