PART D. COMPLEX ANALYSIS

Content. Complex numbers and functions, conformal mapping (Chaps. 13/17)
Complex integration (Chap. 14)
Power series, Taylor series (Chap. 15)
Laurent series, singularities, residue integration (Chap. 16)
Potential theory (Chap. 18)

The Maple notation for $i = \sqrt{-1}$ is `I`. Do not use `I` otherwise. Symbols `+`, `−`, `*`, `/`, `^`, `evalf` are the same in complex as in real. `Re`, `Im`, `abs`, `argument`, and `conjugate` give the real part, imaginary part, absolute value, argument, and complex conjugate, respectively. `evalc` (suggesting 'evaluate complex') gives results in the form $a + ib$. This, as well as the polar form and the plotting of complex numbers, is illustrated in Example 13/17.1.

Chapter 13/17

Complex Numbers and Functions.

Conformal Mapping

Content. Complex numbers, polar form, plots (Ex. 13/17.1, Prs. 13/17.1–13/17.5)
Equations, roots, sets (Ex. 13/17.2, Prs. 13/17.6–13/17.11)
Cauchy-Riemann equations, harmonic functions (Ex. 13/17.3, Pr. 13/17.12)
Conformal mapping (Ex. 13/17.4, Prs. 13/17.13–13/17.15, 13/17.17)
Special complex functions (Exs. 13/17.5, 13/17.6, Prs. 13/17.16–13/17.20)

Examples for Chapter 13/17

EXAMPLE 13/17.1 COMPLEX NUMBERS. POLAR FORM. PLOTTING

Let us begin with illustrating arithmetic in complex. Let

```
> z1 := 6 + 9*I;
```
Resp. $z1 := 6 + 9\,\mathrm{I}$

```
> z2 := 5 - 12*I;
```
Resp. $z2 := 5 - 12\,\mathrm{I}$

Then, with the responses displayed horizontally to save space,

```
> z1 + z2; z1 - z2; z1*z2; z1/z2;
```

$$11 - 3\,\mathrm{I} \qquad 1 + 21\,\mathrm{I} \qquad 138 - 27\,\mathrm{I} \qquad -\frac{6}{13} + \frac{9}{13}\,\mathrm{I}$$

```
> evalf[5](z1/z2);
```
Resp. $-0.46154 + 0.69231\,\mathrm{I}$

```
> (6.0 + 9*I)/(5 - 12*I);    # Use decimal point to get decimal result
```
$$-0.4615384615 + 0.6923076923\,\mathrm{I}$$

If a result is not in the form $a + ib$, but you want it in this form, apply `evalc` (suggesting 'evaluate complex'; type `?evalc`). For instance,

```
> cos(7 + 3*I);                      # Resp. cos(7 + 3 I)
> evalc(cos(7 + 3*I));               # Resp. cos(7) cosh(3) - I sin(7) sinh(3)
> cos(7.0 + 3*I);                    # Resp. 7.590033075 - 6.581609575 I
```

`Re`, `Im`, `abs`, `argument`, and `conjugate` give the real part, imaginary part, absolute value (modulus), argument, and complex conjugate, respectively. For instance,

```
> Re(cos(7 + 3*I));                  # Resp. cos(7) cosh(3)
> evalf(%);                          # Resp. 7.590033077
> Im(cos(7 + 3*I));                  # Resp. -sin(7) sinh(3)
> abs(cos(7 + 3*I));                 # Resp. sqrt(cos(7)^2 cosh(3)^2 + sin(7)^2 sinh(3)^2)
> evalc(abs(x + I*y));               # Resp. sqrt(x^2 + y^2)
> evalc(argument(x + I*y));          # Resp. arctan(y, x)
> conjugate(-5.3 + 4.1*I);           # Resp. -5.3 - 4.1 I
> evalc(conjugate(x + I*y));         # Resp. x - I y
```

Polar form. Maple gives the **principal value** $\operatorname{Arg} z$ of the argument $\theta = \arg z$ of $z = |z| \exp i\theta$, defined by $-\pi < \theta \le \pi$. (Type `?polar` for information.) For instance,

```
> polar(5 + 12*I);                   # Resp. polar(13, arctan(12/5))
> polar(-3*I);                       # Resp. polar(3, -1/2 π)
> polar(a + I*b);                    # Resp. polar(|a + ib|, argument(a + ib))
```

To proceed ***from polar form to Cartesian form*** use `evalc`. For instance,

```
> polar(1 - I);                      # Resp. polar(sqrt(2), -1/4 π)
> evalc(polar(sqrt(2), -Pi/4));      # Resp. 1 - I
```

Multiplication in polar form can be done as follows.

```
> polar(r, t)^2*polar(s, w)^3;       # Resp. polar(r, t)^2 polar(s, w)^3
> simplify(%);                       # Resp. polar(r^2 s^3, 2t + 3w)
> simplify(polar(r, t)^5);           # Resp. polar(r^5, 5t)
```

Plotting complex numbers in the **complex plane.** To plot $3 + 2.5i$ and $3 - 2.5i$ as points, type

```
> plot([[3, 2.5], [3, -2.5]], x = 0..4, style = point, symbolsize = 20,
    labels = [x, y]);
```

Here `x = 0..4` gives the axis as shown. To obtain these numbers as vectors (line segments), type

```
> plot(2.5*x/3, -2.5*x/3, x = 0..4, labels = [x, y]);
```



Example 13/17.1. Complex numbers plotted as points



Example 13/17.1. Complex numbers plotted as vectors (line segments)

To combine both methods of plotting, load the plots package and then use `display`, as shown.

```
> with(plots):
```

```
> Points := plot([[3, 2.5], [3, -2.5]], x = 0..3, style = point,
    symbolsize = 20):
```

```
> Lines := plot(2.5*x/3, -2.5*x/3, x = 0..3):
```

```
> display(Points, Lines, labels = [x, y]);
```



Example 13/17.1. Combination of the two plots for points

Similar Material in AEM: Sec. 13.1

EXAMPLE 13/17.2 EQUATIONS. ROOTS. SETS IN THE COMPLEX PLANE

Quadratic and other equations can be solved by `solve` or `allvalues(RootOf(...))`. (Type `?solve`, `?allvalues` for information.) For instance, solve the equation $z^2 - (7+i)\,z + 18 + i = 0$.

Solution.

```
> eq := z^2 - (7 + I)*z + 18 +I= 0;
```

$$eq := z^2 + (-7 - \mathrm{I})\,z + 18 + \mathrm{I} = 0$$

```
> sol := solve(eq, z);
```

`# Resp.` $sol := 4 + 3\,\mathrm{I},\ 3 - 2\,\mathrm{I}$

```
> sol[1];                                          # Resp. 4 + 3I
> sol2 := allvalues(RootOf(eq));          # Resp. sol2 := 3 - 2I, 4 + 3I
```

Roots of unity. By definition, these are the roots of an equation $z^n = 1$, where n is a given integer. For instance, for $n = 4$ you get 1, i, -1, $-i$. Find and plot the roots of unity for $n = 9$.

Solution. Type

```
> Sol := solve(z^9 = 1.0, z);             # Try with 1 rather than 1.0
```

$$Sol := 1.0, -0.5000000000 - 0.8660254038\,I, -0.5000000000 + 0.8660254038\,I, 0.7660444431 + 0.6427876097\,I, -0.9396926208 + 0.3420201433\,I, 0.1736481776 - 0.9848077530\,I, 0.7660444431 - 0.6427876097\,I, 0.1736481776 + 0.9848077530\,I, -0.9396926208 - 0.3420201433\,I$$

The Maple `plot` cannot plot this *complex* sequence. One way to plot it is to convert it to a real sequence of pairs $[\mathrm{Re}\, z_k, \mathrm{Im}\, z_k]$, where z_k, $k = 1 \ldots 9$, are the roots.

```
> S := seq([Re(Sol[n]), Im(Sol[n])], n = 1..9);
```

$$S := [1., 0.], [-0.5000000000, -0.8660254038], [-0.5000000000, 0.8660254038], [0.7660444431, 0.6427876097], [-0.9396926208, 0.3420201433], [0.1736481776, -0.9848077530], [0.766044443, -0.6427876097], [0.1736481776, 0.9848077530], [-0.9396926208, -0.3420201433]$$

```
> with(plots):
> P1 := plot([S], style = point, symbolsize = 20):
> P2 := plot([cos(t), sin(t), t = 0..2*Pi], scaling = constrained):
> display(P1, P2);
```



Example 13/17.2. Roots of unity for $n = 9$ on the unit circle $|z| = 1$

Circles. The circle of radius 1 with center 0 in the figure is called the **unit circle**. Its interior is called the **open unit disk**. From the plotting command you can infer that a circle of radius r and center z_0, for instance, $r = 5$, $z_0 = -2 - i$, can be plotted by the following command (to get a circle instead of an ellipse, add `scaling = constrained`, as before).

```
> plot([-2 + 5*cos(t), -1 + 5*sin(t), t = 0..2*Pi], labels = [x, y]);
```

Besides disks, other important regions are those bounded by two concentric circles. Such a region is called an **annulus**. For instance, if the center is $-2 - i$ and the radii are 5 (this gives the previous circle) and 2.3, you can type (with `Center` for plotting the center)

```
> Outer := plot([-2 + 5*cos(t), -1 + 5*sin(t), t = 0..2*Pi]):
```

```
> Inner := plot([-2 + 2.3*cos(t), -1 + 2.3*sin(t), t = 0..2*Pi]):
> Center := plot([[-2, -1]], style = point, symbolsize = 20):
> display(Outer, Inner, Center, scaling = constrained, labels = [x, y]);
```



Example 13/17.2. Plot of arbitrary circle without `scaling = constrained`



Example 13/17.2. Annulus with center $z = -2 - i$ and radii 5 and 2.3

Similar Material in AEM: Secs. 13.2, 13.3

EXAMPLE 13/17.3 CAUCHY-RIEMANN EQUATIONS. HARMONIC FUNCTIONS

A standard notation for variables and functions is

$$z = x + iy \qquad \text{and} \qquad f(z) = u(x, y) + iv(x, y)$$

with real u and v. The **Cauchy-Riemann equations** (involving partial derivatives) are

$$u_x = v_y, \qquad u_y = -v_x$$

or, equivalently, $u_x - v_y = 0$, $u_y + v_x = 0$. These are the most important equations in the whole chapter because, roughly speaking, they are necessary and sufficient for $f(z)$ to be an **analytic function**.

For instance, find out whether or not $e^x(\cos y + i \sin y)$ is analytic.

Solution. Type the function and then the partial derivatives needed.

```
> f := exp(x)*(cos(y) + I*sin(y));         # Resp.
```
$f := e^x(\cos(y) + i\sin(y))$

```
> ux := diff(evalc(Re(f)), x);              # Resp.
```
$ux := e^x \cos(y)$

```
> vy := diff(evalc(Im(f)),y);               # Resp.
```
$vy := e^x \cos(y)$

Hence, the first Cauchy-Riemann equation is satisfied. And so is the second one,

```
> CR2 := diff(evalc(Re(f)), y) + diff(evalc(Im(f)), x);
```
$$CR2 := 0$$

Is $|z|^2 = x^2 + y^2$ analytic?

Solution. $f_2 = u + i\,v = x^2 + y^2$, hence $u = x^2 + y^2 = f_2$ and $v = 0$. Type

```
> f2 := x^2 + y^2;                          # Resp.
```
$f2 := x^2 + y^2$

```
> CR1 := diff(evalc(Re(f2)), x) - diff(evalc(Im(f2)), y);
```
$$CR1 := 2x$$

The response is $2x$ (instead of 0). The answer is no. You can stop here.

Harmonic functions. Solutions of Laplace's equation (with continuous second partial derivatives) are called *harmonic functions*. If u and v are harmonic and such that $f = u + iv$ is analytic, then v is called a **conjugate harmonic function** of u.

Show that $u = x^2 - y^2 - y$ is harmonic and find a conjugate harmonic v.

Solution.

```
> u := x^2 - y^2 - y;                    # Resp. u := x^2 - y^2 - y
> with(VectorCalculus):
> Laplacian(u, [x, y]);                  # Resp. 0
```

Hence u is harmonic. Now use the first Cauchy-Riemann equation. Calculate u_x and then integrate $u_x = v_y$ with respect to y, adding an arbitrary "constant" of integration $h(x)$ (not given by Maple).

```
> ux := diff(u, x);                      # Resp. ux := 2x
> v := int(ux, y) + h(x);                # Resp. v := 2xy + h(x)
```

Differentiate this with respect to x, obtaining v_x, and equate it to $-u_y$, thus satisfying the second Cauchy-Riemann equation. Then integrate the result with respect to x.

`> vx := diff(v, x) = -diff(u, y);` # Resp. $vx := 2y + \frac{\mathrm{d}}{\mathrm{d}x}h(x) = 2y + 1$

`> v := int(rhs(vx), x) + c;` # Resp. $v := (2y + 1)\,x + c$

You can now combine this with the given u, obtaining $f(z) = u(x, y) + iv(x, y)$. Using $x = (z + \overline{z})/2$, $y = (z - \overline{z})/2i$, you can express the result in terms of $z = x + iy$ as $f(z) = z^2 + iz + ic$ (c real) because

`> f := u + I*v;` # Resp. $f := x^2 - y^2 - y + \mathrm{I}\,((2y + 1)\,x + c)$

`> subs(x = (z + conjugate(z))/2, y = (z - conjugate(z))/(2*I), f);`

$$\left(\frac{1}{2}z + \frac{1}{2}\overline{z}\right)^2 + \frac{1}{4}(z - \overline{z})^2 + \frac{1}{2}\mathrm{I}\,(z - \overline{z}) + \mathrm{I}\left((-\mathrm{I}\,(z - \overline{z}) + 1)\left(\frac{1}{2}z + \frac{1}{2}\overline{z}\right) + c\right)$$

`> simplify(%);` # Resp. $z^2 + \mathrm{I}z + \mathrm{I}c$

Similar Material in AEM: Sec. 13.4

EXAMPLE 13/17.4 CONFORMAL MAPPING

The mapping of a region in the z-plane onto a region in the w-plane given by an analytic function $w = f(z) = u(x, y) + iv(x, y)$, $z = x + iy$, is **conformal** (angle-preserving in size and sense), except at points where the derivative $f'(z)$ is zero. You can map rectangles by the command `conformal(f, z = z_1..z_2)`. Type `?conformal`. Here z_1 and z_2 are two diagonally opposite vertices of the rectangle. You get the images of 11 lines $x = const$ and 11 lines $y =$ const. Or you get m and n lines, respectively, if you add `grid = [m, n]` in your plot command. Add `scaling = constrained` so that the image curves will intersect orthogonally.

```
> with(plots):
```

A 45-degree rotation of the rectangle $0 \le x \le 3$, $0 \le y \le 2$, combined with a dilatation by a factor $\sqrt{2}$, is obtained by typing

```
> conformal((1+I)*z, z = 0..3 + 2*I, grid = [6, 11], scaling = constrained,
   labels = [u, v]);
```

$w = z^2$ doubles angles at the origin, where $w' = 2z = 0$. For example,

```
> conformal(z^2, z = 0..1 + I, scaling = constrained, labels = [u, v]);
```



Example 13/17.4. Conformal mapping by $w = (1 + i)z$



Example 13/17.4. Conformal mapping of the square $0 \le x \le 1$, $0 \le y \le 1$ by $w = z^2$

An elliptic ring is obtained as the image of the annulus $2 \le |z| \le 3$ by using $w = z+1/z$ and including `coords = polar` in the plot command, so that `z = 2..3 + 2*Pi*I` is interpreted as $2 \le |z| \le 3$, $0 \le \theta \le 2\pi$.

```
> conformal(z + 1/z, z = 2..3 + 2*Pi*I, grid = [5, 11], labels = [u, v],
   scaling = constrained, coords = polar, numxy = [50, 50]);
```

The **linear fractional transformation** $w = (z-i)/(-iz+1)$ maps a large rectangle in the upper half-plane into the unit disk,

```
> conformal((z - I)/(-I*z + 1), z = -10..10 + 10*I, numxy = [100, 100],
   scaling = constrained, labels = [u, v]);
```



Example 13/17.4. Mapping of an annulus by $w = z + 1/z$, giving an elliptic ring Use of polar coordinates



Example 13/17.4. Mapping of a very large rectangle (theoretically: the upper half-plane) into the unit disk by a linear fractional transformation

An airfoil is obtained by $w = z + 1/z$ as the image of a special circle. Indeed, type

```
> z := 1/10*(1 + I + sqrt(122)*exp(I*t));
```

$$z := \frac{1}{10} + \frac{1}{10}\,\mathrm{I} + \frac{1}{10}\sqrt{122}\,\mathrm{e}^{\mathrm{I}t}$$

```
> w := z + 1/z;
```

$$w := \frac{1}{10} + \frac{1}{10}\,\mathrm{I} + \frac{1}{10}\sqrt{122}\,\mathrm{e}^{\mathrm{I}t} + \frac{1}{\frac{1}{10} + \frac{1}{10}\,\mathrm{I} + \frac{1}{10}\sqrt{122}\,\mathrm{e}^{\mathrm{I}t}}$$

```
> plot([Re(w), Im(w), t = 0..2*Pi], scaling = constrained,
  labels = [u, v]);
```



Example 13/17.4. **Joukowski airfoil** (the image of a special circle under $w = z + 1/z$)

Similar Material in AEM: Secs. 17.1 - 17.3

EXAMPLE 13/17.5 EXPONENTIAL, TRIGONOMETRIC, AND HYPERBOLIC FUNCTIONS

Evaluating functions proceeds as in real, except for the possibility of switching back and forth between z and $x + iy$. To return from $z = x + iy$ to z, type the command for unassigning, that is, `z := 'z'`. The following examples will explain the essential commands. These commands produce the given function $f(z)$, its value at $2 + 3\,i$, its real part $u(x, y)$, a plot of $u(x, y)$ as a surface, the absolute value $|f(2 + 3i)|$, the unassignment `z := 'z':`, and the derivative $f'(z)$.

```
> f := -5*z^3 + (12 - 2*I)*z^2 - I*z - 200;
```

$$f := -5z^3 + (12 - 2\,\mathrm{I})\,z^2 - \mathrm{I}z - 200$$

```
> subs(z = 2 + 3*I, f);
```
`# Resp.` $-3 + 107\,\mathrm{I}$

```
> u := evalc(Re(f));
```
`# Resp.` $u := -200 - 5z^3 + 12z^2$

```
> z := x + I*y;
```
`# Resp.` $z := x + \mathrm{I}y$

```
> u := evalc(Re(f));
```

$$u := -5x^3 + 15xy^2 + 12x^2 + 4xy - 12y^2 + y - 200$$

```
> plot3d(u, x = 0..5, y = 0..5, axes = NORMAL, labels = [x, y, "f"],
  orientation = [30, 70]);
```



Example 13/17.5. Real part u of f plotted as surface over the xy-plane

```
> evalf(subs(z = 2 + 3*I, abs(f)));                    # Resp. 107.0420478
```

```
> diff(f, z);
Error, (in VectorCalculus:-diff) invalid input:  diff received x+I*y,
which is not valid for its 2nd argument
```

```
> z := 'z':                                            # Unassign z
```

```
> diff(f, z);                                          # Resp.
```
$-15z^2 + (24 - 4\,\mathrm{I})\,z - \mathrm{I}$

Complex exponential function e^z, also written exp z. The command for the exponential function with basis $e = 2.71828...$ is exp(z). Accordingly,

```
> exp(3.7 - 1.2*I);                                    # Resp.
```
$14.65639438 - 37.69846859\,\mathrm{I}$

```
> exp(3 + 2*I);                                        # Resp.
```
$\mathrm{e}^{3+2\,\mathrm{I}}$

```
> evalc(exp(3 + 2*I));                                 # Resp.
```
$\mathrm{e}^3 \cos(2) + \mathrm{I}\mathrm{e}^3 \sin(2)$

```
> Re(exp(3 + 2*I));                                    # Resp.
```
$\mathrm{e}^3 \cos(2)$

```
> evalf(exp(3 + 2*I));                                 # Resp.
```
$-8.358532651 + 18.26372704\,\mathrm{I}$

```
> evalf[50](exp(1));
```
2.7182818284590452353602874713526624977572470937000

```
> evalc(exp(I*z));                                     # Resp.
```
$\cos(z) + \mathrm{I}\sin(z)$

Complex trigonometric functions cos z, sin z, tan z, cot z, sec z, cosec z. To obtain values of these functions or their real or imaginary parts, etc., use the commands just illustrated. For instance,

```
> z := x + I*y;                                        # Resp.
```
$z := x + \mathrm{I}y$

```
> evalc(cos(z));                                       # Resp.
```
$\cos(x)\cosh(y) - \mathrm{I}\sin(x)\sinh(y)$

```
> evalc(sin(z)) ;                                      # Resp.
```
$\sin(x)\cosh(y) + \mathrm{I}\cos(x)\sinh(y)$

```
> evalf(cos(2 + 3*I));                                 # Resp.
```
$-4.189625691 - 9.109227894\,\mathrm{I}$

```
> evalf[4](tan(1 + 3*I));                              # Resp.
```
$0.004517 + 1.002\,\mathrm{I}$

```
> plot3d(abs(sin(z)), x = 0..2*Pi, y = 0..1.8, axes = NORMAL);
```

The next plot shows the image of the rectangle $-\pi/2+0.5 < x < \pi/2-0.5$, $-1 < y < 1$ under the mapping by sin z.

```
> with(plots):
```

```
> conformal(sin(z), z = -Pi/2 + 0.5 - I..Pi/2 - 0.5 + I,
    labels = ["u", "v"]);
```



Example 13/17.5. Surface of the absolute value of sin z ("**modular surface**")



Example 13/17.5. Conformal mapping of a rectangle by sin z

cosine and **sine** are defined in terms of exponential functions. To obtain the defining formulas on the computer, type

```
> z := 'z':                                        # z was set to x + I*y
> (exp(I*z) + exp(-I*z))/2;                        # Resp.
```
$\frac{1}{2}e^{Iz} + \frac{1}{2}e^{-Iz}$
```
> convert(%, trig);                                # Resp.
```
$\cos(z)$
```
> convert((exp(I*z) - exp(-I*z))/(2*I), trig);     # Resp.
```
$\sin(z)$

Hyperbolic functions cosh *z*, sinh *z*, tanh *z*, coth *z*. In complex, these functions are closely related to the trigonometric functions. To see this for cosh, type the following. For sinh, tanh, and coth the situation is similar. Try it. Numeric values can be obtained as for the other functions just discussed.

```
> cosh(I*z);                                       # Resp.
```
$\cos(z)$
```
> cos(I*z);                                        # Resp.
```
$\cosh(z)$
```
> simplify(cosh(z)^2 - sinh(z)^2);                 # Resp.
```
1
```
> tanh(7 - 3*I);                                   # Resp.
```
$\tanh(7 - 3\,I)$
```
> evalf[5](tanh(2 - 3*I));                         # Resp.
```
$0.96539 + 0.0098844\,I$
```
> z := x + I*y:
> evalc(Re(tanh(z)));                              # Resp.
```
$\frac{\sinh(x)\cosh(x)}{\sinh(x)^2 + \cos(y)^2}$

Similar Material in AEM: Secs. 13.5, 13.6, 17.1, 17.4

EXAMPLE 13/17.6 COMPLEX LOGARITHM

In calculus, the natural logarithm ln x is defined for positive real x only and is single-valued (that is, for each such x it has only one value). In complex, ln z ($z \neq 0$) has infinitely many values $\ln z = \ln|z| + i\,\mathrm{Arg}\,z \pm 2n\pi i$ ($n = 0, 1, 2, ...$). These values all have the same real part $\ln|z|$. Their imaginary parts differ by integer multiples of

2π. Maple gives the **principal value** of $\ln z$, by definition corresponding to $n = 0$ and $-\pi < \operatorname{Arg} z \le \pi$ (called the **principal value** of $\arg z$ and denoted by $\operatorname{Arg} z$, as shown; see also Example 13/17.1 in this Guide). For example,

```
> evalc(ln(x + I*y));
```
\# Resp. $\frac{1}{2}\ln\left(x^2+y^2\right) + \mathrm{I}\arctan(y, x)$

```
> evalc(ln(5 - 12*I));
```
\# Resp. $\ln(13) - \mathrm{I}\arctan\left(\frac{12}{5}\right)$

```
> Re(ln(5 - 12*I));
```
\# Resp. $\ln(13)$

```
> Im(ln(5 - 12*I));
```
\# Resp. $-\arctan\left(\frac{12}{5}\right)$

```
> evalf(5 - 12*I);
```
\# Resp. $5. - 12.\mathrm{I}$

```
> ln(-1);
```
\# Resp. $\mathrm{I}\pi$

```
> evalf[50](ln(-1));
```
$$3.1415926535897932384626433832795028841971693993751\,\mathrm{I}$$

Derivative. The derivative is

```
> diff(ln(z), z);
```
\# Resp. $\frac{1}{z}$

Plotting. Type ?plot, ?plot[style]. Type $\ln(5 - 12i)$ in the form $a + bi$. Type all the values you want to plot; this is the complex sequence S.

```
> L := evalc(ln(5 - 12*I) + 2*n*Pi*I);
```
$$L := \ln(13) + \mathrm{I}\left(-\arctan\left(\frac{12}{5}\right) + 2n\pi\right)$$

```
> S := seq(evalf(L), n = -2..3);
```
$$S := 2.564949357 - 13.74237583\,\mathrm{I},\ 2.564949357 - 7.459190515\,\mathrm{I},\ 2.564949357 - 1.176005207\,\mathrm{I},\ 2.564949357 + 5.107180101\,\mathrm{I},\ 2.564949357 + 11.39036541\,\mathrm{I},\ 2.564949357 + 17.67355071\,\mathrm{I}$$

```
> plot(S, style = point);
Error, (in plot) unexpected options:  [2.564949357-7.459190515*I,
2.564949357-1.176005207*I, 2.564949357+5.107180101*I,
2.564949357+11.39036541*I, 2.564949357+17.67355071*I]
```

So this does not work. There are a couple of methods for plotting such values. The first is by converting to a real sequence S2 of terms $[\Re L, \Im L]$. The outer brackets in S2 seem to be essential. Try without. n goes from 1 to 6, the numbers of the six terms in S, not from -2 to 3. Try S[-2], S[0], S[1], S[2], etc. separately.

```
> S2 := [seq([Re(S[n]), Im(S[n])], n = 1..6)];
```
$$S2 := [[2.564949357, -13.74237583], [2.564949357, -7.459190515], [2.564949357, -1.176005207], [2.564949357, 5.107180101], [2.564949357, 5.107180101], [2.564949357, 11.39036541], [2.564949357, 17.67355071]]$$

```
> plot(S2, style = point, labels = [x, y]);
```

The second method is by use of complexplot which requires a list of complex values (the resulting figure is essentially the same).

```
> with(plots):
> S := [seq(evalf(L), n = -2..3)];
```

$$S := [2.564949357 - 13.74237583\,I, 2.564949357 - 7.459190515\,I,\\ 2.564949357 - 1.176005207\,I, 2.564949357 + 5.107180101\,I,\\ 2.564949357 + 11.39036541\,I, 2.564949357 + 17.67355071\,I]$$

```
> complexplot(S, x = 0..4, style = point, symbol = box, labels = [x, y]);
```



Example 13/17.6. Some values of $\ln(3 - 4i)$ using `plot`



Example 13/17.6. Some values of $\ln(3 - 4i)$ using `complexplot`

Similar Material in AEM: Secs. 13.7, 17.4

Problem Set for Chapter 13/17

Pr.13/17.1 (Complex arithmetic) Let $z_1 = 7 + 11i$, $z_2 = 5 - 3i$. Find $z_1 + z_2$, $z_1 - z_2$, $z_1 z_2$, z_1/z_2, $|z_1/z_2|$, $|z_1|/|z_2|$, $\Re z_1$, $\Im(z_1{}^2)$, $\operatorname{Arg} z_1$. (*AEM* Sec. 13.1)

Pr.13/17.2 (Complex arithmetic) Let $z_1 = 5 + 12i$, $z_2 = 3 - 7i$. Find $z_1\bar{z}_2$, $\bar{z}_1 z_2$ (why must these two products be conjugate?), $1/|z_1|$, $|z_1| + |z_2| - |z_1 + z_2|$ (why must this be nonnegative?), $\mathrm{Re}\,(\mathrm{z_1}^3)$, $(\mathrm{Re}\,\mathrm{z_1})^3$, $\mathrm{Im}\,((\mathrm{z_1} - \mathrm{z_2})/(\mathrm{z_1} + \mathrm{z_2}))$. (*AEM* Sec. 13.1)

Pr.13/17.3 (Real and imaginary parts) Obtain the important formulas $x = (z + \bar{z})/2$ and $y = (z - \bar{z})/2i$ on the computer. (*AEM* Sec. 13.1)

Pr.13/17.4 (Polar form) Using the computer, convert the following complex numbers to polar form and back. $1 - i$, $-3 - 3i$, $(1 - i)(-3 - 3i)$, -15, $(1 - i)/(1 + i)$, $((6 + 8i)/(4 - 3i))^2$. (*AEM* Sec. 13.2)

Pr.13/17.5 (Plotting complex numbers) Plot $(0.9 + 0.4\,i)^n$ for integer $n = -20, ..., 20$. Also plot the unit circle on the same axes. Can you find n for each of the 41 points that you see in the plot? (For plotting points see Example 13/17.6 in this Guide.)

Pr.13/17.6 (Quadratic equation in z^2) Solve $z^4 + (6i)z^2 + 3z^2 - 8 + 6i = 0$. (*AEM* Sec. 13.2)

Pr.13/17.7 (Roots) Find and plot all cube roots of $1 - i$, as well as the circle on which these roots lie. (*AEM* Sec. 13.2)

Pr.13/17.8 (Roots of unity) Find and plot the 16th roots of unity. Can you visualize the values before you plot them? (*AEM* Sec. 13.2)

Pr.13/17.9 (Annulus) Plot the annulus with center $2 + 2i$ whose outer circle passes through the origin and whose inner circle touches the coordinate axes. (*AEM* Sec. 13.3)

Pr.13/17.10 (Complex plane) Find the curve satisfying $|z + i|/|z - i| = 1$ (a) by a geometric argument without calculation, (b) on the computer. (*AEM* Sec. 13.3)

Pr.13/17.11 (Domain) Find the domain for whose points the sum of the distances from -1 and 1 is less than $\sqrt{8}$ (a) by a geometrical argument and inspection, (b) on the computer. (*AEM* Sec. 13.3)

Pr.13/17.12 (Cauchy-Riemann equations) Is $f(z) = \Re(z^3) + i\,\Im(z^3)$ analytic? (*AEM* Sec. 13.4)

Pr.13/17.13 (Experiment on conformal mapping) Make plots of the images of rectangles under $w = z^n$, $n = 2,\ 3,\ 4$, etc. How do these images change as functions of the exponent n? What happens to the image if you change the position of a rectangle in the plane? (*AEM* Secs. 17.1, 17.4)

Pr.13/17.14 (Experiment on conformal mapping) Experiment with the image of a square $a \le x \le a+1$, $b \le y \le b+1$ under the mapping $w = z^3$ and characterize the position and form of the image for various $a \ge 0$ and $b \ge 0$. (*AEM* Sec. 17.1)

Pr.13/17.15 (Inversion in the unit circle) Plot the square $1/2 \le x \le 3/2$, $1/2 \le y \le 3/2$ and its image under $w = 1/z$ as well as the unit circle on common axes. Discuss how the image curves correspond to the straight-line segments in the given square. (*AEM* Sec. 17.2)

Pr.13/17.16 (Exponential function) Find e^z (in the form $u + iv$) and $|e^z|$ if z equals $2 + 3\pi i$, $1 + i$, $2\pi(1 + i)$, $0.95 - 1.6i$, and $-\pi i/2$. (*AEM* Sec. 13.5)

Pr.13/17.17 (Conformal mapping by $\sin z$) Find the image of the rectangle $0 \le x \le 2\pi$, $1/2 \le y \le 1$. What are the images of the vertical sides $x = 0$ and $x = 2\pi$? (*Hint* : What will happen if you shorten the x-interval slightly? (*AEM* Sec. 17.4)

Pr.13/17.18 (Hyperbolic functions) Obtain the basic formulas $\cosh^2 z - \sinh^2 z = 1$, $\cosh(z)^2 + \sinh(z)^2$, and $\cosh^2 z + \sinh^2 z = \cosh 2z$ on the computer. (*AEM* Sec. 13.6)

Pr.13/17.19 (Natural logarithm ln z) Find the principal value $\mathrm{Ln}\, \mathrm{z}$ for $z = -5, -12 - 16i, 1 + i, 1 - i, -10 + 0.1i, -10 - 0.1i$. What does a comparison of the last two values illustrate? (*AEM* Sec. 13.7)

Pr.13/17.20 (General powers) Find (in the form $u + iv$) $(2\,i)^{2\,i}$, 4^{3-i}, $(3 + 6\,i)^i$, $i^{3/2}$. (*AEM* Sec. 13.7)

Chapter 14

Complex Integration

Content. Indefinite integration (Ex. 14.1)
Use of path (Ex. 14.2, Prs. 14.1–14.3, 14.5, 14.7–14.9)
Cauchy theorem and formula (Ex. 14.3, Pr. 14.6, 14.10)
Use of partial fractions (Prs. 14.4, 14.5)
Further integration methods follow in Chap. 15.

Examples for Chapter 14

EXAMPLE 14.1 INDEFINITE INTEGRATION OF ANALYTIC FUNCTIONS

Let $f(z)$ be analytic in a domain D that is **simply connected** (that is, every closed curve in D encloses only points of D). Then there exists an analytic function $F(z)$ such that $F'(z) = f(z)$ everywhere in D, and for all paths in D from any point z_0 to any point z_1 in D,

$$\int_{z_0}^{z_1} f(z)\,dz = F(z_1) - F(z_0).$$

[$F(z)$ is called an **antiderivative** or **indefinite integral** of $f(z)$, and this is the analog of a known formula from calculus, but note well that it applies to *analytic* functions only.] For instance,

```
> int(5*z^4, z = 0..1 + I);
```
`# Resp.` $-4-4\,\mathrm{I}$

```
> int(cos(z/2), z = -2*Pi*I..2*Pi*I);
```
`# Resp.` $2\,\mathrm{I}\left(-1+\mathrm{e}^{2\,\pi}\right)\mathrm{e}^{-\pi}$

```
> evalf[6](%);
```
`# Resp.` $46.1948\,\mathrm{I}$

```
> int(1/z, z = -I..I);
```
`# Resp.` *undefined*

In the last integral, an antiderivative is the principal value $\operatorname{Ln} z$ of $\ln z$, which is not analytic on the negative real axis $x \le 0$, so the paths from $-i$ to i must be restricted accordingly.

Similar Material in AEM: Sec. 14.1

EXAMPLE 14.2 INTEGRATION: USE OF PATH. PATH DEPENDENCE

If the function $f(z)$ you want to integrate is not analytic, the integral will generally depend on path, and you have to use a representation of the path C that is given. Let C be represented by $z = z(t)$, $a \le t \le b$, and be piecewise smooth. Let $f(z)$ be continuous on C. Then

$$\int_C f(z)\,dz = \int_a^b f(z(t))\,\dot{z}(t)\,dt \qquad \dot{z} = dz/dt.$$

For instance, if you want to integrate $f(z) = 1/z^n$ with any given $n = 1, 2, \ldots$ counterclockwise around the unit circle C, there is no simply connected domain containing C in which $1/z^n$ is analytic. ($1/z^n$ is analytic, for instance, in the annulus $D : 1/2 < |z| < 3/2$, but D is not simply connected.) Represent C, obtain $\dot{z}$, and then integrate from $a = 0$ to $b = 2\pi$.

```
> z := exp(I*t);                                  # Resp. z := e^(It)
> zdot := diff(z, t);                             # Resp. zdot := Ie^(It)
> J := int(1/z^n*zdot, t = 0..2*Pi);              # Resp. J := 0
```

This result is true for any **integer n**, but certainly not for any real n. Hence, in using results obtained by a CAS, in particular in complex analysis, one must be cautious.

For $n = 1$ you have $f(z) = 1/z$ and obtain the **very important result**

$$\oint_C \frac{1}{z}\, dz = 2\pi i \qquad \text{(counterclockwise around the unit circle)}$$

that is frequently needed throughout complex analysis,

```
> int(1/z*zdot, t = 0..2*Pi);                     # Resp. 2Iπ
```

Path dependence of the integral can be illustrated by many examples. For instance, integrate $f(z) = |z|^2$ from 0 to $1 + i$ over two different paths, (A) over the straight segment joining these points, (B) over the parabolic arc $y = x^2$ (see the figure).

Solution. (A)

```
> z1 := t + I*t;                                  # Resp. z1 := t + It
> z1dot := diff(z1, t);                           # Resp. z1dot := 1 + I
> f1 := abs(z1)^4;                                # Resp. f1 := (|t + It|)^4
> evalc(f1);                                      # Resp. 4t^4
> int(f1*z1dot, t = 0..1);                        # Resp. 4/5 + 4/5 I
```

(B) The value of the integral taken over the parabolic arc will differ from the value just obtained.

```
> z2 := t + I*t^2;                                # Resp. z2 := t + It^2
> z2dot := diff(z2, t);                           # Resp. z2dot := 1 + 2It
> f2 := abs(z2)^4;                                # Resp. f2 := |t + It^2|^4
> f2 := evalc(f2);                                # Resp. f2 := (t^2 + t^4)^2
> int(f2*z2dot, t = 0..1);                        # Resp. 188/315 + 31/30 I
> with(plots):
> P1 := plot([t, t, t = 0..1]):
> P2 := plot([t, t^2, t = 0..1]):
```

```
> display(P1, P2, xtickmarks = [0, 0.5, 1], ytickmarks = [0, 0.5, 1]);
```



Example 14.2. Paths of the integrals in (A) and (B)

Similar Material in AEM: Sec. 14.1

EXAMPLE 14.3 CONTOUR INTEGRATION BY CAUCHY'S INTEGRAL THEOREM AND FORMULA

Cauchy's integral theorem. Let $f(z)$ be analytic in a simply connected domain D. Let C be any closed path in D. Then the integral of $f(z)$ around C is zero.

Cauchy's integral formula and derivative formulas. Let $f(z)$ and D be as before. Let C be as before and *simple* (that is, without self-intersections). Let z_0 be any point inside C. Then the integral

$$\oint_C \frac{f(z)}{(z-z_0)^{n+1}}\,dz \tag{1}$$

(taken counterclockwise) equals $2\pi i f(z_0)$ for $n = 0$ ("**Cauchy's integral formula**") and ($f^{(n)}$ the nth derivative of f with respect to z)

$$\frac{2\,\pi\,i}{n!}f^{(n)}(z_0) \qquad \text{for } n = 1, 2, \ldots. \tag{2}$$

For instance, taking $f(z) = 1$ and $z_0 = 0$, you see, without any calculation, that for $n = 0$ (that is, for $1/z$) you get $2\pi i$, and for $n = 1, 2, \ldots$ the integrals (of $1/z^2$, $1/z^3, \ldots$) are all zero.

As a second application, find the integral of $\dfrac{e^z}{z\,e^z - 3\,i\,z}$ counterclockwise around the circle $|z| = 0.5$.

Solution. From Cauchy's integral formula with $f(z) = \dfrac{e^z}{e^z - 3\,i}$ and $z_0 = 0$ you obtain

```
> f := exp(z)/(exp(z) - 3*I);
```
Resp. $f := \dfrac{\mathrm{e}^z}{\mathrm{e}^z - 3\,\mathrm{I}}$

```
> 2*Pi*I*subs(z = 0, f);
```
Resp. $\dfrac{2\,\mathrm{I}\pi\,\mathrm{e}^0}{\mathrm{e}^0 - 3\,\mathrm{I}}$

```
> eval(%);
```
Resp. $\left(-\dfrac{3}{5} + \dfrac{1}{5}\,\mathrm{I}\right)\pi$

You still have to make sure that $f(z)$ is analytic everywhere inside and on C. Now, the only points at which something could happen are those at which the denominator of $f(z)$ is zero; that is, $e^z = 3i$, $z = \ln 3i = \ln 3 + \pi i/2 \pm 2n\pi i$, but these points lie outside C, as can be seen from

```
> evalc(solve(exp(z) - 3*I = 0, z));
```
`# Resp.` $\ln(3) + \frac{1}{2}\,\mathrm{I}\pi$

```
> evalf[4](abs(%));
```
`# Resp.` 1.917

and the fact that the other solutions are even greater in absolute value. Your result is now established.

As a third application, find the integral of $(\tan \pi z)/z^6$ counterclockwise around the circle $|z| = 1/4$.

Solution. From (1) and (2) with $f(z) = \tan \pi z$, $z_0 = 0$, and $n+1 = 6$, hence $n = 5$, you obtain

```
> 2*Pi*I/5!*diff(tan(Pi*z), z, z, z, z, z);
```

$$\frac{1}{60}\,\mathrm{I}\pi\left(88\left(1+\tan(\pi z)^2\right)^2\pi^5\tan(\pi z)^2 + 16\left(1+\tan(\pi z)^2\right)^3\pi^5 + 16\tan(\pi z)^4\left(1+\tan(\pi z)^2\right)\pi^5\right)$$

```
> eval(subs(z = 0, %));
```
`# Resp.` $\frac{4}{15}\,\mathrm{I}\pi^6$

A factor π results from $2\pi i$, and π^5 from the chain rule in the five differentiations.

Note further that the circle of integration is small enough that the points $\pm 1/2$, $\pm 3/2, \ldots$ where $\tan \pi z$ is not analytic lie outside the circle.

Similar Material in AEM: Secs. 14.2−14.4

Problem Set for Chapter 14

Pr.14.1 (Use of path) Integrate $\mathrm{Im}\, z$ over the shortest path from $1+i$ to $1+2i$. (*AEM* Sec. 14.1)

Pr.14.2 (Contour integral) Using a representation of the path, integrate $7/(z+i) - 5/(z+i)^2$ clockwise around the circle $|z-i| = 3$. Confirm the answer by the method in Example 14.3 in this Guide (*AEM* Sec. 14.1)

Pr.14.3 (Use of path) Integrate $\mathrm{Re}\, z$ from $2+i$ vertically upward to $2+2i$ and then horizontally to $5+2i$. (*AEM* Sec. 14.1)

Pr.14.4 (Partial fractions) Integrate $(2z^3 + z^2 + 4)/(z^4 - 4z^2)$ clockwise around the circle of radius 5 and center 2 first as given and then by using partial fractions. (*AEM* Sec. 14.3)

Pr.14.5 (Partial fractions) Integrate $(3z^2 + 2z - 1)/(z^3 - z^2 - 21z + 45)$ counterclockwise around the circle $|z| = 4$ by using partial fractions. (Type `?parfrac`. *AEM* Sec. 14.3)

Pr.14.6 (Derivative formulas) Integrate $(\sinh z)/z^6$ counterclockwise around the unit circle. (See Example 14.3 in this Guide)

Pr.14.7 (Path dependence) Integrate $\bar{z}$ from 0 to $1+i$ (A) along the shortest path, (B) along the parabola $y = x^2$. (*AEM* Sec. 14.1)

Pr.14.8 (Experiment on path dependence) Experiment with Pr.14.7 by integrating along $y = x^n$, $n = 2, 3, \ldots$, from 0 to $1 + i$, obtaining the limiting value of the integral as $n \to \infty$, and confirming the result by integration over the "limiting path" from 0 to 1 and then vertically up to $1 + i$.

Pr.14.9 (Use of contour) Integrate $\operatorname{Re} z^2$ clockwise around the boundary of the square with vertices 0, i, $1 + i$, 1. (*AEM* Sec. 14.1)

Pr.14.10 (Derivative formulas) Integrate $(z^4 + \cos\ z)/(z + i)^3$ counterclockwise around the boundary of the square with vertices ± 2 and $\pm 2i$. (*AEM* Sec. 14.4)