Pr.11.2 (Cosine series) Find the Fourier series of the function (sketch it) given by

$$f(n) = \begin{cases} 1 & \text{if } -\pi/2 < x < \pi/2 \\ -1 & \text{if } \quad \pi/2 < x < 3\pi/2\pi \end{cases}$$

and periodic with period 2π. (*AEM* Sec. 11.1)

Pr.11.2 $a_0 = 0$ (the mean value of $f(x)$ is zero). You can integrate over the two subintervals as given. Or you add 1 to the function to get

$$f(n) = \begin{cases} 2 & \text{if } -\pi/2 < x < \pi/2 \\ 0 & \text{if } \quad \pi/2 < x < 3\pi/2\pi \end{cases} .$$

Then take twice the integral of $(1/\pi)\,2\cos(nx)$ from 0 to $\pi/2$ (on the remaining portion of the interval this new function is 0) and drop the constant term 1 from the result. (Make a sketch.)

```
> an := 2/Pi*int(2*cos(n*x), x = 0..Pi/2);   # Resp.
```

$$an := \frac{4\sin\left(\frac{1}{2}n\pi\right)}{n\pi}$$

```
> S := sum(an*cos(n*x), n = 1..10);
```

$$S := \frac{4\cos(x)}{\pi} - \frac{4}{3}\frac{\cos(3x)}{\pi} + \frac{4}{5}\frac{\cos(5x)}{\pi} - \frac{4}{7}\frac{\cos(7x)}{\pi} + \frac{4}{9}\frac{\cos(9x)}{\pi}$$

```
> [4/Pi]*combine(S*Pi/4);
```

$$\left[\frac{4}{\pi}\right]\left(\cos(x) - \frac{1}{3}\cos(3x) + \frac{1}{5}\cos(5x) - \frac{1}{7}\cos(7x) + \frac{1}{9}\cos(9x)\right)$$

```
> plot(S, x = -Pi..2*Pi);
```



Problem 11.2. Partial sum of the Fourier series

Pr.11.4 (Half-wave rectifier) Pass the current $f(t) = \sin t$ through a half-wave rectifier that clips the negative portion of the wave. Find the Fourier series of the resulting periodic function $g(t)$ and plot some of its partial sums. (*AEM* Sec. 11.2)

Pr.11.4 $g(t)$ is neither even nor odd. Its period is 2π. It equals $\sin t$ from 0 to π and 0 from π to 2π. For the cosine coefficients, type

```
> f := sin(t):
```

```
> a0 := 1/(2*Pi)*int(f, t = 0..Pi);
```
`# Resp.` $a0 := \frac{1}{\pi}$

```
> an := 1/Pi*int(f*cos(n*t), t = 0..Pi);
```
`# Resp.` $an := -\frac{1+\cos(n\pi)}{\pi\,(n^2-1)}$

For $n = 1$ the denominator is 0, so a_1 must be evaluated separately. Similarly for b_1 below.

```
> a1 := 1/Pi*int(f*cos(t), t = 0..Pi);
```
`# Resp.` $a1 := 0$

```
> bn := 1/Pi*int(f*sin(n*t), t = 0..Pi);
```
`# Resp.` $bn := -\frac{\sin(n\pi)}{\pi\,(n^2-1)}$

```
> b1 := 1/Pi*int(f*sin(t), t = 0..Pi);
```
`# Resp.` $b1 := \frac{1}{2}$

Because $\sin n\pi = 0$ for $n = 2, 3, \ldots$, you have $b_2 = 0, b_3 = 0, \ldots$. This result of $b_n = 0$ for $n = 2, 3, \ldots$ is surprising. It means that $g(t) - (1/2)\sin t$ is an even function. Can you see it? Type a partial sum and plot it. Note that $g(t)$ is continuous. Hence a sum of few terms gives a good approximation.

```
> S := a0 + b1*sin(t) + sum(an*cos(n*t), n = 2..5);
```
$$S := \frac{1}{\pi} + \frac{1}{2}\sin(t) - \frac{2}{3}\frac{\cos(2t)}{\pi} - \frac{2}{15}\frac{\cos(4t)}{\pi}$$



Pr.11.6 (Behavior near a jump) Find the Fourier series of the periodic function $f(x) = \pi x^5/2$ $(-1 < x < 1)$ of period $p = 2$. Show, by plots, that, at the jumps, the partial sums give the arithmetic mean of the right-hand and left-hand limits of $f(x)$. (*AEM* Sec. 11.2)

Pr.11.6 The arithmetic mean equals 0. You would need many terms for obtaining relatively good approximations near the jumps (except for the Gibbs phenomenon). The function is odd. Hence $a_n = 0$.

```
> f := x -> Pi*x^5/2:
> bn := (1/1)*int(f(x)*sin(n*Pi*x), x = -1..1);
```

$$bn := -\frac{1}{\pi^5 n^6}\left(-5\,n^4\pi^4\sin(n\pi) - 120\sin(n\pi) + 120\cos(n\pi)n\pi - 20\,n^3\pi^3\cos(n\pi) + 60\,n^2\pi^2\sin(n\pi) + n^5\pi^5\cos(n\pi)\right)$$

```
> for j from 1 to 5 do
    s[j] := sum(bn*sin(n*Pi*x), n = 1..j);
  end:
> with(plots):
> P1 := plot(f(x), x = 0..1, thickness = 2):
> P2 := plot(f(x - 2), x = 1..2, thickness = 2):
> P3 := plot(s[1], s[2], s[3], s[4], s[5], x = 0..2,
   linestyle = [dot, spacedot, dashdot, dash, longdash]):
> display(P1, P2, P3, xtickmarks = [0, 1, 2],
   ytickmarks = [-1.57, -1, 0, 1, 1.57]);
```



Problem 11.6. Behavior of partial sums near a jump of the function

Pr.11.8 (Triangular wave) Find the Fourier series of the function (sketch it)

$$f(n) = \begin{cases} 2+x & \text{if } -2 < x < 0 \\ 2-x & \text{if } \quad 0 < x < 2 \end{cases}$$

Plot $f(x)$ and some partial sums. (*AEM* Sec. 11.2)

Pr.11.8 $p = 2L = 4$, $L = 2$. The function is even. Hence $b_n = 0$. You can integrate from 0 to 1/2, saving work.

```
> L := 2:
  f := piecewise(x < -2 and x < 0, 2 + x, x > 0 and x < 2, 2 - x);
```

$$f := f(n) = \begin{cases} x+2 & -2 < x \textbf{ and } \mathrm{x} < 0 \\ 2-x & 0 < x \textbf{ and } \mathrm{x} < 2 \end{cases}$$

```
> a0 := 2/(2*L)*int(f, x = 0..L);                         # Resp. a0 := 1
```

```
> an := (2/L)*int(f*cos(n*Pi*x/2), x = 0..L);
```

$$an := -\frac{4\,(-1+\cos(n\pi))}{n^2\pi^2}$$

```
> S5 := a0 + sum(an*cos(2*n*Pi*x), n = 1..5);
```

$$S5 := 1 + \frac{8\cos(2\,\pi\,x)}{\pi^2} + \frac{8}{9}\,\frac{\cos(6\,\pi\,x)}{\pi^2} + \frac{8}{25}\,\frac{\cos(10\,\pi\,x)}{\pi^2}$$

```
> a0 + [8/Pi^2]*combine((S5 - a0)*Pi^2/8);
```

$$1 + \left[\frac{8}{\pi^2}\right]\left(\cos(2\,\pi\,x) + \frac{1}{9}\,\cos(6\,\pi\,x) + \frac{1}{25}\,\cos(10\,\pi\,x)\right)$$

```
> plot(S5, x = -3..3, scaling = constrained);
```



Problem 11.8. Approximation of a triangular wave by a partial sum (S_5)

Pr.11.10 (Herringbone wave) Find the Fourier series of the function (sketch it)

$$f(n) = \begin{cases} x & \text{if } 0 < x < 1 \\ 1 - x & \text{if } 1 < x < 2\pi \end{cases}$$

Plot $f(x)$ and some partial sums. Observe the Gibbs phenomenon. (*AEM* Sec. 11.3)

Pr.11.10 The integration can be taken over any interval of length $2L$. The function is neither even nor odd. $a_0 = 0$ could be seen without integration.

```
> L := 1:  f := piecewise(x < 1, x, x > 1 and x < 2, 1 - x);
```

$$f := \begin{cases} x & x < 1 \\ 1 - x & 1 < x \textbf{ and } x < 2\pi \end{cases}$$

```
> a0 := 1/(2*L)*int(f, x = 0..2*L);                    # Resp. a0 := 0
```

```
> an := 1*int(f*cos(n*Pi*x), x = 0..2*L);
```

$$an := \frac{\sin(\pi n)\, n\pi + \cos(\pi n) - 1}{\pi^2 n^2} - \frac{2\, n\pi\, \sin(\pi n)\cos(\pi n) + 2\cos(\pi n)^2 - \cos(\pi n) - 1}{\pi^2 n^2}$$

```
> bn := 1*int(f*sin(n*Pi*x), x = 0..2*L);
```

$$bn := \frac{-\cos(\pi n)\,\pi n + \sin(\pi n)}{\pi^2 n^2} - \frac{-2\pi n\cos(\pi n)^2 + 2\sin(\pi n)\cos(\pi n) + \pi n - \sin(\pi n)}{\pi^2 n^2}$$

```
> S5 := sum(an*cos(n*Pi*x) + bn*sin(n*Pi*x), n = 1..5);
```

$$S5 := -4\frac{\cos(\pi x)}{\pi^2} + \frac{2\sin(\pi x)}{\pi} - \frac{4}{9}\frac{\cos(3\pi x)}{\pi^2} + \frac{2}{3}\frac{\sin(3\pi x)}{\pi} - \frac{4}{25}\frac{\cos(5\pi x)}{\pi^2} + \frac{2}{5}\frac{\sin(5\pi x)}{\pi}$$

```
> S50 := sum(an*cos(n*Pi*x) + bn*sin(n*Pi*x), n = 1..50):
> plot(S5, x = -2..2);     plot(S50, x = -2..2);
```



Problem 11.10. Partial sum S_5



Problem 11.10. Gibbs phenomenon in partial sum for S_{50}

Pr.12.2 (Animation) Show 5 cycles of the motion

$$u(x,t) = \sin\, x\, \cos\, t - (1/9) \sin\, 3x\, \cos\, 3t + (1/25) \sin\, 5x\, \cos\, 5t - (1/49) \sin\, 7x\, \cos\, 7t$$

(approximating the motion when the initial deflection of the string is "triangular"). (See Example 12.1 in this Guide. *AEM* Sec. 12.3)

Pr.12.2 $u(x,t)$ results from a partial sum of the Fourier series of the "triangular" initial deflection. Read and follow the instructions on animation in Example 12.1 in this Guide. Type

```
> u := sin(x)*cos(t) - 1/9*sin(3*x)*cos(3*t) + 1/25*sin(5*x)*cos(5*t) -
  1/49*sin(7*x)*cos(7*t);
```

$$u := \sin(x)\cos(t) - \frac{1}{9}\sin(3x)\cos(3t) + \frac{1}{25}\sin(5x)\cos(5t) - \frac{1}{49}\sin(7x)\cos(7t)$$

```
> with(plots):
```

```
> animate(u, x = 0..Pi, t = 0..10*Pi, frames = 100);
```



Problem 12.2. Approximate initial shape $u(x,0)$ of the vibrating string

Pr.12.4 (Separation of variables) Solve $u_{xx} - u_{yy} = 0$, choosing the separation constant positive, zero, and negative. (*AEM* Sec. 12.3)

Pr.12.4 Write $u_{xx} = u_{yy}$. Type

```
> u(x,y) := F(x)*G(y):
```

```
> pde := diff(u(x,y), x, x) = diff(u(x,y), y, y);
```

$$pde := \left(\frac{d^2}{dx^2}F(x)\right)G(y) = F(x)\left(\frac{d^2}{dy^2}G(y)\right)$$

```
> eq := pde/u(x,y);
```

$$eq := \frac{\frac{d^2}{dx^2}F(x)}{F(x)} = \frac{\frac{d^2}{dy^2}G(y)}{G(y)}$$

Now consider three cases, positive, zero, and negative separation constant.

```
> sol1p := dsolve(lhs(eq) = k^2);
```

$$sol1p := F(x) = \_C1\,e^{kx} + \_C2\,e^{-kx}$$

```
> sol2p := dsolve(rhs(eq) = k^2);
```

$$sol2p := G(y) = \_C1\,e^{ky} + \_C2\,e^{-ky}$$

```
> solp := rhs(sol1p)*rhs(sol2p);
```

$$solp := \left(\_C1\,e^{kx} + \_C2\,e^{-kx}\right)\left(\_C1\,e^{ky} + \_C2\,e^{-ky}\right)$$

Note that the four arbitrary constants should be denoted by four different letters. Similarly in the further cases. In the responses, cos and sin may appear in a different order.

```
> sol1n := dsolve(lhs(eq) = -k^2);
```

$$sol1n := F(x) = \_C1\sin(kx) + \_C2\cos(kx)$$

```
> sol2n := dsolve(rhs(eq) = -k^2);
```

$$sol2n := G(y) = \_C1\sin(ky) + \_C2\cos(ky)$$

```
> soln := rhs(sol1n)*rhs(sol2n);
```

$$soln := (\_C1\sin(kx) + \_C2\cos(kx))(\_C1\sin(ky) + \_C2\cos(ky))$$

```
> sol1z := dsolve(lhs(eq) = 0);          # Resp.
```

$sol1z := F(x) = \_C1\,x + \_C2$

```
> sol2z := dsolve(rhs(eq) = 0);          # Resp.
```

$sol2z := G(y) = \_C1\,y + \_C2$

```
> solz := rhs(sol1z)*rhs(sol2z);
```

$$solz := (\_C1\,x + \_C2)(\_C1\,y + \_C2)$$

Note that these nine commands may be combined into a single command

```
> seq(rhs(dsolve(lhs(eq) = k^2*Sign, F(x)))*rhs(dsolve(rhs(eq) =
  k^2*Sign, G(y))), Sign = -1..1);
```

$$(\_C1\sin(kx) + \_C2\cos(kx))(\_C1\sin(ky) + \_C2\cos(ky)),$$
$$(\_C1\,x + \_C2)(\_C1\,y + \_C2),\ \left(\_C1\,e^{kx} + \_C2\,e^{-kx}\right)\left(\_C1\,e^{-ky} + \_C2\,e^{ky}\right)$$

Pr.12.8 (Heat equation) Solve the heat equation in Example 12.3 in this Guide for a bar of length 5 with $c = 1$ and "parabolic" initial temperature $u(x, 0) = x(5 - x)$. (*AEM* Sec. 12.5)

Pr.12.8 The coefficients of the Fourier sine half-range expansion of the initial temperature are

```
> L := 5:  InitTemp := x*(5 - x):
> Bn := 2/L*int(InitTemp*sin(n*Pi*x/L), x = 0..L);
```

$$Bn := -\frac{50\,(n\pi \sin(n\pi) + 2\cos(n\pi) - 2)}{n^3\pi^3}$$

```
> S := seq(Bn, n = 1..7);        # Resp.
```
$S := \frac{200}{\pi^3}, 0, \frac{200}{27\pi^3}, 0, \frac{8}{5\pi^3}, 0, \frac{200}{343\pi^3}$

The first few partial sums of the series are

```
> S1 := sum(Bn*sin(n*Pi*x/L), n = 1..1);
> S3 := sum(Bn*sin(n*Pi*x/L), n = 1..3);
> S5 := sum(Bn*sin(n*Pi*x/L), n = 1..5);
```

S3 almost coincides with $f(x)$, as the following plot shows. The partial sum of the solution corresponding to S3 is

```
> u3 := sum(Bn*sin(n*Pi*x/L)*exp(-n^2*Pi^2/L^2*t), n = 1..3);
```

$$u3 := \frac{200\sin\left(\frac{1}{5}\pi x\right)e^{-\frac{1}{25}\pi^2 t}}{\pi^3} + \frac{200}{27}\frac{\sin\left(\frac{3}{5}\pi x\right)e^{-\frac{9}{25}\pi^2 t}}{\pi^3}$$

```
> plot(InitTemp, S[1]*sin(Pi*x/L),S[1]*sin(Pi*x/L) + S[3]*sin(3*Pi*x/L),
  InitTemp - S[1]*sin(Pi*x/L), InitTemp - S[1]*sin(Pi*x/L)
  - S[3]*sin(3*Pi*x/L), x = 0..L);
```



Problem 12.8. Initial temperature, partial sums of the corresponding Fourier series and their errors