Chapter 2/3

Linear ODEs of Second and Higher Order

Content. General solutions, initial value problems (Exs. 2.1, 3.2, Prs. 2.1, 2.2, 3.3, 3.7
Vibrating mass on a spring (Exs. 2.3, 2.4, Prs. 2.5, 2.8, 2.9, 2.10, 2.11
Euler-Cauchy equations (Ex. 2.5, Prs. 2.13, 2.14)
Wronskian (Exs. 2.6, 3.7, Prs. 2.4, 3.16)
Nonhomogeneous linear ODEs (Exs. 2.8, 3.9, 2.10, 3.11, 2.12, 3.13, Prs. 2.6, 2.18, 2.19, 2.20)
Resonance, beats, electric circuits (Exs. 2.14, 2.15, Prs. 2.15, 2.17, 2.21)

DEtools package, commands for **derivatives**, **integrals**, **solution of ODEs** see the opening of Part A.

The techniques required for ODEs of order two and higher are essentially the same (except, mainly, for the factoring of polynomials). To illustrate this, the examples and problems of Chap. 2 and 3 are intermingled.

Examples for Chapter 2/3

EXAMPLE 2.1 GENERAL SOLUTION. INITIAL VALUE PROBLEM

Find a general solution y of the given ODE. Find and plot the particular solution y_p satisfying the given initial conditions.

$$y'' + y' - 6y = 0, \qquad y(0) = -3, \qquad y'(0) = 4.$$

Solution. Type the ODE in either of two ways

```
> restart:
```

```
> ode := diff(y(x), x, x) + diff(y(x), x) - 6*y(x) = 0;
```

$$ode := \frac{d^2}{dx^2}y(x) + \frac{d}{dx}y(x) - 6\,y(x) = 0$$

```
> ode2 := (D@@2)(y)(x) + D(y)(x) - 6*y(x) = 0;
```

$$ode2 := D^{(2)}(y)(x) + \mathrm{D}(y)(x) - 6\,y(x) = 0$$

Obtain a general solution by `dsolve` from either one of these two equations, the results being the same:

```
> sol := dsolve(ode);                 # Resp. sol := y(x) = _C1 e^(2 x) + _C2 e^(-3 x)
```

```
> sol2 := dsolve(ode2);               # Resp. sol2 := y(x) = _C1 e^(2 x) + _C2 e^(-3 x)
```

You can obtain the particular solution of the initial value problem directly by the command

```
> ypartic := dsolve(ode, y(0) = -3, D(y)(0) = 4);
```

$$ypartic := y(x) = -e^{2\,x} - 2\,e^{-3\,x}$$

You can check this answer by determining the arbitrary constants in the general solution as follows. For this you need the derivative of `sol`

```
> yprime := diff(sol, x);
```

$$yprime := \frac{d}{dx}y(x) = 2\_C1\,e^{2x} - 3\_C2\,e^{-3x}$$

and from this the values of y and y' at $x = 0$,

```
> y0 := subs(x = 0, rhs(sol));
```
`# Resp.` $y0 := \_C1\,e^0 + \_C2\,e^0$

```
> y0 := eval(%);
```
`# Resp.` $y0 := \_C1 + \_C2$

```
> yprime0 := eval(subs(x = 0, rhs(yprime)));
```

$$yprime0 := 2\_C1 - 3\_C2$$

(You needed `eval` because `subs` usually does not evaluate by itself.) Equating `y0` to -3 and `yprime0` to 4 (the given initial values), you obtain the corresponding values of $\_C1$ and $\_C2$ by the command

```
> S := solve(y0 = -3, yprime0 = 4, _C1, _C2);
```

$$S := \{\_C1 = -1, \_C2 = -2\}$$

```
> subs(S, sol);
```
`# Resp.` $y(x) = -e^{2x} - 2\,e^{-3x}$

(If you are familiar with matrices and want to determine the two constants by using matrices, type `?Linsolve`.)

You get the figure of y_p by the commands

```
> plot(ypartic, x = 0..1.5);
Error, vvv invalid input:  plot expects its 1st argument, p, to be of
of type set, array, list, rtable, algebraic, procedure, And(‘module‘,
appliable), but received y(x) = -exp(2*x)-2*exp(-3*x)
```

```
> plot(rhs(ypartic), x = 0..1.5, y = -10..0,
  xtickmarks = [0, 0.5, 1, 1.5], ytickmarks = [0, -2, -4, -6, -8]);
```



Example 2.1. Particular solution $y_p = -e^{2x} - 2\,e^{-3x}$

Similar Material in AEM: Sec. 2.2

EXAMPLE 3.2 GENERAL SOLUTION. INITIAL VALUE PROBLEM

Find a general solution y of the given ODE. Find and plot the particular solution y_p satisfying the given initial conditions.

$$y''' + 2y'' - 5y' - 6y = 0, \qquad y(0) = -5, \qquad y'(0) = 15 \qquad y''(0) == 15.$$

Solution. . Type the ODE as

```
> ode := diff(y(x), x, x, x) + 2*diff(y(x), x, x) - 5*diff(y(x), x)
  - 6*y(x) = 0;
```

$$ode := \frac{d^3}{dx^3}y(x) + 2\left(\frac{d^2}{dx^2}y(x)\right) - 5\left(\frac{d}{dx}y(x)\right) - 6\,y(x) = 0$$

or, equivalently, as

```
> ode2 := (D@@3)(y)(x) + 2*(D@@2)(y)(x) - 5*D(y)(x) - 6*y(x) = 0;
```

$$ode2 := D^{(3)}(y)(x) + 2\,D^{(2)}(y)(x) - 5\,\mathrm{D}(y)(x) - 6\,y(x) = 0$$

Obtain a general solution by `dsolve` from either one of these two equations, the results are, of course, the same:

```
> sol := dsolve(ode);     # Resp.
```

$sol := y(x) = \_C1\,e^{2x} + \_C2\,e^{-x} + \_C3\,e^{-3x}$

```
> sol2 := dsolve(ode2);
```

$$sol2 := y(x) = \_C1\,e^{2x} + \_C2\,e^{-x} + \_C3\,e^{-3x}$$

Obtain the particular solution of the initial value problem directly by the command

```
> ypartic := dsolve(ode, y(0) = -5, D(y)(0) = 15, (D@@2)(y)(0)
  = -15);
```

$$ypartic := y(x) = 2\,e^{2x} - 2\,e^{-3x} - 5\,e^{-x}$$

You get the figure of y_p by the commands

```
> plot(rhs(ypartic), x = 0..1, y = -10..10, xtickmarks = [0, 0.5, 1],
  ytickmarks = [8, 6, 4, 2, 0, -2, -4, -6, -8]);
```



Example 3.2. Particular solution $y_p = 2e^{2x} - 5\,e^{-x} - 2e^{-3x}$

Similar Material in AEM: Sec. 3.2

EXAMPLE 2.3 MASS-SPRING SYSTEM. COMPLEX CHARACTERISTIC ROOTS. DAMPED OSCILLATIONS

Solve $y'' + 0.6\,y' + 5.04\,y = 0, \qquad y(0) = 0, \qquad y'(0) = 1.5.$

Solution. This is the model of the vertical vibrations of a weight (of mass 1) attached to the lower end of an elastic spring (of spring constant 5.04), whose upper end is fixed. This mechanical mass-spring system has damping (damping constant 0.6). It begins its motion at the displacement 0 (the position of static equilibrium) with an initial velocity 2.

Type the equation

```
> m := 1:  c := 0.6:  k := 5.04:
  ode := m*diff(y(t), t, t) + c*diff(y(t), t) + k*y(t) = 0;
```

$$ode := \frac{\mathrm{d}^2}{\mathrm{d}t^2}y(t) + 0.6\left(\frac{\mathrm{d}}{\mathrm{d}t}y(t)\right) + 5.04\,y(t) = 0$$

where t is time. Solve the initial value problem by `dsolve`,

```
> yp := dsolve(ode, y(0) = 0, D(y)(0) = 1.5);
```

$$yp := y(t) = \frac{1}{11}\sqrt{55}\,\mathrm{e}^{-\frac{3}{10}t}\sin\left(\frac{3}{10}\sqrt{55}\,t\right)$$

```
> plot(rhs(yp), t = 0..20);
```



Example 2.3. Damped oscillations

This is the typical form of damped vibrations governed by a linear ODE. The oscillations lie between the exponential curves $+\exp(-0.3t)$ and $-\exp(-0.3t)$. Damping takes energy from the system, so that the maximum amplitudes of the motion decrease with time and eventually go to zero. Note that the motion begins at 0 and with a positive slope, in agreement with the initial conditions.

Similar Material in AEM: Sec. 2.4

EXAMPLE 2.4 THE THREE CASES OF DAMPING

Consider the ODE $y'' + cy' + y = 0$ with $c = 1/2,\ 2,\ 3$. The term cy' is the damping term. No damping, $c = 0$, gives harmonic oscillations. For reasons of continuity, you should expect (decreasing) oscillations when $c > 0$ is small ("**underdamping**"), but non-oscillatory behavior when c is large ("**critical damping**" and "**overdamping**"). Show this by solving the equation with $c = 1/2,\ 2,\ 3$ for the case that the motion starts from rest at $y = 1$.

Solution. Type the equation in the three cases as

```
> restart:
```

```
> ode1 := diff(y1(t), t, t) + 1/2*diff(y1(t), t) + y1(t) = 0;
```

$$ode1 := \frac{d^2}{dt^2}y1(t) + \frac{1}{2}\frac{d}{dt}y1(t) + y1(t) = 0$$

```
> ode2 := diff(y2(t), t, t) + 2*diff(y2(t), t) + y2(t) = 0;
```

$$ode2 := \frac{d^2}{dt^2}y2(t) + 2\left(\frac{d}{dt}y2(t)\right) + y2(t) = 0$$

```
> ode3 := diff(y3(t), t, t) + 3*diff(y3(t), t) + y3(t) = 0;
```

$$ode3 := \frac{d^2}{dt^2}y3(t) + 3\left(\frac{d}{dt}y3(t)\right) + y3(t) = 0$$

The initial conditions are $y(0) = 1$, $y'(0) = 0$. This gives the solutions

```
> yp1 := dsolve(ode1, y1(0) = 1, D(y1)(0) = 0);
```

$$yp1 := y1(t) = \frac{1}{15}\sqrt{15}\,e^{-\frac{1}{4}t}\sin\left(\frac{1}{4}\sqrt{15}\,t\right) + e^{-\frac{1}{4}t}\cos\left(\frac{1}{4}\sqrt{15}\,t\right)$$

```
> yp2 := dsolve(ode2, y2(0) = 1, D(y2)(0) = 0);
```

$$yp2 := y2(t) = e^{-t} + e^{-t}t$$

```
> yp3 := dsolve(ode3, y3(0) = 1, D(y3)(0) = 0);
```

$$yp3 := y3(t) = \left(\frac{1}{2} + \frac{3}{10}\sqrt{5}\right)e^{\frac{1}{2}\left(\sqrt{5}-3\right)t} + \left(\frac{1}{2} - \frac{3}{10}\sqrt{5}\right)e^{-\frac{1}{2}\left(\sqrt{5}+3\right)t}$$

Plot these three particular solutions on common axes by the command

```
> plot(rhs(yp1), rhs(yp2), rhs(yp3), t = 0..20);
```



Example 2.4. Typical solutions for the three cases of damping

$c = 1/2$ (**underdamping**) gives the oscillatory solution. $c = 2$ (**critical damping**) corresponds to the **double root** of the characteristic equation $\lambda^2 + 2\,\lambda + 1 = 0$, and $c = 3$ (**overdamping**) gives a monotone decreasing solution (as does $c = 2$ in the present case).

If you change the initial velocity from 0 to -2, you obtain a critical solution that has a zero at $t = 1$ and is no longer monotone. Indeed, then the solution is

```
> yp2b := dsolve(ode2, y2(0) = 1, D(y2)(0) = -2);
```

$$yp2b := y2(t) = \mathrm{e}^{-t} - \mathrm{e}^{-t}t$$

```
> fsolve(rhs(yp2b) = 0);                                  # Resp. 1.
```

Similar Material in AEM: Sec. 2.4

EXAMPLE 2.5 THE THREE CASES FOR AN EULER–CAUCHY EQUATION

Solve

$$x^2y'' + axy' + y = 0.$$

Solution. The three cases correspond to, say, $a - 1 = 1/2, 2, 3$, thus, $a = 3/2, 3, 4$. Indeed, type the Euler–Cauchy equations as

```
> ode1 := x^2*diff(y1(x), x, x) + (3/2)*x*diff(y1(x), x) + y1(x) = 0;
```

$$ode1 := x^2\left(\frac{\mathrm{d}^2}{\mathrm{d}x^2}y1(x)\right) + \frac{3}{2}\,x\left(\frac{\mathrm{d}}{\mathrm{d}x}y1(x)\right) + y1(x) = 0$$

```
> ode2 := x^2*diff(y2(x), x, x) + 3*x*diff(y2(x), x) + y2(x) = 0;
```

$$ode2 := x^2\left(\frac{\mathrm{d}^2}{\mathrm{d}x^2}y2(x)\right) + 3\,x\left(\frac{\mathrm{d}}{\mathrm{d}x}y2(x)\right) + y2(x) = 0$$

```
> ode3 := x^2*diff(y3(x), x, x) + 4*x*diff(y3(x), x) + y3(x) = 0;
```

$$ode3 := x^2\left(\frac{\mathrm{d}^2}{\mathrm{d}x^2}y3(x)\right) + 4\,x\left(\frac{\mathrm{d}}{\mathrm{d}x}y3(x)\right) + y3(x) = 0$$

Choose initial conditions $y(1) = 2$, $y'(1) = 0$. You cannot choose conditions at $x = 0$, where the coefficients of the equation, divided by x^2, that is, a/x and $1/x^2$, are singular. Apply `dsolve` to obtain

```
> yp1 := dsolve(ode1, y1(1) = 2, D(y1)(1) = 0);
```

$$yp1 := y1(x) = \frac{2}{15}\,\frac{\sqrt{15}\sin\left(\frac{1}{4}\sqrt{15}\ln(x)\right)}{x^{1/4}} + 2\,\frac{\cos\left(\frac{1}{4}\sqrt{15}\ln(x)\right)}{x^{1/4}}$$

```
> yp2 := dsolve(ode2, y2(1) = 2, D(y2)(1) = 0);
```

$$yp2 := y2(x) = \frac{2}{x} + \frac{2\,\ln(x)}{x}$$

```
> yp3 := dsolve(ode3, y3(1) = 2, D(y3)(1) = 0);
```

$$yp3 := y3(x) = \left(1 + \frac{3}{5}\sqrt{5}\right)x^{\frac{1}{2}\sqrt{5} - \frac{3}{2}} + \left(1 - \frac{3}{5}\sqrt{5}\right)x^{-\frac{1}{2}\sqrt{5} - \frac{3}{2}}$$

```
> plot(rhs(yp1), rhs(yp2), rhs(yp3), x = 0..5, y = 0..2,
  ytickmarks = [0, 1, 2]);
```



Example 2.5. The three cases of solutions for an Euler-Cauchy equation

Similar Material in AEM: Sec. 2.5

EXAMPLE 2.6 WRONSKIAN

For $y'' + p_1(x)y' + p\_0(x)y = 0$ the **Wronskian** is the determinant

$$W(y_1, y_2) = \begin{vmatrix} y_1 & y_2 \\ y_1' & y_2' \end{vmatrix} = y_1\, y_2{}' - y_2\, y_1{}'.$$

$\{y_1, y_2\}$ is a basis of solutions of the equation (with continuous p_1 and p_0) if and only if $W(y_1, y_2)$ is different from zero. Show this for the basis $y_1 = \cosh kx$, $y_2 = \sinh kx$ of the ODE $y'' + k^2y = 0$ with $k \neq 0$.

Solution. The basis is

```
> y1 := cosh(k*x):  y2 := sinh(k*x):
```

From this you obtain the Wronskian, by typing

```
> W := y1*diff(y2, x) - y2*diff(y1, x);
```

$$W := \cosh(kx)^2\, k - \sinh(kx)^2\, k$$

```
> simplify(W);                                    # Resp. k
```

If $k = 0$, then $y_1 = 1$, $y_2 = 0$, which is not a basis. If k is not 0, then y_1, y_2 form a basis.

If you know about determinants and matrices, you can use them in connection with Maple. They are part of the `LinearAlgebra` package. Type `?LinearAlgebra[det]` for information. Load the package by typing

```
> with(LinearAlgebra):  with(VectorCalculus):
```

The Wronskian is the determinant of the 2×2 matrix which you can type as

```
> WM := Matrix([[y1, y2], [diff(y1, x), diff(y2, x)]]);
```

$$WM := \begin{bmatrix} \cosh(kx) & \sinh(kx) \\ \sinh(kx)\, k & \cosh(kx)\, k \end{bmatrix}$$

and obtain the determinant of **WM** simply by typing

```
> W := Determinant(WM);              # Resp. W := cosh(kx)^2 k - sinh(kx)^2 k
```

```
> simplify(W);                                    # Resp. k
```

Even more simply, type

```
> y := [y1, y2];                                  # Resp. y := [cosh(kx), sinh(kx)]
```

Then obtain the Wronskian matrix by typing

```
> WM := Wronskian(y, x);
```

$$WM := \begin{bmatrix} \cosh(kx) & \sinh(kx) \\ \sinh(kx)\,k & \cosh(kx)\,k \end{bmatrix}$$

and finally the Wronskian itself by typing

```
> W := Determinant(WM);                           # Resp. W := cosh(kx)^2 k - sinh(kx)^2 k
> simplify(W);                                    # Resp. k
```

EXAMPLE 3.7 WRONSKIAN

For an ODE of order 2 you can get away without knowing about determinants. For an ODE of greater order you will need determinants. We can extend the previous example. For instance, a basis of

$$y''' + 2y'' - 5y' - 6y = 0$$

is $y_1 = e^{2x}$, $y_2 = e^{-x}$, $y_3 = e^{-3\,x}$ (verify that these are solutions), as can be seen by typing the corresponding 3×3 Wronskian matrix (whose third row contains the second derivatives of the three solutions). Hence type the solutions

```
> with(LinearAlgebra):  with(VectorCalculus):
> y1 := exp(2*x):  y2 := exp(-x):  y3 := exp(-3*x):
```

then

```
> y := [y1, y2, y3];                              # Resp. y := [e^(2x), e^(-x), e^(-3x)]
> A := Wronskian(y, x);
```

$$A := \begin{bmatrix} e^{2\,x} & e^{-x} & e^{-3\,x} \\ 2\,e^{2\,x} & -e^{-x} & -3\,e^{-3\,x} \\ 4\,e^{2\,x} & e^{-x} & 9\,e^{-3\,x} \end{bmatrix}$$

```
> W := Determinant(A);                            # Resp. W := -30 e^(2x) e^(-x) e^(-3x)
> simplify(%);                                    # Resp. -30 e^(-2x)
```

Because this is not zero, the three solutions form a basis of solutions for the given ODE on any interval.

Similar Material in AEM: Sec. 3.1

EXAMPLE 2.8 NONHOMOGENEOUS LINEAR ODES

Nonhomogeneous linear ODEs can be solved by `dsolve` in almost the same way as homogeneous ODEs. For instance, to solve

$$y'' + 4y' + 260y = 53\,e^x, \qquad y(0) = 1.5, \qquad y'(0) = -0.8,$$

type the equation as before, say,

```
> restart:
> ode := diff(y(x), x, x) + 4*diff(y(x), x) + 260*y(x) = 53*exp(x);
```

$$ode := \frac{d^2}{dx^2}y(x) + 4\left(\frac{d}{dx}y(x)\right) + 260\,y(x) = 53\,e^x$$

Then `dsolve`, applied to the initial value problem, gives

```
> dsolve(ode);
```

$$y(x) = e^{-2x}\sin(16x)\_C2 + e^{-2x}\cos(16x)\_C1 + \frac{1}{5}e^x$$

```
> yp := dsolve(ode, y(0) = 1.5, D(y)(0) = -0.8);
```

$$yp := y(x) = \frac{1}{10}e^{-2x}\sin(16x) + \frac{13}{10}e^{-2x}\cos(16x) + \frac{1}{5}e^x$$

```
> yp := evalf[2](%);
```

$$yp := y(x) = 0.10\,e^{-2.0\,x}\sin(16.\,x) + 1.3\,e^{-2.0\,x}\cos(16.0\,x) + 0.20\,e^x$$

```
> plot(rhs(yp), x = 0..3, xtickmarks = [1, 2, 3], labels = [x, y]);
```



Example 2.8. Particular solution $y = 0.2e^x + e^{-2x}(1.3\cos 16x + 0.1\sin 16x)$

You see that the oscillatory effect of the second term decreases and the solution curve approaches the curve of the exponential function $0.2\,e^x$.

Similar Material in AEM: Sec. 2.7

EXAMPLE 3.9 NONHOMOGENEOUS LINEAR ODES

Again, the second-order case can be extended to higher order. For instance, to solve

$$y''' - 3y'' + 256y' + 260y = 514\,e^x, \qquad y(0) = 100, \qquad y'(0) = 2, \qquad y''(0) = 500,$$

type the equation as before, say,

```
> ode := diff(y(x), x, x, x) -3*diff(y(x), x, x) + 256*diff(y(x), x)
  + 260*y(x) = 514*exp(x);
```

$$ode := \frac{d^3}{dx^3}y(x) - 3\left(\frac{d^2}{dx^2}y(x)\right) + 256\left(\frac{d}{dx}y(x)\right) + 260\,y(x) = 514\,e^x$$

Then `dsolve`, applied to the initial value problem, gives

```
> dsolve(ode);
```

$$y(x) = e^x + \_C1\,e^{-x} + \_C2\,e^{2\,x}\cos(16\,x) + \_C3\,e^{2\,x}\sin(16\,x)$$

```
> yp := dsolve(ode, y(0) = 100, D(y)(0) = 2, (D@@2)(y)(0) = 500);
```

$$yp := y(x) = e^x + 99\,e^{-x} + \frac{25}{4}\,e^{2\,x}\sin(16\,x)$$

```
> yp := evalf[2](%);
```

$$yp := y(x) = e^x + 99.0\,e^{-1.\,x} + 6.2\,e^{2.\,x}\sin(16.\,x)$$

```
> plot(rhs(yp), x = 0..3, xtickmarks = [1, 2, 3], labels = [x, y]);
```



Example 2.9. Particular solution $y = e^x + 99e^{-x} + 6.2e^{2x}\sin 16x$

Here, the sinusoidal behavior takes over because the "amplitude" is $6.2e^{2x}$.

EXAMPLE 2.10 SOLUTION BY UNDETERMINED COEFFICIENTS

Write a given non-homogeneous ODE, say,

$$y'' - 6y' + 8y = e^{2x},$$

in the form $Ly = r$, where Ly (L suggesting 'linear operator') is the left-hand side of the given ODE,

```
> restart:
```

```
> Ly := diff(y(x), x, x) - 6*diff(y(x), x) + 8*y(x);
```

$$Ly := \frac{d^2}{dx^2}y(x) - 6\left(\frac{d}{dx}y(x)\right) + 8\,y(x)$$

and r is the right-hand side,

```
> r := exp(2*x);                                   # Resp. r := e^(2 x)
```

To see whether the modification rule applies, you must first solve the homogeneous equation $Ly = 0$; thus,

```
> yh := dsolve(Ly = 0);
```
Resp. $yh := y(x) = \_C1\, e^{4\,x} + \_C2\, e^{2\,x}$

You see that r is a solution of the homogeneous ODE, so that the modification rule (for a simple root) does apply. That is, instead of $C\,e^{2x}$ you have to use

```
> yp := C*x*exp(2*x);
```
Resp. $yp := Cxe^{2\,x}$

(In the case of a double root you would have to multiply e^{2x} by x^2 instead of x.) The constant C is unknown. You find its value from the given non-homogeneous equation with $y =$ `yp`; that is,

```
> sol := eval(subs(y(x) = yp, Ly = r));
```

$$sol := -2\,Ce^{2\,x} = e^{2\,x}$$

You see that $C = -1/2$. The command for obtaining this (in a more involved case) would be

```
> C0 := solve(sol, C);
```
Resp. $C0 := -\dfrac{1}{2}$

Substituting this into `yp` and adding `yh`, you obtain a general solution of the given ODE. Perhaps you do this step by step.

```
> rhs(yh);
```
Resp. $\_C1\, e^{4\,x} + \_C2\, e^{2\,x}$

```
> yp;
```
Resp. $Cxe^{2\,x}$

```
> yp2 := subs(C = C0, yp);
```
Resp. $yp2 := -\dfrac{1}{2}\,xe^{2\,x}$

```
> ygen := rhs(yh) + yp2;
```
Resp. $ygen := \_C1\, e^{4\,x} + \_C2\, e^{2\,x} - \dfrac{1}{2}\,xe^{2\,x}$

This example served to explain the method. Clearly, you can obtain the answer directly by typing

```
> dsolve(Ly = r);
```
Resp. $y(x) = \left(-\dfrac{1}{2}\,x + \dfrac{1}{2}\,\_C1\, e^{2\,x} + \_C2\right) e^{2\,x}$

Similar Material in AEM: Sec. 2.7

EXAMPLE 3.11 SOLUTION BY UNDETERMINED COEFFICIENTS

Solve the non-homogeneous ODE,

$$y''' - y'' - 8y' + 12y = e^{2x},$$

Solution. As before, we write a our ODE in the form $Ly = r$, where Ly is the left-hand side of the given ODE,

```
> restart:
> Ly := diff(y(x), x, x, x)-diff(y(x), x, x) - 8*diff(y(x), x) + 12*y(x);
```

$$Ly := \frac{d^3}{dx^3}y(x) - \left(\frac{d^2}{dx^2}y(x)\right) - 8\left(\frac{d}{dx}y(x)\right) + 12\,y(x)$$

and r is the right-hand side,

```
> r := exp(2*x);
```
Resp. $r := e^{2\,x}$

To see whether the modification rule applies, you must first solve the homogeneous equation $Ly = 0$; thus,

```
> yh := dsolve(Ly = 0, y(x));
```

$$yh := y(x) = \_C1\, e^{-3\,x} + \_C2\, e^{2\,x} + \_C3\, e^{2\,x} x$$

You see that r is a solution of the homogeneous ODE, so that the modification rule (for a double root) does apply. That is, instead of $C\,e^{2x}$ you have to use

```
> yp := C*x^2*exp(2*x);
```

Resp. $yp := Cx^2 e^{2\,x}$

The constant C is unknown. You find its value from the given non-homogeneous equation with $y =$ `yp`; that is,

```
> sol := eval(subs(y(x) = yp, Ly = r));
```

Resp. $sol := 10\,Ce^{2\,x} = e^{2\,x}$

You see that $C = 1/10$. The command for obtaining this (in a more involved case) would be

```
> C0 := solve(sol, C);
```

Resp. $C0 := \frac{1}{10}$

Substituting this into `yp` and adding `yh`, you obtain a general solution of the given ODE. Perhaps you do this step by step.

```
> rhs(yh);
```

Resp. $\_C1\, e^{-3\,x} + \_C2\, e^{2\,x} + \_C3\, e^{2\,x} x$

```
> yp;
```

Resp. $Cx^2 e^{2\,x}$

```
> yp2 := subs(C = C0, yp);
```

Resp. $yp2 := \frac{1}{10}\, x^2 e^{2\,x}$

```
> ygen := rhs(yh) + yp2;
```

$$ygen := \_C1\, e^{-3\,x} + \_C2\, e^{2\,x} + \_C3\, e^{2\,x} x + \frac{1}{10}\, x^2 e^{2\,x}$$

As before, this example served to explain the method and you can obtain the answer directly by typing

```
> dsolve(Ly = r);
```

$$y(x) = \_C1\, e^{-3\,x} + \_C2\, e^{2\,x} + \_C3\, e^{2\,x} x + \frac{1}{10}\, x^2 e^{2\,x}$$

Similar Material in AEM: Sec. 3.3

EXAMPLE 2.12 SOLUTION BY VARIATION OF PARAMETERS

The method of undetermined coefficients only works in certain cases. (See AEM Sec. 2.10, p. 99 for details.) To solve

$$y'' + 4y = \csc\, 2x$$

by variation of parameters, write it as $Ly = r$, typing

```
> Ly := diff(y(x), x, x) + 4*y(x);
```

$$Ly := \frac{d^2}{dx^2} y(x) + 4\, y(x)$$

```
> r := csc(2*x);
```

Resp. $r := \csc(2\,x)$

Obtain a general solution of the homogeneous ODE by `dsolve`,

```
> yh := dsolve(Ly = 0);
```

$$yh := y(x) = \_C1\, \sin(2\,x) + \_C2\, \cos(2\,x)$$

Hence a basis of solutions of the homogeneous ODE, as needed in the present method, is

```
> y1 := rhs(subs(_C1 = 1, _C2 = 0, yh));
```
Resp. $y1 := \sin(2x)$

```
> y2 := rhs(subs(_C1 = 0, _C2 = 1, yh));
```
Resp. $y2 := \cos(2x)$

For this basis the Wronskian is (see Example 2.6 in this Guide, if necessary)

```
> W := y1*diff(y2, x) - y2*diff(y1, x);
```

$$W := -2\sin(2x)^2 - 2\cos(2x)^2$$

```
> simplify(W);
```
Resp. -2

The formula for a particular solution is

$$y_p = -y_1 \int \frac{y_2 r}{W}\,dx + y_2 \int \frac{y_1 r}{W}\,dx.$$

For the present ODE you obtain from it

```
> yp := -y1*int(y2*r/W, x) + y2*int(y1*r/W, x);
```

$$yp := \frac{1}{4}\sin(2x)\ln(\sin(2x)) - \frac{1}{2}\cos(2x)\,x$$

Hence a general solution of the given non-homogeneous equation is

```
> ygen := rhs(yh) + yp;
```

$$ygen := \_C1\,\sin(2x) + \_C2\,\cos(2x) + \frac{1}{4}\sin(2x)\ln(\sin(2x)) - \frac{1}{2}\cos(2x)\,x$$

This example served to explain the method. Clearly, you can obtain the solution directly by typing

```
> dsolve(Ly = r);
```

$$y(x) = \sin(2x)\,\_C2 + \cos(2x)\,\_C1 + \frac{1}{4}\sin(2x)\ln(\sin(2x)) - \frac{1}{2}\cos(2x)\,x$$

Similar Material in AEM: Sec. 2.10

EXAMPLE 3.13 SOLUTION BY VARIATION OF PARAMETERS

Solve

$$x^3y''' - 3x^2y'' + 6xy' - 6y = x^4\ln x$$

by variation of parameters.

Solution. After a bit of housekeeping, write the ODE as $Ly = r$,

```
> with(LinearAlgebra):  with(VectorCalculus):
```

```
> Ly := x^3*diff(y(x), x, x, x) - 3*x^2*diff(y(x), x, x)
   + 6*x*diff(y(x), x) - 6*y(x);
```

$$Ly := x^3\left(\frac{d^3}{dx^3}y(x)\right) - 3x^2\left(\frac{d^2}{dx^2}y(x)\right) + 6x\left(\frac{d}{dx}y(x)\right) - 6\,y(x)$$

```
> r := x^4*ln(x);
```
Resp. $r := x^4\ln(x)$

Obtain a general solution of the homogeneous ODE by `dsolve`,

```
> yh := dsolve(Ly = 0);
```
Resp. $yh := y(x) = \_C3\,x^3 + \_C2\,x^2 + \_C1\,x$

Hence a basis of solutions of the homogeneous ODE, as needed in the present method, is

```
> y1 := rhs(subs(_C1 = 1, _C2 = 0, _C3 = 0, yh));        # Resp. y1 := x
> y2 := rhs(subs(_C1 = 0, _C2 = 1, _C3 = 0, yh));        # Resp. y2 := x^2
> y3 := rhs(subs(_C1 = 0, _C2 = 0, _C3 = 1, yh));        # Resp. y3 := x^3
```

For this basis the Wronskian is (see Example 3.7 in this Guide, if necessary)

```
> WM := Wronskian([y1, y2, y3], x);
```

$$WM := \begin{bmatrix} x & x^2 & x^3 \\ 1 & 2\,x & 3\,x^2 \\ 0 & 2 & 6\,x \end{bmatrix}$$

We now need to find the determinant of the Wronskian and compute some other determinants. For details about matrices see the material in Chap. 7.

```
> W := Determinant(WM);                                  # Resp. W := -2 x^3
```

For the other determinants, we make use of

```
> v1 := Vector([0, 0, 1]);                               # Resp. v1 := e_z
> WM1 := Copy(WM): WM2 := Copy(WM): WM3 := Copy(WM):
> WM1[1..3, 1] := v1:  WM2[1..3, 2] := v1:  WM3[1..3, 3] := v1:
> WM2;
```

$$\begin{bmatrix} x & 0 & x^3 \\ 1 & 0 & 3\,x^2 \\ 0 & 1 & 6\,x \end{bmatrix}$$

```
> W1 := Determinant(WM1); W2 := Determinant(WM2); W3 := Determinant(WM3);
```

$$W1 := x^4$$
$$W2 := -2x^3$$
$$W3 := x^2$$

The formula for a particular solution is

$$y_p = -y_1 \int \frac{W_1\,r}{W}\,dx + y_2 \int \frac{W_2 r}{W}\,dx + y_3 \int \frac{W_3 r}{W}\,dx.$$

It is important to remember that the method of variation of parameters requires the ODE to be in standard form. Hence we must divide all terms by x^3. For this ODE you have

```
> yp := y1*int(W1*(r/x^3)/W, x) + y2*int(W2*(r/x^3)/W, x)
        + y3*int(W3*(r/x^3)/W, x);
```

$$yp := x\left(\frac{1}{6}\,x^3\ln(x) - \frac{1}{18}\,x^3\right) + x^2\left(-\frac{1}{2}\,x^2\ln(x) + \frac{1}{4}\,x^2\right) + x^3\left(\frac{1}{2}\,x\ln(x) - \frac{1}{2}\,x\right)$$

Hence a general solution of the given non-homogeneous equation is

```
> ygen := rhs(yh) + yp;
```

$$ygen := \_C3\,x^3 + \_C2\,x^2 + \_C1\,x + x\left(\frac{1}{6}\,x^3\ln(x) - \frac{1}{18}\,x^3\right) + x^2\left(-\frac{1}{2}\,x^2\ln(x) + \frac{1}{4}\,x^2\right) + x^3\left(\frac{1}{2}\,x\ln(x) - \frac{1}{2}\,x\right)$$

```
> expand(ygen);
```
Resp. $\_C3\,x^3 + \_C2\,x^2 + \_C1\,x + \frac{1}{6}\,x^4\ln(x) - \frac{11}{36}\,x^4$

This example served to explain the method. Clearly, you can obtain the solution directly by typing

```
> dsolve(Ly = r);
```
$$y(x) = \_C3\,x^3 + \_C2\,x^2 + \_C1\,x + \frac{1}{6}\,x^4\ln(x) - \frac{11}{36}\,x^4$$

Similar Material in AEM: Sec. 3.3

EXAMPLE 2.14 FORCED VIBRATIONS. RESONANCE. BEATS

Resonance occurs in an **undamped vibrating system** if the frequency of the driving force equals the natural frequency of the free vibrations of the system. For instance, let

$$y'' + 4y = \sin 2t.$$

Type this ODE and solve it.

Solution.

```
> ode := diff(y(t), t, t) + 4*y(t) = sin(2*t);
```
$$ode := \frac{d^2}{dt^2}y(t) + 4\,y(t) = \sin(2\,t)$$

```
> sol := dsolve(ode);
```
$$sol := y(t) = \sin(2\,t)\,\_C2 + \cos(2\,t)\,\_C1 - \frac{1}{4}\cos(2\,t)\,t$$

The last term on the right will be growing without bound as time t increases to infinity. You can single it out by choosing $\_C1 = 0$ and $\_C2 = 0$,

```
> y1 := subs(_C1 = 0, _C2 = 0, rhs(sol));
```
Resp. $y1 := -\frac{1}{4}\cos(2\,t)\,t$

```
> plot(y1, t = 0..50, title = `Resonance`);
```

Beats occur if the frequency of the driving force of the undamped system is close to the natural frequency of the free vibrations. For example, consider the equation $y'' + 4y = 0.39\cos 1.9t$. (The factor 0.39 has been included to have a simpler final answer; it is of no other importance.) Type and solve this ODE.

Solution.

```
> ode2 := diff(y2(t), t, t) + 4*y2(t) = 0.39*cos(1.9*t);
```
$$ode2 := \frac{d^2}{dt^2}y2(t) + 4\,y2(t) = 0.39\,\cos(1.9\,t)$$

```
> sol2 := dsolve(ode2);
```
$$sol2 := y2(t) = \sin(2\,t)\,\_C2 + \cos(2\,t)\,\_C1 + \cos\left(\frac{19}{10}\,t\right)$$

```
> sol3 := subs(_C1 = -1, _C2 = 0, sol2);
```
$$sol3 := y2(t) = -\cos(2\,t) + \cos\left(\frac{19}{10}\,t\right)$$

To comprehend this form of the solution ("**beats**"), note that the difference of two cosine functions can be written as the product of two sine functions, one with a high frequency (this gives the rapid oscillations shown in the figure) and one with a low frequency $(0.05/(2\pi)$, giving a period of about 126). The formula is obtained by

```
> 2*combine(sin(1.95*t)*sin(0.05*t));        # Resp.
```
$\cos(1.90\,t) - \cos(2.00\,t)$

```
> plot(rhs(sol3), t = 0..253, title = 'Beats');
```



Example 2.14. Resonance



Example 2.14. Beats given by $y = 2\sin 1.95t \sin 0.05t$

Similar Material in AEM: Sec. 2.8

EXAMPLE 2.15 ***RLC*-CIRCUIT**

The current $i(t)$ in an RLC-circuit with sinusoidal electromotive force is obtained by solving

$$Li' + Ri + \frac{1}{C}\int i\,dt = E_0 \sin \omega\, t$$

where $Q = \int i\,dt$ is the charge on the capacitor. Differentiation gives the ODE

$$Li'' + Ri' + \frac{1}{C}i = \omega\, E_0 \cos \omega\, t.$$

(Write i because Maple uses I for $\sqrt{-1}$.) For instance, let $L = 0.8$ henry, $R = 50$ ohms, $C = 0.001$ farad, $E_0 = 245$, and $\omega = 377$ (thus 60 hertz). Solve the ODE for these data.

Solution. Write the ODE as $Mi = r$, where Mi denotes the left-hand side. (Note that L here is used for the inductance.)

```
> L := 0.8:  R := 50:  C := 0.001:  E0 := 245:  omega := 377:
  Mi := L*diff(i(t), t, t) + R*diff(i(t), t) + 1/C*i(t);
```

$$Mi := 0.8\left(\frac{d^2}{dt^2}i(t)\right) + 50\left(\frac{d}{dt}i(t)\right) + 1000\,i(t)$$

```
> r := omega*E0*cos(omega*t);                # Resp.
```
$r := 92365\cos(377\,t)$

A general solution of the homogeneous ODE is

```
> igen := evalf[5](dsolve(Mi = 0));
```

$$igen := i(t) = \_C1\, e^{-31.250\,t} \sin(16.536\,t) + \_C2\, e^{-31.250\,t}\cos(16.536\,t)$$

A general solution of the non-homogeneous ODE is

```
> igen := evalf[5](dsolve(Mi = r));
```

$$igen := i(t) = \mathrm{e}^{-31.250\,t} \sin(16.536\,t)\,\_C2 + \mathrm{e}^{-31.250\,t} \cos(16.536\,t)\,\_C1 + 0.13334 \sin(377.0\,t) - 0.79724 \cos(377.0\,t)$$

The particular solution satisfying $i(0) = 0$, $i\,'(0) = 0$ is (see the figure)

```
> ipart := evalf[5](dsolve(Mi = r, i(0) = 0, D(i)(0) = 0));
```

$$ipart := i(t) = -1.5334\,\mathrm{e}^{-31.250\,t} \sin(16.536\,t) + 0.79724\,\mathrm{e}^{-31.250\,t} \cos(16.536\,t) + 0.13334 \sin(377.0\,t) - 0.79724 \cos(377.0\,t)$$

```
> plot(rhs(ipart), t = 0..0.6);
```



Example 2.15. Current in the *RLC*-circuit

The oscillation is very rapid. The **transient solution** becomes practically zero after a short time, so that a transient period appears only for $t < 0.2$ in the figure. For $t = 0.2$, $e^{-31.250t} = 0.0019$ which is very small when compared with an amplitude of about 0.8. $i(0) = 0$ means no current at $t = 0$. Assume that the capacitor is uncharged at $t = 0$; thus $Q(0) = 0$ in the first formula in this example. Solve that formula algebraically for i', obtaining

$$i\,' = \frac{1}{L}(E_0 \sin \omega\, t - R\, i - \frac{Q}{C}).$$

For $t = 0$ all three terms on the right are 0. This shows that the initial condition $i'(0) = 0$ means that the capacitor is initially uncharged.

Similar Material in AEM: Sec. 2.9

Problem Set for Chapter 2/3

Pr.2.1 (General solution, initial value problem, harmonic oscillations) Find a general solution of $y'' - 2y' - 3y = 0$. Find and plot the particular solution satisfying the initial conditions $y(0) = 6$, $y'(0) = -4$. (*AEM* Sec. 2.2)

Pr.2.2 (Maximum of solution) Find and plot the solution y_p of the initial value problem $y'' - 6y' + y = 0$, $y(0) = 4$, $y'(0) = 8$. Find the location and the value of the maximum of y_p, first from the graph and then by calculation. (*AEM* Sec. 2.2)

Pr.3.3 (General solution, initial value problem) Find a general solution of $y''' + 3y'' + 3y' + y = 0$. Find and plot the particular solution satisfying the initial conditions $y(0) = 4$, $y'(0) = 2$, and $y''(0) = 4$. (*AEM* Sec. 3.1)

Pr.3.4 (Linear independence, Wronskian) Show that the solution in Pr.2.1 is a general solution. (*AEM* Sec. 2.6)

Pr.3.5 (Experiment on damped oscillations, dependence on initial conditions) Experiment with Example 2.3 in this Guide by systematically changing the initial conditions (e.g., by keeping $y'(0)$ constant and varying $y(0)$, etc.) in order to find out whether and how solutions depend on initial conditions. (*AEM* Sec. 2.4)

Pr.2.6 (Nonhomogeneous equation, complex roots) Find a general solution of

$$y'' + 2y' + 145y = e^{-0.05\,t}.$$

Find and plot the particular solution which starts at $y = 0$ with initial velocity 0. (*AEM* Sec. 2.7)

Pr.3.7 (Verification of solution) Show that a linear combination of e^x, $x\,e^x$, $x^2\,e^x$ is a solution of $y''' - 3y'' + 3y' - y = 0$. (*AEM* Sec. 3.2)

Pr.2.8 (Experiment on overdamping) Find a general solution of $y'' + 3y' + 2y = 0$. Find and plot (on common axes) the three particular solutions starting from $y = 1$ with different initial velocities, namely, -1, 0, 1. Can you find an initial velocity such that the particular solution becomes 0 for some positive t? For two different positive t? What is the effect of changing the initial displacement? (*AEM* Sec. 2.4)

Pr.2.9 (Critical damping) Find the value of the damping constant c such that the mass-spring system

$$y'' + cy' + 27.52y = 0$$

is critically damped. Characterize the other two cases (underdamping, overdamping) in terms of the values of c. (*AEM* Sec. 2.4)

Pr.2.10 (Logarithmic decrement) Prove, on the computer, that the ratio of two consecutive maximum amplitudes of a damped oscillation $y = C\exp(-at)\cos(wt-d)$ is constant, the natural logarithm of this ratio being called the *logarithmic decrement* and being given by $LD = 2\pi a/w$. (C, a, w, d are constant.) (*AEM* Sec. 2.4)

Pr.2.11 (Pendulum) Find the motion of a pendulum of mass m and length 0.994 meter for small angle $|T|$ (so that sin T can be approximated by T rather accurately). Assume that the pendulum starts from the equilibrium position ($T = 0$) with velocity 2. (*AEM* Sec. 2.4)

Pr.2.12 (Boundary value problem) Solve $y'' + 9y = 0$, $y(0) = 2$, $y(\pi) = -2$.

Pr.2.13 (Euler-Cauchy equation) Determine a and b in the Euler-Cauchy equation $x^2y'' + axy' + by = 0$ so that the auxiliary equation has a double root and determine a general solution for this "critical case". (*AEM* Sec. 2.5)

Pr.2.14 (Euler-Cauchy equation) Find the Euler-Cauchy equation with $x^2 \cos(\ln x)$ and $x^2 \sin(\ln x)$ as a basis of solutions. (*AEM* Sec. 2.5)

Pr.2.15 (Resonance) Find and plot the solution of $y''+81y = -123 \sin 2t$, $y(0) = 10$, $y'(0) = 4$. (*AEM* Sec. 2.8)

Pr.3.16 (Wronskian) Show that $\{e^{3x}, e^{-5x}, e^{6x}\}$ is a basis of solutions of the ODE $y''' - 4y'' - 27y' + 90y = 0$. (*AEM* Sec. 3.2)

Pr.3.17 (Beats) Find and plot the solution of the initial value problem $y'' + 144y = 52 \cos 14t$, $y(0) = 0$, $y'(0) = 0$. Can you see by looking at the ODE, that you will obtain beats, because the frequency of the driving force is close to that of the free harmonic oscillations? (*AEM* Sec. 2.8)

Pr.3.18 (Nonhomogeneous ODE) Show that $-e^{-3x} \ln x$ is a particular solution of

$$y'' + 6y' + 9y = e^{-3x}/x^2.$$

Find a general solution of this ODE and determine its arbitrary constants so that you obtain a particular solution satisfying the initial conditions $y(1) = 1/e^3$, $y'(1) = -2/e^3$. (*AEM* Sec. 2.7)

Pr.3.19 (Undetermined coefficients) Find a particular solution of $y''+3y'-28y = 44 \sinh 4x$ by the method of undetermined coefficients. Check your result by `dsolve.` (*AEM* Sec. 2.7)

Pr.3.20 (Variation of parameters) Find a particular solution of the ODE

$$y'' - 6y' + 9y = 3x^{3/2}\, e^{3x}$$

by the method of variation of parameters. (*AEM* Sec. 2.10)

Pr.3.21 (*RLC*-circuit) Find the current in the RLC-circuit with $R = 16$ ohms, $L = 16$ henrys, $C = 1/4$ farad, and $E = 260 \cos 4t$ volts. Also find and plot the particular solution satisfying $i(0) = 0$, $i'(0) = 0$. (*AEM* Sec. 2.9)