PART C. FOURIER ANALYSIS AND PARTIAL DIFFERENTIAL EQUATIONS (PDEs)

Content. Fourier series, integrals, and transforms (Chap. 11)
Partial differential equations (PDEs) (Chap. 12)

This Part concerns initial and boundary value problems for the "big" PDEs of Applied Mathematics, (the wave, heat, and Laplace equations) their solution by separating variables, and the use of Fourier series and integrals for obtaining solutions sufficiently general to satisfy all the given physical conditions.

Chapter 11

Fourier Series, Integrals, and Transforms

Content. Fourier series (period 2π) (Exs. 11.1, 11.4, Prs. 11.1–11.5)
Fourier series (any period) (Exs. 11.2, 11.3, Prs. 11.6–11.10, 11.15)
Half-range expansions (Ex. 11.3, Pr. 11.9)
Trigonometric approximation (Ex. 11.5, Prs. 11.11–11.13)
Orthogonality. Fourier–Legendre series (Ex. 11.6, Prs. 11.16–11.1)
Fourier integral (Ex. 11.7, Pr. 11.14)

The **Fourier series** of a function $f(x)$ of period $p = 2L$ is obtained by typing

```
> f := a[0] + sum(a[n]*cos(n*Pi*x/L) + b[n]*sin(n*Pi*x/L), n = 1..infinity);
```

$$f := a_0 + \sum_{n=1}^{\infty}\left(a_n \cos\left(\frac{n\pi\, x}{L}\right) + b_n \sin\left(\frac{n\pi\, x}{L}\right)\right)$$

with the **Fourier coefficients** a_n and b_n given by the **Euler formulas**

```
> f := 'f':                                              # Unassign f
```

```
> a[0] := 1/(2*L)*int(f(x), x = -L..L);        # Resp.
```

$$a_0 := \frac{1}{2}\,\frac{\int_{-L}^{L} f(x)\,dx}{L}$$

```
> a[n] := 1/L*int(f(x)*cos(n*Pi*x/L), x = -L..L);
```

$$a_n := \frac{\int_{-L}^{L} f(x)\cos\left(\frac{n\pi\, x}{L}\right)dx}{L}$$

```
> b[n] := 1/L*int(f(x)*sin(n*Pi*x/L), x = -L..L);
```

$$b_n := \frac{\int_{-L}^{L} f(x) \sin\left(\frac{n\pi\, x}{L}\right) dx}{L}$$

Functions of period $p = 2\pi$. Replace L by π, obtaining $\cos nx$, $\sin nx$ in all formulas, and $1/(2\pi)$ and $1/\pi$ as factors before the integrals, and integrate from $-\pi$ to π. Write down these formulas for yourself.

Examples for Chapter 11

EXAMPLE 11.1 FUNCTIONS OF PERIOD 2π. EVEN FUNCTIONS. GIBBS PHENOMENON

Consider the function $f(x)$, of period 2π, which is 0 for $-\pi < x < -\pi/2$, then equals k for $-\pi/2 < x < \pi/2$ and is again 0 for $\pi/2 < x < \pi$ (see Fig. 260 in AEM, with $k = 1$). You may call this function a **periodic rectangular wave** of period 2π. You obtain the Fourier coefficients from the Euler formulas by integrating from $-\pi/2$ to $\pi/2$ only because $f(x)$ is 0 for $-\pi < x < -\pi/2$ and $\pi/2 < x < \pi$. Hence type

```
> a0 := 1/(2*Pi)*int(k, x = -Pi/2..Pi/2);                # Resp.
```
$a0 := \frac{1}{2} k$

Note that this is the mean value of $f(x)$ over the interval from $-\pi$ to π (as it is always the case).

```
> an := 1/Pi*int(k*cos(n*x), x = -Pi/2..Pi/2);
```

$$an := \frac{2 \sin\left(\frac{1}{2} n\pi\right) k}{n\pi}$$

```
> bn := 1/Pi*int(k*sin(n*x), x = -Pi/2..Pi/2);           # Resp.
```
$bn := 0$

Hence this Fourier series is a **Fourier cosine series**, that is, it has only cosine terms, no sine terms. The reason is that $f(x)$ is an **even function**, that is, $f(-x) = f(x)$. The command `seq` gives you the first few coefficients

```
> seq(an, n = 1..8);                                     # Resp.
```
$\frac{2k}{\pi}, 0, -\frac{2}{3}\frac{k}{\pi}, 0, \frac{2}{5}\frac{k}{\pi}, 0, -\frac{2}{7}\frac{k}{\pi}, 0$

For plotting you have to choose a definite value of k, say, $k = 1$,

```
> An := subs(k = 1, an);                                 # Resp.
```
$An := \frac{2 \sin\left(\frac{1}{2} n\pi\right)}{\pi n}$

Find and plot (on common axes) the first few partial sums

```
> S1 := 1/2 + sum(An*cos(n*x), n = 1..1);
```

$$S1 := \frac{1}{2} + \frac{2 \cos(x)}{\pi}$$

```
> S3 := 1/2 + sum(An*cos(n*x), n = 1..3);
```

$$S3 := \frac{1}{2} + \frac{2 \cos(x)}{\pi} - \frac{2}{3} \frac{\cos(3\, x)}{\pi}$$

```
> S5 := 1/2 + sum(An*cos(n*x), n = 1..5);
```

$$S5 := \frac{1}{2} + \frac{2\cos(x)}{\pi} - \frac{2}{3}\frac{\cos(3x)}{\pi} + \frac{2}{5}\frac{\cos(5x)}{\pi}$$

```
> with(plots):
```

```
> P1 := plot(S1, S3, S5, x = -Pi..Pi):
  P2 := plot(1, x = -Pi/2..Pi/2):
  display(P1, P2, ytickmarks = [0, 0.5, 1]);
```



Example 11.1. Given rectangular wave and first three partial sums `S1`, `S3`, `S5`

You see that there are waves near the discontinuities at $-\pi/2$ and $\pi/2$ and you would perhaps expect that these waves become smaller and smaller if you took partial sums with more and more terms – which can easily be done on the computer. However, these waves do not disappear, but they are shifted closer and closer to those points of discontinuity, as is shown in the figure for the partial sum for $n = 1$ to 50. This is called the **Gibbs phenomenon**. Experiment with $n = 100$ or even larger n.

```
> plot(1/2 + sum(An*cos(n*x), n = 1..50), x = -Pi..Pi);
```



Example 11.1. Gibbs phenomenon

Similar Material in AEM: Sec. 11.1

EXAMPLE 11.2 FUNCTIONS OF ARBITRARY PERIOD. ODD FUNCTIONS

For functions $f(x)$ of arbitrary period $p = 2L$, the determination of the Fourier coefficients is the same in principle as for functions of period 2π; the formulas just look a little more complicated.

For instance, consider the function $f(x)$ of period $p = 2L = 2$ in the figure. (Type `?piecewise` for information.)

```
> with(plots):
```

```
> f := piecewise(x < -5, 0, x < -3, 2*x + 8, x < -1, 2*x + 4, x < 1,
  2*x, x < 3, 2*x - 4, x < 5, 2*x - 8, 0);
```

$$f := \begin{cases} 0 & x < -5 \\ 2x+8 & x < -3 \\ 2x+4 & x < -1 \\ 2x & x < 1 \\ 2x-4 & x < 3 \\ 2x-8 & x < 5 \\ 0 & \textit{otherwise} \end{cases}$$

```
> plot(f, x = -7..7, scaling = constrained);
```



Example 11.2. Given function (**Sawtooth wave**)

Clearly, $f(x) = 2x$ if $-1 < x < 1$. For this function you obtain from the Euler formulas with $L = 1$ in the chapter opening

```
> L := 1:  f := 2*x:
```

```
> a0 := 1/(2*L)*int(f, x = -1..1);                    # Resp. a0 := 0
```

```
> an := (1/L)*int(f*cos(n*Pi*x), x = -1..1);          # Resp. an := 0
```

Hence the series has no cosine terms, only sine terms; it is a **Fourier sine series**, because $f(x)$ is an **odd function,** that is, $f(-x) = -f(x)$. The sine terms have the Fourier coefficients

```
> bn := (1/L)*int(f*sin(n*Pi*x), x = -1..1);
```

$$bn := -\frac{4\left(\cos(n\pi)\, n\pi - \sin(n\pi)\right)}{n^2\pi^2}$$

```
> seq(bn, n = 1..8);
```

`# Resp.` $\frac{4}{\pi}, -\frac{2}{\pi}, \frac{4}{3\pi}, -\frac{1}{\pi}, \frac{4}{5\pi}, -\frac{2}{3\pi}, \frac{4}{7\pi}, -\frac{1}{2\pi}$

Now type and then plot some partial sums, including a large one (S50), which will again illustrate the **Gibbs phenomenon** at the points of discontinuity -1 and 1, just as in the preceding example.

```
> S1 := sum(bn*sin(n*Pi*x), n = 1..1):
  S2 := sum(bn*sin(n*Pi*x), n = 1..2):
  S3 := sum(bn*sin(n*Pi*x), n = 1..3):
  S50 := sum(bn*sin(n*Pi*x), n = 1..50):
  plot(S1, S2, S3, S50, x = -1..1, xtickmarks = [-1, -0.5, 0, 0.5, 1],
   ytickmarks = [-2, -1, 0, 1, 2]);
```



Example 11.2. Approximation of $f(x) = 2x$ by partial sums of its Fourier series

EXAMPLE 11.3 HALF-RANGE EXPANSIONS

If you develop a function $f(x)$, given for $0 < x < L$, in a Fourier series, then this series represents a periodic function of period L which for $x = 0 \ldots L$ agrees with $f(x)$. In general, this series contains both cosine and sine terms. In some applications it will be much better to first extend $f(x)$ as an **even** function $f_1(x)$ for $-L < x < L$; then $f_1(x)$ has the period $2L$, and its Fourier series contains only cosine terms because $f_1(x)$ is even. This series is called the **cosine half-range expansion** of $f(x)$ because $f(x)$ is given only over one half of the range (that is, over one half of the interval of periodicity).

Similarly, you may extend the given function $f(x)$ to an **odd** function $f_2(x)$ for $-L < x < L$; then $f_2(x)$ has the period $2L$, and its Fourier series contains only sine terms because $f_2(x)$ is odd. Call this series the **sine half-range expansion** of $f(x)$.

For instance, let $f(x) = x^2$ be given for $0 < x < 2$. Figure (A) shows its even and odd extensions to the interval (the "full range") $-2 < x < 2$. Figure (B) shows its even periodic extension $f_1(x)$ of period $p = 2L = 4$. Figure (C) shows its odd periodic extension $f_2(x)$ of period 4. `scaling = constrained` gives equal scales on the axes. You need a different form for expressing a function so that you can evaluate expressions such as `f(x - 2)`.

```
> with(plots):  f := x -> x^2:
```

```
> P := plot(f(x), x = 0..2, thickness = 2):
```

```
> PE := plot(f(x), x = -2..0, thickness = 2):
```

```
> PO := plot(-f(x), x = -2..0, linestyle = dash):
```

```
> display(P, PE, PO, scaling = constrained);
```



***Example 11.3.*A** $f(x) = x^2$ (given for $0 < x < 2$) and its even and odd extensions to $-2 < x < 2$

```
> PE1 := plot(f(x + 4), x = -6..-4, thickness = 2):
> P1 := plot(f(x + 4), x = -4..-2):
> PE2 := plot(f(x), x = -2..0, thickness = 2):
> P2 := plot(f(x), x = 0..2):
> PE3 := plot(f(x - 4), x = 2..4, thickness = 2):
> P3 := plot(f(x - 4), x = 4..6):
> PE4 := plot(f(x - 8), x = 6..8, thickness = 2):
> P4 := plot(f(x - 8), x = 8..10):
```

To draw figure (B),

```
> display(PE1, P1, PE2, P2, PE3, P3, PE4, P4, scaling = constrained);
> PO1 := plot(-f(x + 4), x = -6..-4, linestyle = dash):
> P1 := plot(f(x + 4), x = -4..-2):
> PO2 := plot(-f(x), x = -2..0, linestyle = dash):
> P2 := plot(f(x), x = 0..2):
> PO3 := plot(-f(x - 4), x = 2..4, linestyle = dash):
> P3 := plot(f(x - 4), x = 4..6):
> PO4 := plot(-f(x - 8), x = 6..8, linestyle = dash):
> P4 := plot(f(x - 8), x = 8..10):
```

To draw figure (C),

```
> display(PO1, P1, PO2, P2, PO3, P3, PO4, P4, scaling = constrained);
```



***Example 11.3.*B** Even periodic extension $f_1(x)$ of $f(x)$



***Example 11.3.*C** Odd periodic extension $f_2(x)$ of $f(x)$

Now determine the Fourier coefficients a_n of the Fourier cosine series of $f_1(x)$ from the Euler formula. Because $f_1(x)$ and the cosine are both even, so is the product (the integrand). Hence the integral from -2 to 2 equals twice the integral from 0 to 2 (the interval on which $f(x)$ is given). Remembering that $L = 2$, type the commands for a_0, a_n, and for a partial sum S. Figure (D) shows that a good approximation can already be obtained with a small number of terms.

```
> L := 2:
```

```
> a0 := 2/(2*L)*int(f(x), x = 0..2);                    # Resp. 4/3
```

```
> an := 2/(L)*int(f(x)*cos(n*Pi*x/2), x = 0..2);
```

$$an := \frac{8\left(2\cos(n\pi)\,n\pi + n^2\pi^2\sin(n\pi) - 2\sin(n\pi)\right)}{n^3\pi^3}$$

```
> S := a0 + sum(an*cos(n*Pi*x/2), n = 1..2);
```

$$S := \frac{4}{3} - \frac{16\cos\left(\frac{1}{2}\pi x\right)}{\pi^2} + \frac{4\cos(\pi x)}{\pi^2}$$

```
> P11 := plot(S, x = -6..6, scaling = constrained):
```

```
> display(P11, PE1, P1, PE2, P2, PE3, P3);
```



***Example 11.3.*D** Approximation of $f_1(x)$ by a partial sum of its Fourier series

For $f_2(x)$ you will need a partial sum consisting of more terms, as Fig. (E) will show, because $f_2(x)$ is discontinuous – and it will be quite interesting to see how a sum of ***continuous*** terms approximates a ***discontinuous*** function. Both $f_2(x)$ and the sine in the Euler formula for b_n are odd, hence the integrand is even, so that the integral from -2 to 2 equals twice the integral from 0 to 2. Hence type

```
> bn := 2/(L)*int(f(x)*sin(n*Pi*x/2), x = 0..2);
```

$$bn := -\frac{8\left(n^2\pi^2\cos(n\pi) - 2\sin(n\pi)n\pi - 2\cos(n\pi) + 2\right)}{n^3\pi^3}$$

```
> S2 := sum(bn*sin(n*Pi*x/2), n = 1..7);
```

$$S2 := -\frac{8\left(-\pi^2+4\right)\sin\left(\frac{1}{2}\pi x\right)}{\pi^3} - \frac{4\sin(\pi x)}{\pi} - \frac{8}{27}\,\frac{\left(-9\pi^2+4\right)\sin\left(\frac{3}{2}\pi x\right)}{\pi^3} - 2\,\frac{\sin(2\pi x)}{\pi}$$
$$- \frac{8}{125}\,\frac{\left(-25\pi^2+4\right)\sin\left(\frac{5}{2}\pi x\right)}{\pi^3} - \frac{4}{3}\,\frac{\sin(3\pi x)}{\pi} - \frac{8}{343}\,\frac{\left(-49\pi^2+4\right)\sin\left(\frac{7}{2}\pi x\right)}{\pi^3}$$

```
> P12 := plot(S2, x = -6..6):
```

```
> display(P12, P01, P1, P02, P2, P03, P3, scaling = constrained);
```



***Example 11.3.*E** Approximation of $f_2(x)$ by a partial sum of its Fourier series

Similar Material in AEM: Sec. 11.3

EXAMPLE 11.4 RECTIFIER

A sinusoidal current, for simplicity, $f(t) = \sin t$, where t is time, is passed through a full-wave rectifier which converts the negative half-waves to positive half-waves (and leaves the positive half-waves as they are), resulting in the function $g(t) = |\sin t|$. Find the Fourier series of $g(t)$.

Solution. $g(t)$ is even. Hence you obtain a Fourier cosine series, and the integral for a_n from $t = -\pi$ to π equals twice the integral from 0 to π, an interval in which $g(t) = \sin t$. Using the Euler formulas for a function of period 2π, thus type

```
> f := sin(t):
```

```
> a0 := 1/Pi*int(f, t = 0..Pi);                    # Resp. a0 := 2/π
```

```
> an := 2/Pi*int(f*cos(n*t), t = 0..Pi);
```

$$an := -\frac{2(\cos(n\pi) + 1)}{\pi\,(n^2 - 1)}$$

For $n = 1$ the denominator is zero. Hence show a_1 separately and then some partial sums.

```
> a1 := 2/Pi*int(sin(t)*cos(t), t = 0..Pi);        # Resp. a1 := 0
```

```
> S3 := a0 + a1*cos(t) + sum(an*cos(n*t), n = 2..3):
```

```
> S7 := a0 + a1*cos(t) + sum(an*cos(n*t), n = 2..7);
```

$$S7 := \frac{2}{\pi} - \frac{4}{3}\frac{\cos(2t)}{\pi} - \frac{4}{15}\frac{\cos(4t)}{\pi} - \frac{4}{35}\frac{\cos(6t)}{\pi}$$

S7 is already fairly accurate and can hardly be distinguished from $g(t)$ in the figure.

```
> with(plots):
```

```
> P := plot(S3, S7, t = -3*Pi..3*Pi, linestyle = [dot, dash]):
```

```
> Pg := plot(abs(f), t = -3*Pi..3*Pi):
```

```
> display(P, Pg);
```



Example 11.4. Full-wave rectification $g(t)$ of $f(t) = |\sin t|$ and partial sums S_3 (dots) and S_7 (dashes) of its Fourier series

Similar Material in AEM: Sec. 11.2

EXAMPLE 11.5 TRIGONOMETRIC APPROXIMATION. MINIMUM SQUARE ERROR

In approximating a given function f by another function, F the quality of the approximation can be measured in various ways, depending on the purpose. A quantity that characterizes the *overall goodness of fit* on the interval of interest, say, $-\pi < x < \pi$, is the so-called **square error** of F (relative to f) defined by

$$E = \int_{-\pi}^{\pi} (f - F)^2 \, dx.$$

If

$$F(x) = A_0 + \sum_{n=1}^{N} (A_n \cos n\,x + B_n \sin n\,x),$$

then $F(x)$ is called a **trigonometric polynomial** and the approximation is called a **trigonometric approximation**. In this case the square error of F relative to f (with fixed N) is minimum if and only if you choose the Fourier coefficients of f as the coefficients of F. It can be shown (see AEM, Sec. 11.4) that this **minimum square error** is

$$E^* = \int_{-\pi}^{\pi} f^2 \, dx - \pi \left[2\,a_0^2 + \sum_{n=1}^{N} (a_n^2 + b_n^2) \right].$$

For instance, find the minimum square errors for the approximation of the given function in Example 11.1 in this Guide with $k = 1$ by the partial sums of its Fourier series (illustrated by the figures in that example).

Solution. The b_n are zero because f is even. Type the formula for the a_n, then the formula for the minimum square error, call it `Emin`, and finally find numeric values of `Emin` for $N = 10, 20, ..., 250$. (Because f is discontinuous, the convergence of the sequence of the minimum square errors to 0 will be slow.)

```
> f := 1:
```

```
> an := 1/Pi*int(f*cos(n*x), x = -Pi/2..Pi/2);
```

$$an := \frac{2\sin\left(\frac{1}{2}n\pi\right)}{n\pi}$$

```
> Emin := int(f, x = -Pi/2..Pi/2) - Pi*(2*(1/2)^2 + evalf(sum(an^2,
  n = 1..10*N))):
```

```
> seq(Emin, N = 1..25);
```

0.063430340, 0.031777326, 0.021184281, 0.015883058, 0.012701236, 0.010579667, 0.009064117, 0.007927385, 0.007043227, 0.006335884, 0.005757134, 0.005274837, 0.004866735, 0.004516929, 0.004213763, 0.003948491, 0.003714426, 0.003506368, 0.003320210, 0.003152668, 0.003001080, 0.002863276, 0.002737450, 0.002622113, 0.002516001

Similar Material in AEM: Sec. 11.4

EXAMPLE 11.6 ORTHOGONALITY. FOURIER–LEGENDRE SERIES

The importance of the Legendre polynomials results in part from their orthogonality. They are orthogonal on the interval from -1 to 1. By definition, this means that the integral of the product of two Legendre polynomials of different orders, integrated from -1 to 1, is zero,

$$\int_{-1}^{1} P(m,\,x)\,P(n,\,x)\,dx = 0 \qquad (m \neq n).$$

It can be shown that, for $n = m$, the integral equals $2/(2m+1)$, as is stated on p. 212 of AEM (without the tricky proof). For specific cases, e.g. for all m and n up to 100 (or more), you can verify these statements, using the following command (which does it for 0, 1, ..., 10).

```
> with(orthopoly):
```

```
> seq(seq(int(P(m, x)*P(n, x), x = -1..1), m = 0..n), n = 0..10);
```

$2,\, 0,\, \frac{2}{3},\, 0,\, 0,\, \frac{2}{5},\, 0,\, 0,\, 0,\, \frac{2}{7},\, 0,\, 0,\, 0,\, 0,\, \frac{2}{9},\, 0,\, 0,\, 0,\, 0,\, 0,\, \frac{2}{11},\, 0,\, 0,\, 0,\, 0,\, 0,\, 0,\, \frac{2}{13},\, 0,$
$0,\, 0,\, 0,\, 0,\, 0,\, 0,\, \frac{2}{15},\, 0,\, 0,\, 0,\, 0,\, 0,\, 0,\, 0,\, 0,\, \frac{2}{17},\, 0,\, 0,\, 0,\, 0,\, 0,\, 0,\, 0,\, 0,\, 0,\, \frac{2}{19},\, 0,\, 0,\, 0,$
$0,\, 0,\, 0,\, 0,\, 0,\, 0,\, 0,\, \frac{2}{21}$

You see the values $2/(2m+1) = 2,\ 2/3,\ 2/5$, ... for $m = n = 0, 1, 2, \ldots$ as well as the zeros for the index pairs (0,1), (0,2), (1,2), (0,3), (1,3), (2,3), and so on.

Orthogonal expansions are series for a given function f in terms of orthogonal functions, in which orthogonality helps determining the coefficients in a simple fashion – this is a main point of orthogonality. For the Legendre polynomials this gives **Fourier–Legendre series**, as follows. Take, for instance $f(x) = \sin \pi x$,

```
> f := sin(Pi*x);                              # Resp. f := sin(π x)
```

The series will be $f = a_0\,P_0(x) + a_1\,P_1(x) + ...$ with coefficients a_m equal $(2m+1)/2$ times the integral of $f P_m$ from $x = -1$ to 1. Recall the simplest procedure in Example 5.2 in this Guide and type (note that the orthopoly package must be loaded before)

```
> c := proc(m)
    (2*m + 1)/2*int(f*P(m, x), x = -1..1);
  end:
```

Then you get single coefficients or, better, several at once, say, a_0, a_1, ..., a_5, by the command seq applied to c(m) – don't forget the m,

```
> seq(c(m), m = 0..5);
```

$$0,\ \frac{3}{\pi},\ 0,\ \frac{7\left(\pi^2-15\right)}{\pi^3},\ 0,\ \frac{11\left(\pi^4-105\,\pi^2+945\right)}{\pi^5}$$

```
> evalf[6](%);                     # Resp.
```

$0,\ 0.954930,\ 0,\ -1.15825,\ 0,\ 0.219340$

The even-numbered coefficients are zero because f is odd. You should now design a procedure that gives you the *terms* of your series, not just the coefficients,

```
> term := proc(m)
    (2*m + 1)/2*int(f*P(m, x), x = -1..1)*P(m, x);
  end:
```

```
> term(3);                     # Resp.
```

$$\frac{7\left(\pi^2-15\right)\left(\frac{5}{2}\,x^3-\frac{3}{2}\,x\right)}{\pi^3}$$

This is not what you want and need. The Legendre polynomials should remain untouched, as shown above in the series. To accomplish this, type '$P_n(x)$' in primes as shown,

```
> term := proc(m)
    (2*m + 1)/2*int(f*P(m, x), x = -1..1)*'P(m, x)';
  end:
```

```
> term(3);                     # Resp.
```

$$\frac{7\left(\pi^2-15\right)P\left(3,x\right)}{\pi^3}$$

```
> evalf(%);                     # Resp.
```

$-2.895604778\,x^3 + 1.737362867\,x$

Similarly, type your series (it is short), and use it to show that often a small number of terms gives good approximations

```
> sum('term(m)', m = 0..3);                     # Resp.
```

$$\frac{3P\left(1,x\right)}{\pi}+\frac{7\left(\pi^2-15\right)P\left(3,x\right)}{\pi^3}$$

To have the coefficients as decimal fractions, use evalf. Then plot.

```
> S := sum(evalf(term(m)), m = 0..3);
```

$$S := 2.692292526\,x - 2.895604781\,x^3$$

```
> plot(f, S, x = -1..1, xtickmarks = [-1, -0.5, 0, 0.5, 1],
  ytickmarks = [-1, -0.5, 0, 0.5, 1], linestyle = [solid, dash]);
```



Example 11.6. Sine function and approximation by $a_1\,P_1(x) + a_3\,P_3(x)$

The accuracy is surprising (see what happens with 5 terms). The areas under the curves seem almost equal, and so are the locations and size of the extrema. The zeros of $\sin \pi x$ are not too well approximated.

Similar Material in AEM: Sec. 5.8

EXAMPLE 11.7 FOURIER INTEGRAL, FOURIER TRANSFORM

The **Fourier integral representation** of a given function $f(x)$ defined on the x-axis (and satisfying certain regularity conditions; see AEM, Sec. 11.7) is

$$f(x) = \int_0^\infty [A(w)\,\cos\,wx + B(w)\,\sin\,wx]\,dw$$

where

$$A(w) = \frac{1}{\pi}\int_{-\infty}^{\infty} f(v)\,\cos\,wv\,dv,$$
$$B(w) = \frac{1}{\pi}\int_{-\infty}^{\infty} f(v)\,\sin\,wv\,dv.$$

For instance, find the Fourier integral representation of the function $f(x)$ which equals 1 for $-1 < x < 1$ and zero everywhere else on the x-axis (*not* periodic).

Solution. The integration in $A(w)$ and $B(w)$ extends from -1 to 1 only because $f(x)$ is zero elsewhere. Thus type

```
> A := 1/Pi*int(1*cos(w*v), v = -1..1);
```
Resp. $A := \dfrac{2\sin(w)}{\pi\,w}$

```
> B := 1/Pi*int(1*sin(w*v), v = -1..1);
```
Resp. $B := 0$

```
> f := 2/Pi*'int(sin(w)/w*cos(w*x), w = 0..infinity)';
```

$$f := \frac{2\left(\displaystyle\int_0^\infty \frac{\sin(w)\cos(wx)}{w}dw\right)}{\pi}$$

Note the ' before and after the `int`egral, which forces Maple to leave the integral unevaluated. Try it without to see that then the computer simply reproduces the given function (type `?signum` for information).

It can be shown (see AEM, Sec. 11.9) that the Fourier integral may be converted to complex form, and from it one can derive the **Fourier transform** $F(w)$ of $f(x)$ given by

$$F(w) = \frac{1}{\sqrt{2\,\pi}} \int_{-\infty}^{\infty} f(x)\, e^{-iwx}\, dx$$

and the **inverse Fourier transform** of $F(w)$ given by

$$f(x) = \frac{1}{\sqrt{2\,\pi}} \int_{-\infty}^{\infty} F(w)\, e^{iwx}\, dw.$$

For instance, find the Fourier transform of $f(x) = e^{-x^2}$.

Solution. Let `F` denote the Fourier transform of the given $f(x)$. Type

```
> F := 1/(sqrt(2*Pi))*int(exp(-x^2)*exp(-I*w*x), x = -infinity..infinity);
```

$$F := \frac{1}{2}\sqrt{2}\, e^{-\frac{1}{4}w^2}$$

Similar Material in AEM: Sec. 11.9

Problem Set for Chapter 11

Pr.11.1 (Rectangular wave. Gibbs phenomenon) Find the Fourier series of the following function of period 2π and make plots that show the Gibbs phenomenon. (*AEM* Sec. 11.1)

$$f(n) = \begin{cases} -4 & \text{if } -\pi < x < 0 \\ 4 & \text{if } \quad 0 < x < \pi \end{cases}$$

Pr.11.2 (Cosine series) Find the Fourier series of the function (sketch it) given by

$$f(n) = \begin{cases} 1 & \text{if } -\pi/2 < x < \pi/2 \\ -1 & \text{if } \quad \pi/2 < x < 3\pi/2\pi \end{cases}$$

and periodic with period 2π. (*AEM* Sec. 11.1)

Pr.11.3 (Cosine and sine terms) Find the Fourier series of $f(x)$ given by $f(x) = [x/(2\pi)]^4$ if $0 < x < 2\pi$ and of period 2π. (*AEM* Sec. 11.1)

Pr.11.4 (Half-wave rectifier) Pass the current $f(t) = \sin\, t$ through a half-wave rectifier that clips the negative portion of the wave. Find the Fourier series of the resulting periodic function $g(t)$ and plot some of its partial sums. (*AEM* Sec. 11.2)

Pr.11.5 (Periodic force, resonance) If a force $f(t) = t^2$ for $0 < t < 2\pi$ and periodic with period 2π is acting as the driving force on a mechanical system, which term of its Fourier series has the greatest coefficient in absolute value (so that the corresponding frequency should be watched for possible resonance effects)? (*AEM* Sec. 11.5)

Pr.11.6 (Behavior near a jump) Find the Fourier series of the periodic function $f(x) = \pi\, x^5/2$ $(-1 < x < 1)$ of period $p = 2$. Show, by plots, that, at the jumps, the partial sums give the arithmetic mean of the right-hand and left-hand limits of $f(x)$. (*AEM* Sec. 11.2)

Pr.11.7 (Continuous function) Find the Fourier series of the periodic function $f(x) = 5x^2$ $(-1 < x < 1)$ of period $p = 2$. Show, by a plot, that a partial sum of few terms gives a relatively good approximation (except near the cusps). (*AEM* Sec. 11.2)

Pr.11.8 (Triangular wave) Find the Fourier series of the function (sketch it)

$$f(n) = \begin{cases} 2+x & \text{if } -2 < x < 0 \\ 2-x & \text{if } \quad 0 < x < 2 \end{cases}$$

Plot $f(x)$ and some partial sums. (*AEM* Sec. 11.2)

Pr.11.9 (Half-range expansion) Find the Fourier cosine and sine series of $f(x) = x$, where $0 < x < L$ and L is arbitrary. (*AEM* Sec. 11.3)

Pr.11.10 (Herringbone wave) Find the Fourier series of the function (sketch it)

$$f(n) = \begin{cases} x & \text{if } 0 < x < 1 \\ 1-x & \text{if } 1 < x < 2\pi \end{cases}$$

Plot $f(x)$ and some partial sums. Observe the Gibbs phenomenon. (*AEM* Sec. 11.3)

Pr.11.11 (Error distribution) Find and plot $f(x) - S_5$ for x^2 from $-\pi$ to π, where $f(x) = x^3$ $(-\pi < x < \pi)$ and periodic with period $2\,\pi$ and S_5 is the partial sum of the Fourier series containing terms of sin x to sin $5x$. (*AEM* Sec. 11.6)

Pr.11.12 (Minimum square error) Let $f(x) = x^4$ $(-\pi < x < \pi)$ and periodic with period 2π. Find the minimum square error for the trigonometric approximation of $f(x)$ by polynomials of degree $N = 1, ..., 10$ and $N = 100$. (See Example 11.5 in this Guide. *AEM* Sec. 11.4)

Pr.11.13 (Minimum square error) Do the same task as in Pr.11.12 for the periodic function $f(x) = x^3$ $(-\pi < x < \pi)$ of period 2π.

Pr.11.14 (Fourier integral, Fourier cosine integral) Using the Fourier integral representation in Example 11.7 in this Guide, represent

$$f(n) = \begin{cases} \pi e^{-3x} & \text{if } x > 0 \\ 0 & \text{otherwise} \end{cases}$$

by a Fourier integral. (*AEM* Sec. 11.7)

Pr.11.15 (Experiment on Gibbs phenomenon) Study and plot the Gibbs phenomenon for functions of your choice. On what does the height of the "spikes" depend? Or is it always the same? Is the speed of the shift of the waves toward the discontinuity different for different functions? On what does it depend? What about the number of "spikes"?

Pr.11.16 (Orthogonality) The functions cos nx with positive integer are orthogonal on any interval of length 2π and have the norm $\sqrt{\pi}$. Verify this for $n = 1, ..., 10$ by integration. (*AEM* Sec. 11.5)

Pr.11.17 (Experiment on Fourier–Legendre series) Represent e^{-x} on the interval from -3 to 3 by partial sums S_n of a Fourier–Legendre series. Plot e^{-x} and the approximating partial sum on common axes, beginning with 2 terms and taking more and more terms stepwise until the two curves practically coincide in the plot. Describe what you can see regarding the increase of accuracy (a) on the subinterval from -1 to 1, (b) on the entire interval from -3 to 3. Calculate e approximately from S_n with increasing n. Find an empirical formula for the (approximate) size of the error as a function of n. (*AEM* Sec. 11.6)

Pr.11.18 (Fourier–Legendre series) Develop x^7 in a Fourier–Legendre series. Why will this series reduce to a polynomial? Plot x^7 and the corresponding Fourier–Legendre series of powers up to x^4 and comment on the quality of approximation. Show graphically that the error (as a function of x) is oscillating. (*AEM* Sec. 11.6)

Chapter 12

Partial Differential Equations (PDEs)

Content. Wave equation, animation (Exs. 12.1, 12.2, Prs. 12.1–12.3)
Tricomi, Airy equations, vibrating beam (Prs. 12.6, 12.7)
Heat equation, Laplace equation (Exs. 12.3, 12.4, Prs. 12.8–12.12)
Vibrating membrane (Exs. 12.5, 12.6, Prs. 12.13–12.15)

PDEtools package. Load it by typing `with(PDEtools)`. It may help you in certain tasks (see some of our examples). Type `?PDEtools`.

The PDEs that we consider are solved by separating variables, which reduces them to ODEs, and by the subsequent use of Fourier series and integrals.

Examples for Chapter 12

EXAMPLE 12.1 WAVE EQUATION. SEPARATION OF VARIABLES. ANIMATION

The **one-dimensional wave equation** is $u_{tt} = c^2\, u_{xx}$. It governs the vertical vibrations of an **elastic string**, such as a violin string. Then $u(x,t)$ is the displacement from rest along the x-axis at a point x and time t. The string of length L is fixed at its ends $x = 0$ and $x = L$. This gives the boundary conditions $u(0,t) = 0$ and $u(L,t) = 0$ for all t. In separating variables you look for solutions of the form $u(x,t) = F(x)G(t)$ satisfying the boundary conditions. So, type this $u(x,t)$ and then the wave equation.

```
> u(x,t) := F(x)*G(t);
```
Resp. $u(x,t) := F(x)\, G(t)$

```
> pde := diff(u(x,t), t, t) = c^2*diff(u(x,t), x, x);
```

$$pde := F(x)\left(\frac{d^2}{dt^2}G(t)\right) = c^2\left(\frac{d^2}{dx^2}F(x)\right)G(t)$$

Now separate the variables by dividing by $c^2\, u(x,t) = c^2\, F(x)\, G(t)$ and then set each of the two sides equal to a constant, which must be negative, say $-p^2$, in order to avoid ending up with an identically vanishing solution. First obtain a general solution of the ordinary differential equation for $F(x)$, written as `rhs(eq)` $= -p^2$. Reduce this solution by the left boundary condition. Then determine p from the right boundary condition. Finally solve the ordinary differential equation for $G(t)$, obtaining a general solution (`solG`, below).

```
> eq := pde/(c^2*u(x,t));
```
Resp. $eq := \dfrac{\frac{d^2}{dt^2}G(t)}{c^2 G(t)} = \dfrac{\frac{d^2}{dx^2}F(x)}{F(x)}$

```
> sol1 := dsolve(rhs(eq) = -p^2);
```

$$sol1 := F(x) = \_C1 \sin(px) + \_C2 \cos(px)$$

```
> s2 := eval(subs(x = 0, sol1));
```
Resp. $s2 := F(0) = \_C2$

This tells us that $\_C2 = 0$ because the string is fixed at its left end, i.e. $x = 0$. Also, you may take $\_C1 = 1$ because arbitrary constants will be provided by $G(t)$.

```
> s3 := subs(_C1 = 1, _C2 = 0, sol1);
```
Resp. $s3 := F(x) = \sin(px)$

```
> s4 := subs(x = L, s3);
```
Resp. $s4 := F(L) = \sin(pL)$

To satisfy the boundary condition at $x = L$, i.e. $F(L) = 0$, requires that $pL = n\pi$ with integer n; thus $p = n\,\pi/L$, $n = 1, 2, \ldots$. (Because $\sin(-\alpha) = -\sin\alpha$, you need not consider $n = -1, -2, \ldots$.) Type

```
> p := n*Pi/L;
```
Resp. $p := \dfrac{n\pi}{L}$

```
> solF := s3;
```
Resp. $solF := F(x) = \sin\left(\dfrac{n\pi x}{L}\right)$

Now obtain the ODE for $G(t)$ by setting the left-hand side of `eq` equal to $-p^2$ and solve the ODE

```
> solG := dsolve(lhs(eq) = -p^2);
```

$$solG := G(t) = \_C1 \sin\left(\frac{\pi n c t}{L}\right) + \_C2 \cos\left(\frac{\pi n c t}{L}\right)$$

From the solutions $F(x)$ and $G(t)$ in `solF` and `solG` you obtain $u(x,t) = G(t)\,F(x)$

```
> u_n(x,t) := rhs(solG)*rhs(solF);
```

$$u\_n(x,t) := \left(\_C1 \sin\left(\frac{\pi n c t}{L}\right) + \_C2 \cos\left(\frac{\pi n c t}{L}\right)\right) \sin\left(\frac{\pi n x}{L}\right)$$

Taking series of these solutions with $n = 1$ to ∞ and suitable constants, one can determine solutions that satisfy given initial conditions (displacement and velocity), as explained in Sec. 12.3 of AEM.

Animation shows the string in motion. The simplest solution $u(x,t)$ is the product $\sin t \sin x$ (when $n = 1$, $c = 1$, $L = \pi$ in the previous formula). Type

```
> with(plots):
```

```
> animate(sin(t)*sin(x), x = 0..Pi, t = Pi/2..10*Pi, frames = 100);
```

Press `Enter` (as usual). The maximally displaced string (for $t = \pi/2$, the beginning) will appear. Click with the mouse anywhere in the figure. A frame will appear, and above (outside the worksheet) you will see a row of symbols (a tool-bar). Click on the triangle (play). The motion will begin and make 4 1/2 cycles. Click again for a repetition. The fewer frames you choose (e.g. `frames = 50`), the more rapid the motion will be. (Type `?plots,animate` for information.)



Example 12.1. Fundamental mode of the vibrating string (Animation)

Similar Material in AEM: Sec. 12.3

EXAMPLE 12.2 WAVE EQUATION: D'ALEMBERT'S SOLUTION METHOD. COMMAND `pdsolve`

D'Alembert's solution method for the one-dimensional wave equation

$$u_{tt} - c^2\, u_{xx} = 0$$

consists of transforming the equation, by introducing two new independent variables, so that one can immediately solve the new equation by two successive integrations. Accordingly, type the wave equation, then the formula `tr` for the transformed variables and then the transformed equation `pde2`. (In the response to `pde2` the factors c may appear in different positions).

```
> with(PDEtools):
```

```
> pde1 := diff(u(x,t), t, t) - c^2*diff(u(x,t), x, x) = 0;
```

$$pde1 := \frac{\partial^2}{\partial t^2} u(x,t) - c^2 \left(\frac{\partial^2}{\partial x^2} u(x,t)\right) = 0$$

```
> tr := x = (v + z)/2, t = (v - z)/(2*c);
```

$$tr := \left\{t = \frac{1}{2}\frac{v-z}{c}, x = \frac{1}{2}v + \frac{1}{2}z\right\}$$

```
> pde2 := dchange(tr, pde1, [v, z]);
```

$$pde2 := c\left(c\left(\frac{\partial^2}{\partial v^2} u(v,z,c)\right) - c\left(\frac{\partial^2}{\partial z \partial v} u(v,z,c)\right)\right) - c\left(c\left(\frac{\partial^2}{\partial z \partial v} u(v,z,c)\right) - c\left(\frac{\partial^2}{\partial z^2} u(v,z,c)\right)\right) - c^2\left(\frac{\partial^2}{\partial v^2} u(v,z,c) + 2\left(\frac{\partial^2}{\partial z \partial v} u(v,z,c)\right) + \frac{\partial^2}{\partial z^2} u(v,z,c)\right) = 0$$

```
> pde2 := simplify(pde2);
```
\# Resp. $pde2 := -4c^2 \left(\frac{\partial^2}{\partial z \partial v} u(v,z,c)\right) = 0$

Drop $-4\,c^2$, which is not essential (also, as c is a constant, drop it from u), and integrate this PDE with respect to z. From the left-hand side you obtain

```
> int(lhs(pde2)/(-4*c^2), z);
```
\# Resp. $\left(\frac{\partial}{\partial v} u(v,z)\right)$

Add an arbitrary integration "constant" $h(v)$ (the result of integrating 0 on the right with respect to z). (Maple puts all integration constants to 0.)

```
> eq := diff(u(v,z), v) - h(v);
```
\# Resp. $eq := \frac{\partial}{\partial v} u(v,z) - h(v)$

Integrate this with respect to v

```
> int(diff(u(v,z), v), v)-int(h(v), v);
```
\# Resp. $u(v,z) - \int h(v)dv$

Add another integration "constant" $-g(z)$. Hence $u - \int h(v)\,dv = g(z)$. Call the integral $f(v)$. Then, in the original variables ($v = x + ct$, $z = x - ct$) the solution, called **D'Alembert's solution**, is

$$u(x,t) = f(x+ct) + g(x-ct).$$

Practically the same solution can be obtained by the command `pdsolve`, as follows, where _F1 and _F2 are arbitrary functions.

```
> sol := pdsolve(pde1, u(x,t));
```
$$sol := u(x,t) = \_F1\,(ct+x) + \_F2\,(ct-x)$$

Similar Material in AEM: Sec. 12.4

EXAMPLE 12.3 ONE-DIMENSIONAL HEAT EQUATION

The **one-dimensional heat equation** is $u_t = c^2\,u_{xx}$, where $u(x,t)$ is the temperature in a straight bar or wire and the boundary conditions are $u(0,t) = 0$ and $u(\pi,t) = 0$ for all t. Separation of variables leads to solutions $u = u_n$,

$$u_n(x,\,t) = F_n(x)\,G_n(t) = B_n \sin nx \exp\left(-(c\,n)^2 t\right)$$

satisfying the boundary conditions. A series of these will satisfy a given initial condition $f(x) = u(x,0)$, for instance, an initial temperature (see the figure)

$$f(x) = \begin{cases} x^3 & \text{if} \quad 0 < x < \pi/2 \\ (\pi - x)^3 & \text{if} \quad \pi/2 < x < \pi \end{cases}$$

if the coefficients B_n of the series are the Fourier coefficients of the Fourier sine half-range expansion of $f(x)$ (of period $2\,\pi$). Show this in a plot, which also exhibits the temperatures for various constant values of t. Take $c = 1$ for simplicity.

Solution. Compute B_n and then find a partial sum of the series of the u_n with these coefficients B_n for various $t = 0,\ 0.1,\ 0.2,\ \ldots$. For information on one of the commands below, type `?piecewise`. Read the command as follows. For $x < 0$ the function f is 0. For $x < \pi/2$ it is x^3. For $x < \pi$ it is $(\pi - x)^3$. Elsewhere it is 0.

```
> f := piecewise(x > 0 and x < Pi/2, x^3, x > Pi/2 and x < Pi, (Pi - x)^3):
```

```
> Bn := simplify(2/Pi*(int(f*sin(n*x), x = 0..Pi)));
```

$$Bn := \frac{3\left(n^2 \sin\left(\frac{1}{2}n\pi\right)\pi^2 - 8\sin\left(\frac{1}{2}n\pi\right) + 4\sin(n\pi)\right)}{\pi n^4}$$

```
> S := sum(Bn*sin(n*x)*exp(-n^2*t), n = 1..13);
```

$$S := \frac{3(\pi^2-8)\sin(x)\,e^{-t}}{\pi} + \frac{1}{27}\frac{(-9\pi^2+8)\sin(3x)\,e^{-9t}}{\pi} + \frac{3}{625}\frac{(25\pi^2-8)\sin(5x)\,e^{-25t}}{\pi} + \frac{3}{2401}\frac{(-49\pi^2+8)\sin(7x)\,e^{-49t}}{\pi} + \frac{1}{2187}\frac{(81\pi^2-8)\sin(9x)\,e^{-81t}}{\pi} + \frac{3}{14641}\frac{(-121\pi^2+8)\sin(11x)\,e^{-121t}}{\pi} + \frac{3}{28561}\frac{(169\pi^2-8)\sin(13x)\,e^{-169t}}{\pi}$$

```
> S0 := eval(subs(t = 0, S)):     S1 := eval(subs(t = 0.1, S)):
> S2 := eval(subs(t = 0.2, S)):   S10 := eval(subs(t = 1.0, S)):
> S20 := eval(subs(t = 2.0, S)):  S30 := eval(subs(t = 3.0, S)):
> plot(S0, S1, S2, S10, S20, S30, f, x = 0..Pi, color = black);
```



Example 12.3. "Triangular" initial temperature and its decrease with time

Similar Material in AEM: Sec. 12.5

EXAMPLE 12.4 HEAT EQUATION, LAPLACE EQUATION

Find the steady-state temperature in the rectangular plate $0 \le x \le \pi$, $0 \le y \le \pi/2$ if the upper edge is kept at temperature $u(x, \pi/2) = f(x) = 2$ and the other three edges are kept at temperature 0. Plot the temperature as a surface over the rectangle.

Solution. The two-dimensional heat equation is $u_t = c^2(u_{xx} + u_{yy})$. Because u is assumed to be steady-state (time independent), this equation reduces to **Laplace's equation** $u_{xx} + u_{yy} = 0$. Separation of variables leads to solutions

$$u_n(x, y) = B_n \sin nx \sinh ny,$$

that satisfy the three zero boundary conditions. It can be shown that a series of these also satisfies $u(x, \pi/2) = f(x) = 2$ if you choose the $B_n \sinh(n\pi/2)$ to be the coefficients of the Fourier sine half-range expansion of $f(x)$. Thus type the following. (Type `?orientation` for information. Actually, you can turn the surface by clicking on any point in the figure and then moving the mouse while holding the button down. Try it.) By the Euler formulas

```
> upEdge := 2:
```

```
> Bn := 2/(Pi*sinh(n*Pi/2))*int(upEdge*sin(n*x), x = 0..Pi);
```

$$Bn := -\frac{4\left(-1+\cos\left(n\pi\right)\right)}{\pi \sinh\left(\frac{1}{2}n\pi\right) n}$$

```
> u30 := sum(Bn*sin(n*x)*sinh(n*y), n = 1..30):
```

```
> with(plots):
```

```
> plot3d(u30, x = 0..Pi, y = 0..Pi/2, axes = NORMAL,
    orientation = [30, 60]);
```

To obtain a figure showing the approximation of the temperature along the upper edge, type

```
> w := subs(y = Pi/2, u30):
```

```
> plot(w, x = 0..Pi, y = 0..2.5, scaling = constrained);
```



Example 12.4. Temperature given by `u30` shown as a surface over the rectangular plate



Example 12.4. Temperature given by `u30` on the upper edge

Note the beginning Gibbs phenomenon near 0 and π.

Similar Material in AEM: Sec. 12.5

EXAMPLE 12.5 RECTANGULAR MEMBRANE. DOUBLE FOURIER SERIES

Vibrations of elastic membranes (such as drumheads) are governed by the **two-dimensional wave equation**

$$u_{tt} = c^2\,(u_{xx} + u_{yy}),$$

where $u(x, y, t)$ is the displacement of the membrane at a point (x, y) and time t from its position at rest in the xy-plane. Let the density of the material and the tension be such that $c^2 = 49$. The membrane is 3×5. Let the initial velocity be zero and the initial displacement $f(x, y)$ as typed below and as shown in the figure. (You can turn the figure by clicking on any point of it and then moving the mouse.)

```
> f := 1/20*(3*x - x^2)*(5*y - y^2);
```

$$f := \frac{1}{20}\left(-x^2 + 3x\right)\left(-y^2 + 5y\right)$$

```
> plot3d(f, x = 0..3, y = 0..5, axes = NORMAL);
```



Example 12.5. Initial displacement of the membrane

The solution method is similar to that for the vibrating string. Separation of variables leads to a double sequence of solutions

$$u_{mn} = B_{mn}\cos\left(\lambda_{mn}\, t\right)\sin\left(\frac{m\pi x}{a}\right)\sin\left(\frac{n\pi y}{b}\right) \qquad (m,\ n = 1, 2, \ldots)$$

satisfying the boundary condition $u = 0$ on the edges of the membrane. A double series of these (a "**double Fourier series**") will satisfy the initial conditions $u(x, y, 0) = f(x, y)$ (given initial displacement) and $u_t(x, y, 0) = 0$ (zero initial velocity) if the B_{mn} are the coefficients of the double Fourier series of $f(x, y)$. Accordingly, type the following. (The resulting terms may come out in a different order.) The **Euler formulas for a double Fourier series** give

```
> a := 3:  b := 5:  c := 7:
```

```
> Bmn := 4/(a*b)*int(int(f*sin(m*Pi*x/a)*sin(n*Pi*y/b), x = 0..a),
  y = 0..b);
```

$$Bmn := \frac{1}{m^3\pi^6 n^3}\left(45\left(\pi^2\sin(m\pi)\sin(n\pi)mn + 2\pi\sin(m\pi)\cos(n\pi)m\right.\right.$$
$$+ 2\pi\cos(m\pi)\sin(n\pi)n - 2m\pi\sin(m\pi) - 2n\pi\sin(n\pi)$$
$$\left.\left.+ 4\cos(m\pi)\cos(n\pi) - 4\cos(m\pi) - 4\cos(n\pi) + 4\right)\right)$$

Tell the computer that m and n are integers, so that B_{mn} will simplify. (The tildes after m and n remind you of these assumptions. Type `?assume`.)

```
> assume(m, integer):  assume(n, integer):
```

```
> Bmn := simplify(Bmn);
```

$$Bmn := \frac{180\left(1 + (-1)^{m\sim+n\sim} + (-1)^{1+m\sim} + (-1)^{1+n\sim}\right)}{m\sim^3\pi^6 n\sim^3}$$

For the time function in u_{mn} you need λ_{mn} as follows.

```
> lambda_mn := c*Pi*sqrt(m^2/a^2 + n^2/b^2);
```

$$lambda\_mn := \frac{7}{15}\pi\sqrt{25m{\sim}^2 + 9n{\sim}^2}$$

```
> umn := Bmn*cos(lambda_mn*t)*sin(m*Pi*x/a)*sin(n*Pi*y/b);
```

$$umn := \frac{1}{m{\sim}^3\pi^6 n{\sim}^3}\left(180\left(1 + (-1)^{m\sim+n\sim} + (-1)^{1+m\sim} + (-1)^{1+n\sim}\right)\right.$$
$$\left.\cos\left(\frac{7}{15}\pi\sqrt{25m{\sim}^2 + 9n{\sim}^2}t\right)\sin\left(\frac{1}{3}m{\sim}\pi x\right)\sin\left(\frac{1}{5}n{\sim}\pi y\right)\right)$$

Now type a partial sum `S` of these $u_{mn}(x, y, t)$. Very few terms will do. In fact, let us take just a single term. (Try more terms.) Then type `S0`, which is `S` for $t = 0$, that is, an approximation of the initial shape of the membrane.

```
> S := subs(n = 1, subs(m = 1, umn));
```

$$S := \frac{720\cos\left(\frac{7}{15}\pi\sqrt{34}t\right)\sin\left(\frac{1}{3}\pi x\right)\sin\left(\frac{1}{5}\pi y\right)}{\pi^6}$$

```
> S0 := eval(subs(t = 0, S));
```

$$S0 := \frac{720\sin\left(\frac{1}{3}\pi x\right)\sin\left(\frac{1}{5}\pi y\right)}{\pi^6}$$

```
> plot3d(S0, x = 0..a, y = 0..b, axes = NORMAL);
```

Similar Material in AEM: Sec. 12.8

EXAMPLE 12.6 LAPLACIAN. CIRCULAR MEMBRANE. BESSEL EQUATION

This concerns the vertical vibrations of a **circular membrane** fixed along its edge $x^2 + y^2 = R^2$ in the xy-plane. The vibrations are rotationally symmetric (i.e., independent of θ). The membrane has an initial displacement ($u(r, 0) = f(r)$) but no initial velocity ($u_t(r, 0) = 0$). You must first obtain the two-dimensional wave equation $u_{tt} = c^2(u_{xx} + u_{yy})$ in polar coordinates. (Type `?Laplacian` for information.) Relevant Maple commands are as follows.

```
> with(VectorCalculus):
```

```
> lap := Laplacian(u(x,y), [x, y]);
```

$$lap := \frac{\partial^2}{\partial x^2}u(x, y) + \frac{\partial^2}{\partial y^2}u(x, y)$$

```
> with(PDEtools):
```

```
> tr := x = r*cos(theta), y = r*sin(theta):
```

```
> dchange(tr, lap);                  # Change variables - long response.
```

```
> combine(%);            # Resp.
```

$$\frac{\left(\frac{\partial^2}{\partial r^2}u(r,\theta)\right)r^2 + \left(\frac{\partial}{\partial r}u(r,\theta)\right)r + \frac{\partial^2}{\partial \theta^2}u(r,\theta)}{r^2}$$

This is the Laplacian in polar coordinates. You may confirm this by typing

```
> simplify(Laplacian(u(r, theta), 'polar'[r, theta]));
```

$$\frac{\left(\frac{\partial^2}{\partial r^2}u(r,\theta)\right)r^2+\left(\frac{\partial}{\partial r}u(r,\theta)\right)r+\frac{\partial^2}{\partial \theta^2}u(r,\theta)}{r^2}$$

By rotationally symmetry, we can remove the dependence on θ, i.e. $u = u(r,t)$, so

```
> pde := diff(u(r,t), t, t) = c^2*Laplacian(u(r,t), 'polar'[r, theta]);
```

$$pde := \frac{\partial^2}{\partial t^2}u(r,t) = \frac{c^2\left(\frac{\partial}{\partial r}u(r,t)+r\left(\frac{\partial^2}{\partial r^2}u(r,t)\right)\right)}{r}$$

Now separate variables. Start from

```
> pde2 := eval(subs(u(r,t) = W(r)*G(t), pde));
```

$$pde2 := W(r)\left(\frac{d^2}{dt^2}G(t)\right) = \frac{c^2\left(\left(\frac{d}{dr}W(r)\right)G(t)+r\left(\frac{d^2}{dr^2}W(r)\right)G(t)\right)}{r}$$

```
> eq1 := pde2/(c^2*W(r)*G(t));
```

$$eq1 := \frac{\frac{d^2}{dt^2}G(t)}{c^2G(t)} = \frac{\left(\frac{d}{dr}W(r)\right)G(t)+r\left(\frac{d^2}{dr^2}W(r)\right)G(t)}{W(r)\,G(t)\,r}$$

```
> eq2 := simplify(eq1);
```

$$eq2 := \frac{\frac{d^2}{dt^2}G(t)}{c^2G(t)} = \frac{\left(\frac{d^2}{dr^2}W(r)\right)r+\frac{d}{dr}W(r)}{W(r)r}$$

The variables are separated. Each side must equal a constant, which must be negative, say, $-k^2$, in order to finally obtain solutions that are not identically zero.

To obtain the solutions for G, set the left-hand side equal to $-k^2$ or write `lhs(eq2) +` $k^2 = 0$, and then multiply by $c^2G(t)$.

```
> eq3 := lhs(eq2) + k^2 = 0;
```

Resp. $eq3 := \frac{\frac{d^2}{dr^2}G(t)}{c^2G(t)} + k^2 = 0$

```
> eq4 := simplify(eq3*c^2*G(t));
```

Resp. $eq4 := k^2c^2G(t) + \frac{d^2}{dr^2}G(t) = 0$

Assume $k > 0$ (which Maple indicates by a tilde after k) and solve this ordinary differential equation by `dsolve`.

```
> assume(k > 0):
```

```
> with(DEtools):
```

```
> solG := dsolve(eq4, G(t));
```

$$solG := G(t) = \_C1\sin(k{\sim}ct) + \_C2\cos(k{\sim}ct)$$

To obtain the solutions for W, set the right-hand side equal to $-k^2$ or write `rhs(eq2) +` $k^2 = 0$, and then multiply by $W(r)\,r$.

```
> eq5 := rhs(eq2) + k^2 = 0;
```

$$eq5 := \frac{\left(\frac{d^2}{dr^2}W(r)\right)r + \frac{d}{dr}W(r)}{W(r)\,r} + k\sim^2 = 0$$

```
> eq6 := simplify(eq5*W(r)*r);
```

$$eq6 := k\sim^2 W(r)\,r + \left(\frac{d^2}{dr^2}W(r)\right)r + \frac{d}{dr}W(r) = 0$$

This is the **Bessel equation** with parameter $\nu = 0$ (and independent variable $s = kr$). Solve this ordinary differential equation by `dsolve`.

```
> solW := dsolve(eq6, W(r)):
```

$$solW := W(r) = \_C1\,\mathrm{BesselJ}(0, k\sim r) + \_C2\,\mathrm{BesselY}(0, k\sim r)$$

Hence a solution is the Bessel function $J_0(k\,r)$. (The Bessel function Y_0 cannot be used because it becomes infinite as $r \to 0$. Type `?Bessel`, `?BesselJZeros` for information.)

The eigenfunctions, for this problem, are (for $m = 0, 1, 2, 3, \ldots$)

```
> u_m(r, t) := rhs(solG)*subs(_C1 = 1, _C2 = 0, rhs(solW));
```

$$u\_m(r,t) := (\_C1\sin(k\sim ct) + \_C2\cos(k\sim ct))\mathrm{BesselJ}(0, k\sim r)$$

Because we have assumed no initial velocity ($u_t(r, 0) = 0$), we set

```
> evalf(subs(t = 0, diff(u_m(r, t), t)));
```

$$1.\_C1 k\sim c\,\mathrm{BesselJ}(0., kr)$$

to 0, from which $\_C1 = 0$, and

$$u_m(r,t) := \_C2\cos(kct))\mathrm{BesselJ}(0, kr)$$

The eigenvalue, k, must be determined so that $J_0(kr) = 0$ on the edge $r = R$ of the membrane. For simplicity, take $R = 1$ and $c = 1$ (the constant in the wave equation). Then k must equal the first positive zero (this gives the simplest solution), the second positive zero, etc. (See the graph of J_0 in Example 5.6 in this Guide.) To obtain the first three zeros, type

```
> zeros := evalf(BesselJZeros(0, 1..3));
```

$$zeros := 2.404825558, 5.520078110, 8.653727913$$

```
> z1 := zeros[1]; z2 := zeros[2]; z3 := zeros[3];
```

$$z1 := 2.404825558$$
$$z2 := 5.520078110$$
$$z3 := 8.653727913$$

You can now plot three eigenfunctions ($t = 0$) by using the first three zeros:

```
> plot3d([r*cos(theta), r*sin(theta), BesselJ(0, z1*r)], r = 0..1,
    theta = 0..2*Pi, scaling = constrained, axes = none);
```

```
> plot3d([r*cos(theta), r*sin(theta), BesselJ(0, z2*r)], r = 0..1,
    theta = 0..2*Pi, scaling = constrained, axes = none);
```

```
> plot3d([r*cos(theta), r*sin(theta), BesselJ(0, z3*r)], r = 0..1,
    theta = 0..2*Pi, scaling = constrained, axes = none);
```



Example 12.6. Circular membrane, simplest forms of vibrations

Similar Material in AEM: Sec. 12.9

Problem Set for Chapter 12

Pr.12.1 (Normal modes of the vibrating string) The solutions `F(x)` in `solF` in Example 12.1 in this Guide are called the *normal modes* of the string. Plot them (separately) for $n = 1, 2, 3, 4, 5$ (with $L = \pi$). Multiply them by the corresponding sines of t (with $c = 1$) and apply animation. (See Example 12.1 in this Guide. *AEM* Sec. 12.3)

Pr.12.2 (Animation) Show 5 cycles of the motion

$$u(x,t) = \sin x \cos t - (1/9)\sin 3x \cos 3t + (1/25)\sin 5x \cos 5t - (1/49)\sin 7x \cos 7t$$

(approximating the motion when the initial deflection of the string is "triangular"). (See Example 12.1 in this Guide. *AEM* Sec. 12.3)

Pr.12.3 (Extension of D'Alembert's method) Solve $u_{xx} + u_{xy} - 2\,u_{yy} = 0$ by setting $v = 2x - y$, $z = x + y$. (*AEM* Sec. 12.4)

Pr.12.4 (Separation of variables) Solve $u_{xx} - u_{yy} = 0$, choosing the separation constant positive, zero, and negative. (*AEM* Sec. 12.3)

Pr.12.5 (Checking solutions) Checking is important. The computer may sometimes give you false or insufficient results. Check whether $u = (c_1\, e^{kx} + c_2\, e^{-kx})(c_3\, e^{ky} + c_4\, e^{-ky})$ is a solution of $u_{xx} - u_{yy} = 0$. Can you replace k in the functions depending on y by another constant?

Pr.12.6 (Tricomi equation. Airy equation) Find solutions $u(x,y) = F(x)G(y)$ of the Tricomi equation $yu_{xx} + u_{yy} = 0$. Show that for $G(y)$ this gives $G'' + kyG = 0$. (With $k = -1$ or 1 this is called *Airy's equation*.) Find a general solution $G(y)$ involving the Airy functions Ai and Bi (see Ref. [1] in Appendix 1 for information). Obtain $\mathrm{Ai}(-y)$ from that general solution and plot it. (*AEM* Sec. 12.4)

Pr.12.7 (Vibrating beam. Command `pdsolve`) Vertical vibrations of a horizontal elastic beam of homogeneous material and constant cross section are governed by the equation $u_{tt} = c^2\, u_{xxxx}$. Solve this fourth-order PDE by `pdsolve`. (Type `?pdsolve` for information. *AEM* Sec. 12.3)

Pr.12.8 (Heat equation) Solve the heat equation in Example 12.3 in this Guide for a bar of length 5 with $c = 1$ and "parabolic" initial temperature $u(x, 0) = x(5 - x)$. (*AEM* Sec. 12.5)

Pr.12.9 (Heat equation) Derive the solutions $u_n(x, t)$ in Example 12.3 in this Guide by separation of variables. (*AEM* Sec. 12.5)

Pr.12.10 (Heat flow in a long bar. Fourier integral. Error function) An infinite bar (the x-axis), practically a very long bar, is heated to 100 degrees at some point, for simplicity, between -1 and 1, the rest being kept at 0. At $t = 0$, heating is terminated and heat begins to flow away to both sides. It can be shown that the temperature is given by $100/\sqrt{\pi}$ times the integral of $\exp(-t^2)$ from $-(1+x)/(2c\sqrt{t})$ to $(1-x)/(2c\sqrt{t})$. Study the decrease of the temperature by animation, assuming that $c = 1$. The integral of $\exp(-t^2)$ from 0 to v (times $2/\sqrt{\pi}$) is called the **error function** and is denoted by $\operatorname{erf} v$. (Type `?animate`; see also the instruction on animation in Example 12.1 in this Guide. *AEM* Sec. 12.6)

Pr.12.11 (Two-dimensional heat equation) Solve Example 12.4 in this Guide when the upper edge is kept at the temperature $\sin x$, the other data being as before. (*AEM* Sec. 12.5)

Pr.12.12 (Isotherms) Find the isotherms (curves of constant temperature) in Pr.12.11 and plot some of them. Do these curves look physically reasonable? (*AEM* Sec. 12.5)

Pr.12.13 (Animation, rectangular membrane) Show the motion of the membrane in Example 12.5 in this Guide for u_{22} with $B_{22} = 1$. (*AEM* Sec. 12.8)

Pr.12.14 (Circular membrane) Show the motion of the three solutions in the figure of Example 12.6 in this Guide with $c = 1$, $R = 1$, so that $k^2 = {z_n}^2$. (Type `?BesselJZeros`. Sec. 12.9)

Pr.12.15 (Circular membrane, vibration depending on angle) Study the motion of the circular membrane given by $u_{11}(r, \theta, t) = \sin kt\, J_1(kr) \cos\theta$, which has the y-axis ($\theta = \pi/2$ and $3\pi/2$) as a nodal line (line along which the membrane does not move). Here, k is the first positive zero of J_1 (type `?BesselJZeros`).
(*AEM* Sec. 12.9)