Chapter 4

Systems of ODEs.

Phase Plane, Qualitative Methods

Content. Solution of systems, use of matrices (Exs. 4.1, 4.2)
Critical points, pendulum ODE (Exs. 4.3–4.5, Prs. 4.1–4.4, 4.6, 4.11)
Nonhomogeneous systems (Exs. 4.6, 4.7)
Van der Pol and related ODEs (Ex. 4.8, Prs. 4.12–4.15)
Mixing problems, networks (Prs. 4.8–4.10)

Writing and solving systems see Ex. 4.1, etc. **Use of matrices** needs the LinearAlgebra package; see Ex. 4.2. **Plotting trajectories** see Exs. 4.3, 4.5, 4.8.

Examples for Chapter 4

EXAMPLE 4.1 HOW TO WRITE A SYSTEM OF ODES? INITIAL VALUE PROBLEM

A system of ODEs

$$\begin{aligned} y_1' &= -4y_1 + 3y_2 \\ y_2' &= 3y_1 - 4y_2 \end{aligned}$$

can be written

```
> sys := D(y1)(t) = -4*y1(t) + 3*y2(t),
         D(y2)(t) = 3*y1(t) - 4*y2(t);
```

$$sys := \mathrm{D}(y1)(t) = -4\, y1(t) + 3\, y2(t),\ \mathrm{D}(y2)(t) = 3\, y1(t) - 4\, y2(t)$$

If, for some reason, you need the two equations separately, you can get them by using `sys[1]` and `sys[2]`; thus,

```
> sys[1];
```
Resp. $\mathrm{D}(y1)(t) = -4\, y1(t) + 3\, y2(t)$

```
> sys[2];
```
Resp. $\mathrm{D}(y2)(t) = 3\, y1(t) - 4\, y2(t)$

You can solve this system by `dsolve`,

```
> sol := dsolve(sys);
```

$$sol := \{y1(t) = \_C1\, \mathrm{e}^{-7\,t} + \_C2\, \mathrm{e}^{-t},\ y2(t) = \_C1\, \mathrm{e}^{-7\,t} - \_C2\, \mathrm{e}^{-t}\}$$

Note that the solutions may appear in a different order and with the notations for the arbitrary constants interchanged. Observe the use of braces in `{sol}` (this is the set of the two equations).
Initial value problems can be solved as for a single equation. Thus, let $y_1(0) = 0$ and $y_2(0) = 8$. You obtain the solution from `dsolve` and can plot the two curves as usual.

```
> yp := dsolve(sys, y1(0) = 0, y2(0) = 8);
```

$$yp := \{y1(t) = -4\,\mathrm{e}^{-7\,t} + 4\,\mathrm{e}^{-t}, y2(t) = 4\,\mathrm{e}^{-7\,t} + 4\,\mathrm{e}^{-t}\}$$

```
> plot(rhs(yp[1]), rhs(yp[2]), t = 0..5);
```



Example 4.1. Solutions y_1 (lower curve) and y_2 of the initial value problem

Similar Material in AEM: Sec. 4.3

EXAMPLE 4.2 USE OF MATRICES IN SOLVING SYSTEMS OF ODES

A general solution of a homogeneous linear system of ODEs turns out to involve the eigenvalues and eigenvectors of the coefficient matrix of the system and can readily be obtained from these eigenvalues and vectors. We explain this for the system in Example 4.1 in this Guide, $y_1{}' = -4y_1 + 3y_2$, $y_2{}' = 3y_1 - 4y_2$. To treat matrices and vectors with Maple, load the `LinearAlgebra` package by

```
> with(LinearAlgebra):
```

Type the coefficient matrix **A**,

```
> A := Matrix([[-4, 3], [3, -4]]);                # Resp.
```

$$A := \begin{bmatrix} -4 & 3 \\ 3 & -4 \end{bmatrix}$$

An **eigenvalue** of **A** is a number λ such that the vector equation $\mathbf{Ax} = \lambda\mathbf{x}$ has a vector solution **x** which is not the zero vector. (A zero vector is a solution of this equation for any value of λ, and is called the '***trivial solution***'.) This nonzero **x** is called an **eigenvector** of **A** corresponding to that eigenvalue λ. The eigenvalues are the solutions of the **characteristic equation** $\det(\mathbf{A} - \lambda\mathbf{I}) = 0$. Here, **I** is the **unit matrix**. If you want the eigenvalues, type

```
> Eigenvalues(A);                # Resp.
```

$$\begin{bmatrix} -1 \\ -7 \end{bmatrix}$$

If you want the eigenvalues as well as eigenvectors (as it will usually be the case), type

```
> eig := Eigenvectors(A);
```

$$eig := \begin{bmatrix} -1 \\ -7 \end{bmatrix}, \begin{bmatrix} 1 & -1 \\ 1 & 1 \end{bmatrix}$$

You should be aware that, in eig, the eigenvalues (and corresponding eigenvectors) may appear in a different order, and the eigenvectors may differ from the response shown here by a nonzero multiplicative constant. This gives the general solution of the system

$$\mathbf{y} = c_1\,\mathbf{x}_1 e^{-1\,t} + c_2\,\mathbf{x}_2 e^{-7\,t}.$$

You can pick from eig the items you need for composing this solution,

```
> eig[1];                    # Resp. [-1, -7]
> L1 := eig[1][1];           # Resp. L1 := -1
> x1 := eig[2][1];           # Resp. x1 := [ 1  -1 ]
> L2 := eig[1][2];           # Resp. L2 := -7
> x2 := eig[2][2];           # Resp. x2 := [ 1  1 ]
```

With these notations, you can now write the corresponding general solution as

```
> y := c1*x1*exp(L1*t) + c2*x2*exp(L2*t);
```

$$y := \left[\ c1\,\mathrm{e}^{-t} + c2\,\mathrm{e}^{-7\,t} \quad -c1\,\mathrm{e}^{-t} + c2\,\mathrm{e}^{-7\,t}\ \right]$$

This agrees with the answer in Example 4.1 of this Guide , except for the notations for the arbitrary constants.

Similar Material in AEM: Sec. 4.0, 4.3

EXAMPLE 4.3 CRITICAL POINTS. NODE

From a given linear system of ODEs we obtain

$$\frac{dy_2}{dy_1} = \frac{dy_2}{d\,t}\bigg/\frac{dy_1}{dt} = \frac{a_{21}\,y_1 + a_{22}\,y_2}{a_{11}\,y_1 + a_{12}\,y_2}.$$

At each point $(y_1,\, y_2)$ in the $y_1\,y_2$-plane (the **phase plane**) this determines a tangent direction of the solution graphed as a curve in the $y_1\,y_2$-plane. Such a curve is called a **trajectory** (or *path* or *orbit*). An exception is the point $(y_1,\, y_2) = (0,\ 0)$, at which the numerator and the denominator on the right are both zero. This is called a **critical point** of the system. It can be classified in terms of three quantities related to the coefficient matrix $\mathbf{A} = [a_{jk}]$ of the system. These quantities are

$p = a_{11} + a_{22}$	the **trace** of the matrix
$q = \det \mathbf{A} = a_{11}\,a_{22} - a_{12}\,a_{21}$	the **determinant** of the matrix
$\delta = p^2 - 4\,q.$	

The critical point is a:

Node	if $q > 0$ and $\delta \geq 0$,
Saddle point	if $q < 0$,
Center	if $p = 0$ and $q > 0$,
Spiral point	if $p \neq 0$ and $\delta < 0$.

For example, let the matrix of the system be (see the previous example)

```
> with(LinearAlgebra):
```

```
> A := Matrix([[-4, 3], [3, -4]]);
```
Resp. $A := \begin{bmatrix} -4 & 3 \\ 3 & -4 \end{bmatrix}$

Then the quantities needed for classifying the critical point are

```
> p := Trace(A);
```
Resp. $p := -8$

```
> q := Determinant(A);
```
Resp. $q := 7$

```
> delta := p^2 - 4*q;
```
Resp. $\delta := 36$

Because q and δ are positive, the critical point at the origin is a node. More precisely, it is an **improper node,** characterized by the property that, at that point, all trajectories, except for two of them, have the same tangent direction, as the figure shows. (A **proper node** is one at which for each direction there is a trajectory having it as the tangent direction at the node.) Each trajectory shown is obtained by choosing a pair of constants c_1, c_2 in the general solution in the preceding example. For $c_1 = 0$ you get one of the two straight lines and for $c_2 = 0$ the other. For any other choice of c_1, c_2 you get a quadratic parabola because in that general solution, $e^{-4t} = \left(e^{-2t}\right)^2$. In the plot commands, set $e^{-t} = \tau$ and reuse t for τ^2, for simplicity. In the following, five representative trajectories will be selected — based on the selection of the constants.

```
> with(plots):
```

```
> c1s := [1, 0, 1, 0, 1]:  c2s := [0, 1/10, 1/10, 1/20, 1/20]:
```

```
> Traj := [c1s[i]*t+c2s[i]*t^2, -c1s[i]*t+c2s[i]*t^2]:
```

```
> i := 1:  Traj; p1 := plot(Traj, t = -10..10, y=-10..20):
```
$$[t, -t]$$

```
> i := 2:  Traj; p2 := plot(Traj, t = -10..10, linestyle=dash):
```
$$\left[\frac{1}{10}t^2, \frac{1}{10}t^2\right]$$

```
> i := 3:  Traj; p3 := plot(Traj, t = -10..10, linestyle= dot):
```
$$\left[\frac{1}{10}t^2 + 1, \frac{1}{10}t^2 - 1\right]$$

```
> i := 4:  Traj; p4 := plot(Traj, t = -10..10, linestyle=dashdot):
```
$$\left[\frac{1}{20}t^2, \frac{1}{20}t^2\right]$$

```
> i := 5:  Traj; p5 := plot(Traj, t = -10..10, linestyle=longdash):
```
$$\left[\frac{1}{20}t^2 + t, \frac{1}{20}t^2 - t\right]$$

```
> display(p1, p2, p3, p4, p5, title = `Trajectories near a node`,
   scaling = constrained);
```



Example 4.3. Trajectories near an improper node

Similar Material in AEM: Sec. 4.3, 4.4

EXAMPLE 4.4 PROPER NODE, SADDLE POINT, CENTER, SPIRAL POINT

Improper nodes were discussed and shown in the previous example. Phase plane plots ("**phase portraits**") of the other types of critical points are given on pp. 143 – 146 of AEM. We present here the corresponding calculations for typical examples.

A ***proper node*** with trajectories $y_2 = ky_1$ and $y_1 = 0$ (straight lines) is obtained for the system ${y_1}' = y_1$, ${y_2}' = y_2$, whose matrix is the unit matrix.

A ***saddle point*** with trajectories $y_1\, y_2 = \text{const}$ (hyperbolas) is obtained for $y_1' = y_1$, $y_2' = -y_2$ with matrix

```
> with(LinearAlgebra):
```

```
> A := Matrix([[1, 0], [0, -1]]);
```

$$A := \begin{bmatrix} 1 & 0 \\ 0 & -1 \end{bmatrix}$$

for which $q = \det \mathbf{A} = -1 < 0$, so that we have a saddle (see the criteria in the previous example). The eigenvalues of $\mathbf{A}$ have opposite signs, as is typical, their values being 1 and -1.

A ***center*** with trajectories $y_1^2 + y_2^2/4 = \text{const}$ (ellipses) is obtained for $y_1' = y_2$, ${y_2}' = -4y_1$ with matrix

```
> A := Matrix([[0, 1], [-4, 0]]);
```

$$A := \begin{bmatrix} 0 & 1 \\ -4 & 0 \end{bmatrix}$$

for which $p = 0$ and $q = 4 > 0$, which gives a center. The eigenvalues of $\mathbf{A}$ are pure imaginary, which is typical, the values being $-2i$ and $2i$.

```
> Eigenvalues(A);
```

$$\begin{bmatrix} 2\,\mathrm{I} \\ -2\,\mathrm{I} \end{bmatrix}$$

A ***spiral point*** with trajectories $r = c\,\exp(-\theta)$ (spirals, in polar coordinates) is obtained for the system with matrix

```
> A := Matrix([[-1, 1], [-1, -1]]);
```

$$A := \begin{bmatrix} -1 & 1 \\ -1 & -1 \end{bmatrix}$$

for which $p = -2 \neq 0$ and $\delta = p^2 - 4q = 4 - 8 < 0$, which gives a spiral point. In this case, **A** has complex eigenvalues,

```
> Eigenvalues(A);
```

$$\begin{bmatrix} -1 + \mathrm{I} \\ -1 - \mathrm{I} \end{bmatrix}$$

Similar Material in AEM: Sec. 4.3

EXAMPLE 4.5 PENDULUM EQUATION

The ODE governing the motion of a pendulum (e.g., in a pendulum clock) is $y'' + k \sin y = 0$ (here we have disregarded damping) and is a classical case of an ODE that is not exactly solvable in terms of finitely many elementary functions. Physically, $y(t)$ is the angle of displacement from rest (the position when the pendulum is vertical), and t is time. k is a positive constant (we choose $k = 1$ for simplicity). We set $y = y_1$ and $y' = y_1' = y_2$. Then $y'' = y_2' = -k \sin y = -\sin y = -\sin y_1$. Hence you can write the given equation (with $k = 1$) as a system $y_1' = y_2$, ${y_2}' = -\sin y_1$. This illustrates the standard method of ***transforming an ODE of second order into a system of two first-order ODEs.*** In Maple, type the system as

```
> with(DEtools):
```

```
> sys := D(y1)(t) = y2(t), D(y2)(t) = -2*sin(y1(t));
```

$$sys := \mathrm{D}(y1)(t) = y2(t),\ \mathrm{D}(y2)(t) = -2\sin(y1(t))$$

Now plot a direction field of the system along with three approximate trajectories (phase plane representations of three solutions) that will turn out to be typical; specify each of them by one of the three initial conditions

```
> inits := [0, 0, 1], [0, 3.14, 0], [0, 4, 4];
```

where [0, 0, 1] means $t = 0$, $y_1 = 0$, $y_2 = 1$, etc.; thus, each condition specifies a t-value (0) and a point in the phase plane through which the trajectory should pass. Then plot. (For information, type `?DEplot`.)

```
> DEplot([sys[1], sys[2]], [y1(t), y2(t)], t = -10..10, y1 = -8..8,
    y2 = -5..5, [inits], linecolor=black, scaling = constrained,
    title = `Phase portrait of the undamped pendulum`);
```



Example 4.5. Direction field with typical trajectories for undamped pendulum

Critical points are at 0 and at $\pm 2\pi$, $\pm 4\pi$,... by periodicity. These are centers. One of the three solutions is typical of the circular-shaped closed curves surrounding 0. Further critical points are at π and at $-\pi$, $\pm 3\pi$, $\pm 5\pi$, ... by periodicity. These are saddle points. Your second curve passes (approximately) through two of them. The third kind of solution corresponds to swings of the pendulum through 360 degrees (full rotations), which, of course, an ordinary pendulum could not do.

Similar Material in AEM: Sec. 4.5

EXAMPLE 4.6 NONHOMOGENEOUS SYSTEM

Find a general solution of the non-homogeneous linear system

$$\mathbf{y}' = \mathbf{A}\mathbf{y} + \mathbf{g} = \begin{bmatrix} 4 & -8 \\ 2 & -6 \end{bmatrix} \mathbf{y} + \begin{bmatrix} 2t^2 + 8 \\ t^2 + (3/4)t + 6 \end{bmatrix}.$$

Solution. Type the system as shown and obtain a general solution by use of the command `dsolve`.

```
> with(LinearAlgebra):
```

```
> A := <<4, 2>|<-8,-6>>; g := <2*t^2+8, t^2+3/4*t+6>;
```

$$\begin{bmatrix} 4 & -8 \\ 2 & -6 \end{bmatrix}$$

$$\begin{bmatrix} 2t^2 + 8 \\ t^2 + \frac{3}{4}t + 6 \end{bmatrix}$$

```
> Ayg := A.<y1(t), y2(t)> + g;
```

$$\begin{bmatrix} 4y1(t) - 8y2(t) + 2t^2 + 8 \\ 2y1(t) - 6y2(t) + t^2 + \frac{3}{4}t + 6 \end{bmatrix}$$

```
> sys := D(y1)(t) = Ayg[1], D(y2)(t) = Ayg[2];
```

$$sys := \mathrm{D}(y1)(t) = 4\,y1(t) - 8\,y2(t) + 2\,t^2 + 8,\ \mathrm{D}(y2)(t) = 2\,y1(t) - 6\,y2(t) + t^2 + \frac{3}{4}\,t + 6$$

```
> sol := dsolve({sys}, {y1(t), y2(t)});
```

$$sol := \left\{ y1(t) = \mathrm{e}^{-4\,t}\_C2 + \mathrm{e}^{2\,t}\_C1 - \frac{1}{2}\,t^2 - \frac{1}{8},\right.$$
$$\left. y2(t) = \mathrm{e}^{-4\,t}\_C2 + \frac{1}{4}\mathrm{e}^{2\,t}\_C1 + \frac{1}{8}\,t + \frac{15}{16} \right\}$$

From `sol` you see that a particular solution of the *non-homogeneous* system is $y_{1\,p} = -1/8 - t^2/2$, $y_{2\,p} = t/8 + 15/16$.

Similar Material in AEM: Sec. 4.6

EXAMPLE 4.7 METHOD OF UNDETERMINED COEFFICIENTS

Find a particular solution of the non-homogeneous system in the previous example by the method of undetermined coefficients.

Solution. The system is $\mathbf{y}' = \mathbf{A}\mathbf{y} + \mathbf{g}$, where

$$\mathbf{g} = \begin{bmatrix} 2t^2 + 8 \\ t^2 + (3/4)t + 6 \end{bmatrix}.$$

The non-homogeneous part $\mathbf{g}$ suggests choosing $\mathbf{y}_p = \mathbf{u} + \mathbf{v}t + \mathbf{w}t^2$, with vectors $\mathbf{u} = [u_1,\, u_2]$, $\mathbf{v} = [v_1,\, v_2]$, $\mathbf{w} = [w_1,\, w_2]$ to be determined by substitution. Thus, type

```
> yp := [y1(t) = u1 + v1*t + w1*t^2, y2(t) = u2 + v2*t + w2*t^2];
```

$$yp := [y1(t) = t^2\,w1 + t\,v1 + u1,\, y2(t) = t^2\,w2 + t\,v2 + u2]$$

Here, `y1(t)` and `y2(t)` are the components of the vector `yp` to be determined, and you get them individually by typing

```
> r1 := rhs(yp[1]);                          # Resp. r1 := t^2 w1 + t v1 + u1
> r2 := rhs(yp[2]);                          # Resp. r2 := t^2 w2 + t v2 + u2
> with(LinearAlgebra):
> A := <<4, 2>|<-8,-6>>; g := <2*t^2+8, t^2+3/4*t+6>;
```

$$\begin{bmatrix} 4 & -8 \\ 2 & -6 \end{bmatrix}$$

$$\begin{bmatrix} 2t^2 + 8 \\ t^2 + \frac{3}{4}t + 6 \end{bmatrix}$$

```
> Arg := A.<r1, r2> + g;
```

$$\begin{bmatrix} 4t^2w1 - 8t^2v2 + 2t^2 + 4tv1 - 8tv2 + 4u1 - 8u2 + 8 \\ 2w1t^2 - 6v2t^2 + 2v1t - 6v2t + 2u1 - 6u2 + t^2 + \frac{3}{4}t + 6 \end{bmatrix}$$

Accordingly, type the system as

```
> sys := diff(r1, t) = Arg[1], diff(r2, t) = Arg[2];
```

$$sys := 2tw1 + v1 = 4\,t^2 w1 - 8t^2\,w2 + 2\,t^2 + 4\,tv1 - 8\,tv2 + 4\,u1 - 8\,u2 + 8,$$
$$2\,tw2 + v2 = 2\,w1\,t^2 - 6\,w2\,t^2 + 2\,v1\,t - 6\,v2\,t + 2\,u1 - 6\,u2 + t^2 + \frac{3}{4}\,t + 6$$

Determine the six unknown components of the three unknown vectors **u, v, w.** From `sys` get three vector equations `eq1`, `eq2`, `eq3` (involving the components of those unknown vectors) by setting $t = 0,\ 1,\ -1$,

```
> eq1 := eval(subs(t = 0, sys));
```

$$eq1 := \{v1 = 4\,u1 - 8 + 8\,u2, v2 = 6 + 2\,u1 - 6\,u2\}$$

```
> eq2 := eval(subs(t = 1, sys));
```

$$eq2 := \Big\{v1 + 2\,w1 = 4\,u1 - 8\,u2 + 4\,v1 - 8\,v2 + 4\,w1 - 8\,w2 + 10,$$
$$v2 + 2\,w2 = 2\,w1 - 6\,w2 + 2\,v1 + 6\,v2 + 2\,u1 - 6\,u2 + \frac{31}{4}\Big\}$$

```
> eq3 := eval(subs(t = -1, sys));
```

$$eq3 := \Big\{v1 - 2\,w1 = 4\,u1 - 8\,u2 - 4\,v1 + 8\,v2 + 4\,w1 - 8\,w2 + 10,$$
$$v2 - 2\,w2 = 2\,w1 - 6\,w2 - 2\,v1 + 6\,v2 + 2\,u1 - 6\,u2 + \frac{25}{4}\Big\}$$

Solve this linear system of six equations for the six components `eq1[1]`, `eq1[2]`, `eq2[1]`, `eq2[2]`, `eq3[1]`, `eq3[2]` of these three vector equations by the command

```
> s := solve({eq1[1], eq1[2], eq2[1], eq2[2], eq3[1], eq3[2]},
    u1, u2, v1, v2, w1, w2);
```

$$s := \left\{u1 = -\frac{1}{8}, u2 = \frac{15}{16}, v1 = 0, v2 = \frac{1}{8}, w1 = -\frac{1}{2}, w2 = 0\right\}$$

Substitution of these six values into `yp` gives the answer, in agreement with y_p in the previous example,

```
> subs(seq(s[n], n = 1..6), yp);
```

$$\left[y1(t) = -\frac{1}{2}\,t^2 - \frac{1}{8}, y2(t) = \frac{15}{16} + \frac{1}{8}\,t\right]$$

Similar Material in AEM: Sec. 4.6

EXAMPLE 4.8 VAN DER POL EQUATION. LIMIT CYCLE

The famous Van der Pol equation

$$y'' - \mu(1 - y^2)y' + y = 0 \qquad (\mu > 0)$$

models, for instance, vacuum tube circuits. It governs "**self-sustained oscillations**", that is, for small $y^2 < 1$ the second term is negative ("negative damping"), so that energy is fed into the physical system, whereas for $y^2 > 1$, we have damping that takes energy from the system and makes the amplitudes y to become smaller. This gives a **limit cycle** corresponding to a periodic solution), which the other trajectories are approaching. We take $\mu = 2$ and consider

$$y'' - 2(1 - y^2)y' + y = 0.$$

Convert the equation to a system of equations by the standard procedure (explained and also used in Example 3.5 in this Guide), namely, $y_1' = y_2$, $y_2' = (1 - y_1^2)y_2 - y_1$. Type this as

```
> mu := 2:
```

```
> sys := D(y1)(t) = y2(t), D(y2)(t) = mu*(1 - y1(t)^2)*y2(t) - y1(t);
```

$$sys := \mathrm{D}(y1)(t) = y2(t), \mathrm{D}(y2)(t) = 2\left(1 - y1(t)^2\right) y2(t) - y1(t)$$

Pick three points (triples t, y_1, y_2) through which your trajectories should pass, say,

```
> inits := [0, 0, 6], [0, 0, 1], [0, 0, 0.1];
```

These triples mean $t = 0$, $y_1 = 0$, $y_2 = 6$, etc. Then plot a set of phase portraits (for t going from 0 to 3, 0 to 6, and 0 to 10) by the commands (type `?DEplot` for information),

```
> with(DEtools):
```

```
> DEplot([sys[1], sys[2]], [y1(t), y2(t)], t = 0..3, y1 = -5..5,
   y2 = -6..8, [inits], linecolor=[black, blue, green],
   scaling = constrained, stepsize = 0.05);
```

```
> DEplot([sys[1], sys[2]], [y1(t), y2(t)], t = 0..6, y1 = -5..5,
   y2 = -6..8, [inits], linecolor=[black, blue, green],
   scaling = constrained, stepsize = 0.05);
```

```
> DEplot([sys[1], sys[2]], [y1(t), y2(t)], t = 0..10, y1 = -5..5,
   y2 = -6..8, [inits], linecolor=[black, blue, green],
   scaling = constrained, stepsize = 0.05);
```



$t = 0$ to 3 $\quad$ $t = 0$ to 6 $\quad$ $t = 0$ to 10

Example 4.8. Limit cycle and trajectories for the Van der Pol equation with $\mu = 2$

These three portraits are snapshots in time. They all start at different points on the y_2-axis but quickly move to a common trajectory. This is different from a center because the **limit cycle** is approached by trajectories from the interior and from the exterior. "Small solutions" increase, "large solutions" decrease, hence the existence of a common limit curve seems plausible.

For Maple help type `?odeadvisor`.

Similar Material in AEM: Sec. 4.5

Problem Set for Chapter 4

Pr.4.1 (Node) Using the criteria, show that $y_1' = 2y_1$, $y_2' = 3y_2$ has a node. Find a general solution. Plot a phase portrait. (*AEM* Sec. 4.3)

Pr.4.2 (Saddle point) Find and plot (by `DEplot`) the solution of the initial value problem $y_1' = 2\,y_1 - 3\,y_2$, $y_2' = (3/4)y_1 - 3\,y_2$, $y_1(0) = 10$, $y_2(0) = 0$ as a trajectory. (*AEM* Sec. 4.3)

Pr.4.3 (Center) Solve $y'' + 16y = 0$ by first converting it to a system of equations. Plot a phase portrait. (*AEM* Sec. 4.3)

Pr.4.4 (Spiral point) Find a general solution of the system $y_1' = -y_1 + 9y_2$, $y_2' = -4y_1 - y_2$. From it find the particular solution satisfying the initial conditions $y_1(0) = 1$, $y_2(0) = 1$. Plot it as a trajectory in the phase plane. (*AEM* Sec. 4.3)

Pr.4.5 (No basis of eigenvectors available) Determine the eigenvalues and eigenvectors of the coefficient matrix of the system

$$y_1' = 7\,y_1 + y_2, \qquad y_2' = -y_1 + 5\,y_2.$$

Show that this gives no basis of eigenvectors, so that you get at first only one solution. It is shown in AEM, Sec. 4.3, that a second independent solution is of the form

$$\mathbf{y} = (\mathbf{x}t + \mathbf{u})e^{\lambda\,t}.$$

Show that `dsolve` does indeed give a general solution of the corresponding form. (*AEM* Sec. 4.3)

Pr.4.6 (Damped pendulum) Add a damping term cy' to the pendulum equation in Example 4.5 in this Guide, choosing $c = 1/3$. Plot a phase portrait. Compare with Example 4.5. (*AEM* Sec. 4.5)

Pr.4.7 (Harmonic oscillations) Indicate the kind of trajectories of $y'' + (1/9)y = 0$ in the phase plane. (*AEM* Sec. 4.4)

Pr.4.8 (Mixing problem involving two tanks) Initially, tank T_1 contains 100 gal of pure water and tank T_2 contains 100 gal of water in which 200 lb of fertilizer are dissolved. Liquid circulates through the tanks (by means of two tubes connecting the tanks) at a constant rate of 4 gal/min and the mixture is kept uniform by stirring. Find the amounts of fertilizer $y_1(t)$ and $y_2(t)$ in T_1 and T_2, respectively, where t is time. (*AEM* Sec. 4.1)

Pr.4.9 (Electrical network) Find and plot the currents $i_1(t)$, $i_2(t)$ in a network governed by the equations

$$i_1' + 3(i_1 - i_2) = 24$$

$$8i_2 + 3(i_2 - i_1) + 4\int i_2\,dt = 0,$$

assuming $i_1(0) = 0$, $i_2(0) = 0$. The equations result from the two loops of the network. Loop 1 contains an inductor of $L = 1$ henry (giving i_1'), a resistor of $R = 3$ ohms (common to both loops), and a battery of 24 volts. Loop 2 contains another resistor of 8 ohms and a capacitor of 1/4 farad. (Differentiate the second equation to get rid of the integral.) (*AEM* Sec. 4.1)

Pr.4.10 (Electrical network) Find a general solution in Pr.4.9 by using matrices. (*AEM* Sec. 4.1)

Pr.4.11 (Damped oscillations) Indicate the kind of trajectories of $y'' + 8y' + 4y = 0$. (*AEM* Sec. 4.4)

Pr.4.12 (Duffing equation, soft spring) Plot a phase portrait of the Duffing equation with a soft spring, say, $y''+y-2y^3 = 0$. Include some trajectories. What kind of trajectories would you expect if $|y|$ is small? If it is large? Think it over before you plot. (*AEM* Sec. 4.5)

Pr.4.13 (Duffing equation, hard spring) Plot a phase portrait of the Duffing equation with a hard spring, say, $y''+y+2y^3 = 0$. Include some trajectories. Do they look physically reasonable? (*AEM* Sec. 4.5)

Pr.4.14 (Van der Pol equation) Plot a direction field for the equation with $\mu = 1/2$. Describe how it differs from that for $\mu = 2$ in Example 4.8 in this Guide. (*AEM* Sec. 4.5)

Pr.4.15 (Experiment on Van der Pol equation) Experiment with the Van der Pol equation in the previous problem, to find out how the shape of the limit cycle changes when you change μ.

Chapter 5

Series Solutions of ODEs. Special Functions

Content. Solving in series, plotting (Ex. 5.1, Prs. 5.1–5.3)
Legendre's ODE, Fourier–Legendre series (Exs. 5.2, 5.3, Prs. 5.4–5.6)
Frobenius method, hypergeometric ODE (Ex. 5.4, Prs. 5.7–5.11)
Bessel's ODE (Ex. 5.5, Prs. 5.12–5.17)

Series solutions of ODEs are obtained by `dsolve({ode, y(0) = K0, D(y)(0) = K1}, y(x), series)`, as explained in Example 5.1, along with **plotting** and **calculating** values from series.
Obtaining series independently of ODEs. Use `series(f( x - a), b)`, `a` the center, `b` the order of the error term. Type `?series`, `?powseries`. For instance, the **sine series** (center at $x = \pi/2$ and truncation order 10) is obtained by typing

```
> series(sin(x - Pi/2), x = 0, 10);
```

$$-1 + \frac{1}{2}x^2 - \frac{1}{24}x^4 + \frac{1}{720}x^6 - \frac{1}{40320}x^8 + O\left(x^{10}\right)$$

Or use `seq` for a sequence and `sum` for a sum (a polynomial). For example,

```
> sum((x - a)^n, n = 0..8);
```

$$1 + x - a + (x-a)^2 + (x-a)^3 + (x-a)^4 + (x-a)^5 + (x-a)^6 + (x-a)^7 + (x-a)^8$$

Special functions, elementary and higher, are known to Maple. For instance, `series(BesselJ(0, x), x, 20)` gives the (Maclaurin) series of the Bessel function $J_0(x)$, etc. Try it. See Examples 5.2-5.5.

Orthopoly package. This package handles the Legendre polynomials (called by `P(n, x)`) and other orthogonal polynomials. Type `?orthopoly`. See Example 5.2 in this Guide.

Examples for Chapter 5

EXAMPLE 5.1 POWER SERIES SOLUTIONS. PLOTS FROM THEM. NUMERIC VALUES

Find a **power series solution** of the initial value problem $(1-x^2)\,y'' - 2xy' + 90y = 0$, $y(0) = 0$, $y'(0) = 1$. Plot the solution and find its value at $x = 0.8$.

Solution. . Type the ODE.

```
> ode := (1 - x^2)*diff(y(x), x, x) - 2*x*diff(y(x), x) + 90*y(x) = 0;
```

$$ode := \left(-x^2 + 1\right)\left(\frac{d^2}{dx^2}y(x)\right) - 2\,x\left(\frac{d}{dx}y(x)\right) + 90\,y(x) = 0$$

Type the number of terms of the series you want, say, by taking a remainder $O(x^{12})$. Then type `dsolve(...,series)`, including the initial conditions. This is a set of data, hence include it in braces `{...}`.

```
> Order := 12:
```

```
> sol := dsolve({ode, y(0) = 0, D(y)(0) = 1}, y(x), series);
```

$$sol := y(x) = x - \frac{44}{3}x^3 + \frac{286}{5}x^5 - \frac{572}{7}x^7 + \frac{2431}{63}x^9 + O\left(x^{12}\right)$$

Plotting. Convert this to a polynomial by dropping the remainder (the error term).

```
> p := convert(sol, polynom);
```

$$p := y(x) = x - \frac{44}{3}x^3 + \frac{286}{5}x^5 - \frac{572}{7}x^7 + \frac{2431}{63}x^9$$

```
> plot(p, x = -1..1);
Error, invalid input:  plot expects its 1st argument, p, to be of type
set, array, list, rtable, algebraic, procedure, And(`module`,
appliable), but received y(x) = -(44/3)*x^3 + (286/5)*x^5 - (572/7)*x^7
+(2431/63)*x^9
```

Type `rhs(p)`, suggesting 'right-hand side of `p`'. You will get the polynomial without the left-hand side `y(x) =`

`> rhs(p);` `# Resp.` $x - \frac{44}{3}x^3 + \frac{286}{5}x^5 - \frac{572}{7}x^7 + \frac{2431}{63}x^9$

```
> plot(rhs(p), x = -1..1);
```



Example 5.1. Plot of the series solution

Numeric values can be obtained by the command `subs`. For instance,

`> subs(x = 0.3, p);` `# Resp.` $y(0.3) = 0.02588459957$

The zeros of the polynomial `p` are

```
> evalf[6](fsolve(rhs(p) = 0, x));
```

$-0.968160, -0.836031, -0.613371, -0.324253, 0.0, 0.324253, 0.613371, 0.836031, 0.968160$

We mention that the ODE is Legendre's equation with parameter $n = 9$, but the series solution obtained is not a Legendre polynomial (to be discussed in the next example).

Similar Material in AEM: Sec. 5.3

EXAMPLE 5.2 LEGENDRE POLYNOMIALS. THE ORTHOPOLY PACKAGE. PROCEDURES

The Legendre polynomials $P_0(x)$, $P_1(x)$, $P_2(x)$, ... are, after the Bessel functions, probably the most important special functions in applications. In your work you have to load the package orthopoly. First type ?orthopoly[legendre] for information. Then load by typing

```
> with(orthopoly):
```

You can now obtain the Legendre polynomials by typing, for instance,

```
> P5 := P(5, x);
```
Resp. $P5 := \frac{63}{8}x^5 - \frac{35}{4}x^3 + \frac{15}{8}x$

A formula for defining any of these polynomials is ***Rodrigues's formula***

$$P_n(x) = \frac{1}{2^n\, n!}\frac{d^n}{d\,x^n}[(x^2-1)^n].$$

This can be used for obtaining individual Legendre polynomials by what is called a ***procedural definition*** or simply a **procedure**. For information, type ?proc and ?procedure[paramtype]. In the simplest case, type,

```
> f := proc(x)
    x^3
  end:
```

```
> f(2.5);
```
Resp. 15.625

Do a few examples of your own, before you go on.

Using Rodrigues's formula, type a procedure for the Legendre polynomials.

```
> Leg := proc(n)
    1/(2^n*n!)*diff((x^2 - 1)^n, x$n)
  end:
```

x$n means x, x, ..., x, that is, **differentiate *n* times**. You can now simply type Leg(1), Leg(2),

```
> Leg(5);
```
Resp. $x^5 + 5\left(x^2-1\right)x^3 + \frac{15}{8}\left(x^2-1\right)^2 x$

```
> simplify(%);
```
Resp. $\frac{63}{8}x^5 - \frac{35}{4}x^3 + \frac{15}{8}x$

and so on. Now if you type Leg(0), which is $P_0 = 1$, you get

```
> Leg(0);
Error, (in Leg) invalid input:  diff expects 2 or more arguments, but
received 1
```

that is, Maple does not interpret the zeroth derivative as the function itself. But proc is flexible; to include $n = 0$, extend the procedure to

```
> Legendre := proc(n)
    if n = 0
    then
       1
    else
       1/(2^n*n!)*diff((x^2 - 1)^n, x$n)
    end if;
  end:
```

`end if` indicates the end of the `if-then-else` statement. Note further that a procedure `proc` needs `end:` (More complicated procedures will be considered in Chapter 19 of this Guide.) You now obtain the polynomials as before, including

```
> Legendre(0);                                    # Resp. 1
```

You may now put the procedure into the command `seq`, obtain the first few polynomials at once, and plot them on common axes. You can identify the polynomials by counting the number of zeros each one has. We get,

```
> S := seq(Legendre(n), n = 0..4);
```

$$S := 1,\ x,\ \frac{3}{2}x^2 - \frac{1}{2},\ x^3 + \frac{3}{2}\left(x^2-1\right)x,\ x^4 + 3\left(x^2-1\right)x^2 + \frac{3}{8}\left(x^2-1\right)^2$$

To obtain these polynomials in the usual form, use the commands `simplify` or `expand`. Type

```
> seq(expand(S[m]), m = 1..5);
```

$$1,\ x,\ \frac{3}{2}x^2 - \frac{1}{2},\ \frac{5}{2}x^3 - \frac{3}{2}x,\ \frac{35}{8}x^4 - \frac{15}{4}x^2 + \frac{3}{8}$$

```
> plot(S, x = -1.2..1.2, y = -1.1..1.1, xtickmarks = [-1, 0, 1],
   ytickmarks = [-1, -0.5, 0, 0.5, 1], title = `Legendre polynomials`);
```



Example 5.2. Legendre polynomials $P_0, \ldots, P_4$

Similar Material in AEM: Sec. 5.3

EXAMPLE 5.3 LEGENDRE'S DIFFERENTIAL EQUATION

The Legendre polynomials $P_n(x), n = 0,\ 1,\ 2,\ \ldots,$ just considered in Example 5.2

are solutions of **Legendre's differential equation**

$$(1-x^2)y'' - 2xy' + n(n+1)y = 0$$

satisfying suitable initial conditions (depending on n). Type

```
> restart:
> LEG := (1 - x^2)*diff(y(x), x, x) - 2*x*diff(y(x), x) + n*(n + 1)*y(x)
  = 0;
```

$$LEG := \left(-x^2+1\right)\left(\frac{d^2}{dx^2}y(x)\right) - 2\,x\left(\frac{d}{dx}y(x)\right) + n\,(n+1)\,y(x) = 0$$

For instance, consider $n = 6$,

```
> LEG6 := subs(n = 6, LEG);
```

$$LEG6 := \left(-x^2+1\right)\left(\frac{d^2}{dx^2}y(x)\right) - 2\,x\left(\frac{d}{dx}y(x)\right) + 42\,y(x) = 0$$

To obtain the particular solution $P_6(x)$ of this ODE, note that $P_6(0) = -5/16$ and its derivative at $x = 0$ is 0. Thus, type

```
> dsolve(LEG6, y(0) = -5/16, D(y)(0) = 0);
```

$$y(x) = -\frac{5}{16} + \frac{231}{16}\,x^6 - \frac{315}{16}\,x^4 + \frac{105}{16}\,x^2$$

Legendre polynomials are very convenient in connection with **boundary value problems** because at 1, they equal +1, while at −1 they equal 1 when n is even or −1 when n is odd (see the figure in Ex. 5.2). Thus you also obtain P_6 from

```
> dsolve(LEG6, y(-1) = 1, y(1) = 1);
```

$$y(x) = -\frac{5}{16} + \frac{231}{16}\,x^6 - \frac{315}{16}\,x^4 + \frac{105}{16}\,x^2$$

Coefficient recursion. For a power series solution of the Legendre equation, with arbitrary positive integer n, the coefficient recursion will involve three subsequent powers x^s, x^{s+1}, x^{s+2}. Accordingly, type

```
> ser := sum(a[m]*x^m, m = s..s + 2);
```

$$ser := a_s x^s + a_{1+s}\,x^{1+s} + a_{s+2}\,x^{s+2}$$

Substitute this into the Legendre equation, where `eval` evaluates the derivatives and `simplify` multiplies out $1 - x^2$ (the first coefficient of the Legendre equation) times the very long corresponding expression and $-2x$ (the other coefficient of the equation) times the corresponding long expression.

```
> eval(subs(y(x) = ser, LEG));
> simplify(%);
```

$$\begin{aligned}&-5x^{s+2}sa_{s+2} + x^{1+s}n^2a_{1+s} + x^{s+2}n^2a_{s+2} + x^sn^2a_s + x^ss^2a_{s+2} + x^{-1+s}s^2a_{1+s}\\ &\quad + x^{1+s}na_{1+s} + x^{s+2}na_{s+2} + x^sna_s + 3a_{s+2}x^ss + a_{1+s}x^{-1+s}s + 2a_{s+2}x^s - a_sx^ss\\ &\quad - 2a_{1+s}x^{1+s} - 6a_{s+2}x^{s+2} - x^{s+2}s^2a_{s+2} - x^{1+s}s^2a_{1+s} - 3x^{1+s}sa_{1+s} + a_sx^{s-2}s^2\\ &\quad - a_sx^{s-2}s - a_sx^ss^2 = 0\end{aligned}$$

The coefficient of x^s (a sum of expressions) is now obtained by the command

```
> coeff(lhs(%), x^s);
```

$$n^2a_s - s^2a_s + s^2a_{s+2} + na_s - sa_s + 3sa_{s+2} + 2a_{s+2}$$

This gives the coefficient recursion

```
> a[s+2] := solve(%, a[s+2]);
```
Resp. $a_{s+2} := -\frac{a_s\left(n^2-s^2+n-s\right)}{s^2+3\,s+2}$

```
> a[s+2] := factor(%);
```
Resp. $a_{s+2} := -\frac{a_s\,(n+1+s)\,(n-s)}{(s+2)\,(1+s)}$

This is the basic recursion relation for the coefficients of the power series solution of the Legendre equation with positive integer parameter n that leads to the Legendre polynomials.

Similar Material in AEM: Sec. 5.3

EXAMPLE 5.4 FROBENIUS METHOD

The equation $xy'' + 2y' + xy = 0$ requires the Frobenius method (why?). Try to identify the result in terms of known series. (*AEM* Sec. 5.4)

Solution. Type the ODE.

```
> restart:
> ode := x*diff(y(x), x, x) + 2*diff(y(x), x) + x*y(x) = 0;
> dsolve(ode);
```
Resp. $y(x) = \frac{\_C1\ \sin(x)}{x} + \frac{\_C2\ \cos(x)}{x}$

Use the first few terms to find the indicial equation.

```
> ser3 := sum(a[m]*x^(m+r), m = 0..2);
> ser3ode := x*diff(ser3, x, x) + 2*diff(ser3, x) + x*ser3;
```

$$ser3ode := x\left(\frac{a_0x^rr^2}{x^2} - \frac{a_0x^rr}{x^2} + \frac{a_1x^{1+r}\,(1+r)^2}{x^2} - \frac{a_1x^{1+r}\,(1+r)}{x^2} + \frac{a_2x^{2+r}\,(2+r)^2}{x^2}\right.$$
$$\left. - \frac{a_2x^{2+r}(2+r)}{x^2}\right) + 2\,\frac{a_0x^rr}{x} + 2\,\frac{a_1x^{1+r}\,(1+r)}{x} + 2\,\frac{a_2x^{2+r}(2+r)}{x}$$
$$+x\left(a_0x^r + a_1x^{1+r} + a_2x^{2+r}\right)$$

Comment 1 To find the coefficient of lowest degree term, we might try using `coeff` but, if we try it on `ser3`, we get

```
> coeff(ser3, x, r);
```
Resp. $\mathit{coeff}\left(a_0x^r + a_1x^{1+r} + a_2x^{2+r}, x, r\right)$

which is not what is needed. `coeff` does not seem to work with a variable exponent so we try a little trick to create a polynomial (in `A`).

```
> tempA := subs({x^(r-1) = A, x^(r) = A^2, x^(r+1) = A^3,
  x^(r+2) = A^4, x^(r+3) = A^5}, simplify(ser3ode));
```

$$tempA := A^5a_2 + A^3r^2a_2 + A^4a_1 + 5\,A^3ra_2 + A^2r^2a_1 + A^3a_0 + 6\,A^3a_2 + 3\,A^2ra_1$$
$$+Ar^2a_0 + 2\,A^2a_1 + Ara_0$$

from which

```
> Low = coeff(tempA, A)
```
Resp. $\mathtt{Low} := \mathtt{r}\,\mathtt{a}_0 + \mathtt{r}^2\,\mathtt{a}_0$

as desired. Find values for `r` and the minimum value

```
> rt = solve(Low, r)
```
Resp. $\mathtt{rt} := -1, 0$

```
> rm = min(rt)
```
Resp. -1

In this case the complete solution arises from the use of the smaller index so find the recurrence using `rm`

```
> ser := sum(a[s]*x^(s+rm), s = m - 1..m + 2);
```

$$ser := a_{m-1}x^{-2+m} + a_m x^{m-1} + a_{1+m}x^m + a_{m+2}x^{1+m}$$

Comment 2. After substituting this into the ODE the coefficients of x^n need to be found. One way to achieve this is to note that `x ser′` increases the degree of each term of the polynomial by 1, `ser′` decreases the degree by 1, `x ser'` would decrease it by 0, `ser″` decreases it by 2, `x ser″` by 1, and so forth. Because $a_m x^m$ is the third term in `ser`, the equation for the recurrence relation comes from

```
> Rec := x*diff(op(4, ser), x, x) + 2*diff(op(4, ser), x) + x*op(2, ser);
```

$$Rec := x\left(\frac{a_{m+2}x^{1+m}(1+m)^2}{x^2} - \frac{a_{m+2}x^{1+m}(1+m)}{x^2}\right) + 2\,\frac{a_{m+2}x^{1+m}(1+m)}{x} + x a_m x^{m-1}$$

```
> Rec1 := expand(Rec);
```

$$Rec1 := a_{m+2}x^m m^2 + 3\,a_{m+2}x^m m + 2\,a_{m+2}x^m + a_m x^m$$

```
> am := solve(Rec1, a[m+2]);                # Resp.
```

$$am := -\frac{a_m}{m^2 + 3\,m + 2}$$

```
> Rec2 := factor(am);                       # Resp.
```

$$Rec2 := -\frac{a_m}{(m+2)\,(1+m)}$$

```
> seq({a[m+2], Rec2}, m = 0..4, 2);
```

$$\left\{a_2, -\frac{1}{2}\,a_0\right\},\ \left\{a_4, -\frac{1}{12}\,a_2\right\},\ \left\{a_6, -\frac{1}{30}\,a_4\right\}$$

The problem with this is that the denominators are shown as products and so it become difficult to see what is happening. Again, we can apply a trick by decomposing the expression

```
> temp := numer(Rec2)/ ([op(1, denom(Rec2))].[op(2, denom(Rec2))]);
```

$$temp := -\frac{a_m}{[m+2]\,.\,[1+m]}$$

```
> SE0 := seq({a[m + 2], temp}, m = 0..4, 2);
```

$$SE0 := \left\{a_2, -\frac{a_0}{[2]\,.\,[1]}\right\},\ \left\{a_4, -\frac{a_2}{[4]\,.\,[3]}\right\},\ \left\{a_6, -\frac{a_4}{[6]\,.\,[5]}\right\}$$

From which

$$a_6 = -\frac{a_4}{6\cdot 5} = (-1)^2\frac{a_2}{6\,.\,5\,.\,4\,.\,3} = (-1)^3\frac{a_0}{6\,.\,5\,.\,4\,.\,3\,.\,2\,.\,1} = (-1)^3\frac{a_0}{6!}$$

```
> SEn := [subs(m = 2*(n - 2),{a[m + 2], temp}), subs(m = 2*(n - 1),
  {a[m + 2], temp})];
```

$$SEn := \left[\left\{a_{2\,n-2}, -\frac{a_{2\,n-4}}{[2\,n-2]\,.\,[-3+2\,n]}\right\},\left\{a_{2\,n}, -\frac{a_{2\,n-2}}{[2\,n]\,.\,[-1+2\,n]}\right\}\right]$$

From which

$$a_{2n} = -\frac{a_{2n-2}}{2n\cdot(2n-1)} = (-1)^2\frac{a_{2n-4}}{2n\cdot(2n-1)\cdot(2n-2)\cdot(2n-3)} = \cdots$$

The coefficient a_{2n} of x^{2n} is $(-1)^n/(2n)!$ i.e. the series is that of a cosine

```
> SO1 := seq({a[m + 2], temp}, m = 1..7, 2);
```

$$SO1 := \left\{a_3, -\frac{a_1}{[3].[2]}\right\}, \left\{a_5, -\frac{a_3}{[5].[4]}\right\}, \left\{a_7, -\frac{a_5}{[7].[6]}\right\}, \left\{a_9, -\frac{a_7}{[9].[8]}\right\}$$

```
> SOn := [subs(m = 2*n - 3,{a[m + 2], temp}), subs(m = 2*n - 1,
  {a[m + 2], temp})];
```

$$SOn := \left[\left\{a_{-1+2n}, -\frac{a_{-3+2n}}{[-1+2n].[2n-2]}\right\}, \left\{a_{1+2n}, -\frac{a_{-1+2n}}{[1+2n].[2n]}\right\}\right]$$

The coefficient a_{2n+1} of x^{2n+1} is $1/(2n+1)$ i.e. the series is that of a sine.

These series are multiplied by $x^{rm} = x^{-1}$.

EXAMPLE 5.5 FROBENIUS METHOD

Solve the equation $(x^2 - x)y'' - xy' + y = 0$ by the Frobenius method. (*AEM* Sec. 5.4)

Solution. Type the ODE.

```
> restart:
> ode := (x^2 - x)*diff(y(x), x, x) - x*diff(y(x), x) + y(x) = 0;
> dsolve(ode);                            # Resp.
```

$y(x) = \_C1\, x + \_C2\, (\ln(x)\, x + 1)$

Put the first few terms to find the indicial equation.

```
> ser3 := add(a[m]*x^(m+r), m = 0..2);
> ser3ode := (x^2 - x)*diff(ser3, x, x) - x*diff(ser3, x) + ser3;
```

See Comment 1 Example 4 regarding finding the lowest degree term.

```
> tempA := subs({x^(r-1) = A, x^(r) = A^2, x^(r+1) = A^3,
  x^(r+2) = A^4, x^(r+3) = A^5}, simplify(ser3ode));
```

$$tempA := A^4 r^2 a_2 + 2A^4 r a_2 + A^3 r^2 a_1 - A^3 r^2 a_2 + A^4 a_2 - 3A^3 r a_2 + A^2 r^2 a_0 - A^2 r^2 a_1 - 2A^3 a_2 - 2A^2 r a_0 - A^2 r a_1 - Ar^2 a_0 + A^2 a_0 + Ara_0$$

```
> Low := coeff(tempA, A);                 # Resp.
```

$Low := -r^2 a_0 + r a_0$

Find values for `r`.

```
> rt := solve(Low = 0, r);                # Resp.
```

$rt := 0, 1$

In this case the smaller index will give the same solution as the larger so use $r = 1$.

```
> ser := sum(a[s]*x^(s+1), s = m - 2..m + 1);
```

$$ser := a_{-2+m}x^{m-1} + a_{m-1}x^m + a_m x^{1+m} + a_{1+m}x^{m+2}$$

```
> Rec := x^2*diff(op(3, ser), x, x) - x*diff(op(4, ser), x, x)
  - x*diff(op(3, ser), x) + op(3, ser);
```

$$Rec := x^2\left(\frac{a_m x^{1+m}(1+m)^2}{x^2} - \frac{a_m x^{1+m}(1+m)}{x^2}\right) - x\left(\frac{a_{1+m}x^{m+2}(m+2)^2}{x^2} - \frac{a_{1+m}x^{m+2}(m+2)}{x^2}\right) - a_m x^{1+m}(1+m) + a_m x^{1+m}$$

```
> Rec1 := expand(Rec);
```

$$Rec1 := x a_m x^m m^2 - x a_{1+m} x^m m^2 - 3\, x a_{1+m} x^m m - 2\, x a_{1+m} x^m$$

```
> solve(Rec1, a[m+1]);
```
Resp. $\dfrac{m^2 a_m}{m^2 + 3\,m + 2}$

```
> Rec2 := factor(%);
```
Resp. $Rec2 := \dfrac{m^2 a_m}{(m+2)\,(1+m)}$

```
> seq([a[m+1], Rec2], m = 0..4);
```

$$[a_1, 0],\ [a_2, \frac{1}{6}\, a_1],\ [a_3, \frac{1}{3}\, a_2],\ [a_4, \frac{9}{20}\, a_3],\ [a_5, \frac{8}{15}\, a_4]$$

From this, $a_1 = 0$, so $a_2 = 0$, $a_3 = 0, \ldots, a_n = 0$. Hence, the only nonzero coefficient is a_0. Because other constants will appear in the next part, set $a_0 = 1$.

The other solution is obtained by Reduction of Order, i.e.

```
> y1(x) := x^1;
```
Resp. $y1\,(x) := x$

```
> y2(x) = y1(x)*u(x);
```
Resp. $y2\,(x) = xu\,(x)$

```
> NewODE := (x^2 - x)*diff(y2(x), x, x) - x*diff(y2(x), x) + y2(x) = 0;
```

```
> uSol := dsolve(NewODE);
```
Resp. $uSol := y2\,(x) = \_C1\,x + \_C2\,(\ln(x)\,x + 1)$

```
> y1(x)*uSol;
```
Resp. $xy2\,(x) = x\,(\_C1\,x + \_C2\,(\ln(x)\,x + 1))$

EXAMPLE 5.6 BESSEL'S EQUATION. BESSEL FUNCTIONS

Bessel's differential equation

$$x^2\, y'' + xy' + (x^2 - \nu^2)y = 0$$

appears in various practical problems, in particular in those that exhibit cylindrical symmetry. ν (Greek nu) is a given real number (a real parameter suggested by the kind of problem). Type the equation as

```
> BES := x^2*diff(y(x), x, x) + x*diff(y(x), x)+(x^2 - nu^2)*y(x)=0;
```

$$BES := x^2 \left(\frac{d^2}{dx^2} y(x)\right) + x \left(\frac{d}{dx} y(x)\right) + \left(-\nu^2 + x^2\right) y(x) = 0$$

Obtain a general solution by the command

```
> sol := dsolve(BES);
```

$$sol := y(x) = \_C1\,\mathrm{BesselJ}\,(\nu,\, x) + \_C2\,\mathrm{BesselY}\,(\nu,\, x)$$

The first term involves the **Bessel function of the first kind** $J_\nu(x)$ and the second term the **Bessel function of the second kind** $Y_\nu(x)$. When ν is an integer, one writes n instead of ν.

The functions J_0 and J_1 are particularly important. J_0 satisfies Bessel's equation with parameter 0, that is,

```
> subs(nu = 0, BES)/x;
```
Resp. $\dfrac{x^2 \left(\frac{d^2}{dx^2} y(x)\right) + \left(\frac{d}{dx} y(x)\right) x + y(x) x^2}{x} = 0$

```
> BES0 := simplify(%);
```
Resp. $BES0 := x \left(\frac{d^2}{dx^2} y(x)\right) + y(x) x + \frac{d}{dx} y(x) = 0$

Maple did not drop the factor x automatically, hence our trick. From the general solution (or equally well from the Bessel equation with $n = 0$) you obtain J_0, its Maclaurin series, and the conversion of the response to a polynomial by the commands

```
> subs(_C1 = 1, _C2 = 0, nu = 0, sol);
```

$$y(x) = \mathrm{BesselJ}(0, x)$$

```
> ser0 := series(rhs(%), x, 12);
```

$$ser0 := 1 - \frac{1}{4}x^2 + \frac{1}{64}x^4 - \frac{1}{2304}x^6 + \frac{1}{147456}x^8 - \frac{1}{14745600}x^{10} + O\left(x^{12}\right)$$

```
> p0 := convert(%, polynom);
```

$$p0 := 1 - \frac{1}{4}x^2 + \frac{1}{64}x^4 - \frac{1}{2304}x^6 + \frac{1}{147456}x^8 - \frac{1}{14745600}x^{10}$$

Similarly for the Bessel function $J_1(x)$

```
> subs(_C1 = 1, _C2 = 0, nu = 1, sol);
```

$$y(x) = \mathrm{BesselJ}(1, x)$$

```
> ser1 := series(rhs(%), x, 12);
```

$$ser1 := \frac{1}{2}x - \frac{1}{16}x^3 + \frac{1}{384}x^5 - \frac{1}{18432}x^7 + \frac{1}{1474560}x^9 - \frac{1}{176947200}x^{11} + O\left(x^{12}\right)$$

```
> p1 := convert(%, polynom);
```

$$p1 := \frac{1}{2}x - \frac{1}{16}x^3 + \frac{1}{384}x^5 - \frac{1}{18432}x^7 + \frac{1}{1474560}x^9 - \frac{1}{176947200}x^{11}$$

Plot the two polynomials and the two functions jointly, so that you see that the approximation seems good for x to about 3 or 4. For larger x you would need more terms (but be careful as the power series is not valid if x gets too large). The other figure shows that the curves of the two functions look similar to those of cosine and sine, due to the similarity of the respective series.

```
> plot(p0, p1, BesselJ(0, x), BesselJ(1, x), x = 0..6, y = -0.8..1,
    xtickmarks = [1, 2, 3, 4, 5, 6], ytickmarks = [-0.5, 0, 0.5, 1]);
> plot(BesselJ(0, x), BesselJ(1, x), x = 0..15, y = -0.8..1,
    xtickmarks = [5, 10, 15], ytickmarks = [-0.5, 0, 0.5, 1]);
```



Example 5.5. Partial sum approximations of the Bessel functions J_0 and J_1



Example 5.5. Bessel functions J_0 (starting from 1) and J_1 (starting from 0)

Zeros of Bessel functions are often needed in applications (vibration of membranes, celestial mechanics, etc.) and have been extensively tabulated, as have the functions themselves, see Ref. [1] in Appendix 1 of this Guide. You obtain the zeros by the command `fsolve`. For instance, for the first positive zero of J_0 you obtain as an approximation (exact to 5 digits) and as an exact 10-digit value

```
> x01appr := fsolve(p0 = 0, x, 2..3);      # Resp. x01appr := 2.404792604
```

```
> x01 := fsolve(BesselJ(0, x) = 0, x, 2..3); # Resp. x01 := 2.404825558
```

For the second zero (near 5.5) you see, from the first figure, that you cannot expect great accuracy. Try it.

Finally, remember that `dsolve(.., series)` gives series directly. In the present case,

```
> Order := 8:
```

```
> dsolve(BES0, y(x), series);
```

$$y(x) = \_C1\left(1 - \frac{1}{4}x^2 + \frac{1}{64}x^4 - \frac{1}{2304}x^6 + O\left(x^8\right)\right) + \_C2\left(\ln(x)\left(1 - \frac{1}{4}x^2 + \frac{1}{64}x^4 - \frac{1}{2304}x^6 + O\left(x^8\right)\right) + \left(\frac{1}{4}x^2 - \frac{3}{128}x^4 + \frac{11}{13824}x^6 + O\left(x^8\right)\right)\right)$$

The second line results from the Bessel function of the second kind $Y_0(x)$ (as explained in Sec. 5.5 of AEM).

Similar Material in AEM: Secs. 5.4, 5.5

Problem Set for Chapter 5

Pr.5.1 (Power series) Find the Maclaurin series for $\csc x$ (with 5 nonzero terms). Find the coefficient of x^7 (as a decimal fraction with 6 digits) by applying a suitable command to that series. (*AEM* Sec. 5.1)

Pr.5.2 (Power series) Find the Maclaurin series of $f = \sin \pi x$ on the computer (powers up to x^{19}, inclusively). Plot f and the series jointly to see for what x the series gives useful approximations. (*AEM* Sec. 5.1)

Pr.5.3 (Coefficient recursion. Do-loop) Obtain the coefficient recursion for the power series solution of $y' = 4xy$ with $a_0 = 1$ and write the series thus obtained. (For another coefficient recursion, see Example 5.3 in this

Pr.5.4 (Legendre polynomials) Find and solve an initial value problem for which $P_6(x)$ is a solution. (*AEM* Sec. 5.3)

Pr.5.5 (Legendre's equation) Solve Legendre's equation with $n = 0$ by `dsolve` both as a series and in closed form. (*AEM* Sec. 5.3)

Pr.5.6 (Boundary value problem) Solve the boundary value problem consisting of the Legendre equation with parameter $n = 7$ and boundary values $y(-1) = -1$ and $y(1) = 1$. Plot the solution. (*AEM* Sec. 5.2)

Pr.5.7 (Frobenius method) Find the series solution for the Euler-Cauchy equation $x^2y'' + bxy' + cy = 0$ with arbitrary constant b and c.

Pr.5.8 (Frobenius method) Find a general solution of $(x-1)^2\, y'' + (x-1)\, y' - 9y = 0$. (*AEM* Sec. 5.4)

Pr.5.9 (Hypergeometric equation) Apply the command `dsolve` to the hypergeometric equation

$$x(1-x)y'' + [c - (a+b+1)\, x]\, y' - aby = 0.$$

In many cases, the solutions of this equation become elementary functions. Show that this is the case when $a = 1$, $b = 1$, $c = 2$, $x = -t$. As another case, consider $a = -n$, $c = b$, $x = -t$. (*AEM* Sec. 5.4)

Pr.5.10 (Frobenius method) Find a general solution of $x(x-1)\, y'' + (3\,x-1)\, y' + y = 0$. (*AEM* Sec. 5.4)

Pr.5.11 (Hypergeometric equation) Solve the initial value problem

$$x(1-x)y'' + (2-4x)y' - 2y = 0, \qquad y(0) = 1, \qquad y'(0) = 1.$$

Give the solution also in the form of a series. To what a, b, c in the general hypergeometric equation does the present equation correspond? (*AEM* Sec. 5.3)

Pr.5.12 (Bessel functions) The Bessel functions J_ν satisfy the basic relations

$$[x^\nu\, J_\nu(x)]' = x^\nu\, J_{\nu-1}(x), \qquad [x^{-\nu}\, J_\nu(x)]' = -x^{-\nu}\, J_{\nu+1}(x),$$

where $' = d/dx$. Try to obtain these relations on the computer, either directly or, if this will not work, in integrated form on both sides. (*AEM* Sec. 5.4)

Pr.5.13 (Zeros of Bessel functions) Find approximations of the first two negative zeros of $J_1(x)$ from the Maclaurin series, using powers up to x^{19}, and compare with the values obtained directly from $J_1(x) = 0$. (*AEM* Sec. 5.5)

Pr.5.14 (Experiment on asymptotics of Bessel functions) For large x the Bessel function $J_n(x)$ is approximately equal to

$$\sqrt{\frac{2}{\pi\, x}}\, \cos\left(x - \frac{n\,\pi}{2} - \frac{\pi}{4}\right).$$

Experiment with this formula for various n; find out empirically from the graphs how accurate the formula is for small integers n and for larger ones.
(*AEM* Sec. 5.4)

Pr.5.15 (Bessel functions Y_0 and Y_1) Solve the general Bessel equation

$$x^2\, y'' + xy' + (x^2 - \nu^2)y = 0$$

by `dsolve`. Obtain from the response the particular solutions Y_0 and Y_1 and plot them on common axes. (*AEM* Sec. 5.5)

Pr.5.16 (Bessel's equation) A large number of ODEs can be reduced to Bessel's equation. Show that $x^2y'' + xy' + (9x^6 - 1/9)y = 0$ is of this kind, by solving it by `dsolve`, which will give a general solution in terms of Bessel functions.
(*AEM* Sec. 5.5)

Pr.5.17 (Reduction to Bessel's equation) Find out from the solution by `dsolve` what transformation will reduce $y'' + x^2\, y = 0$ to Bessel's equation.
(*AEM* Sec. 5.5)